Verilog HDL

A guide to Digital Design and Synthesis

Samir Palnitkar

SunSoft Press
1996

Part 1 Basic Verilog Topics

1

Overview of Digital Design with Verilog HDL

Evolution of CAD, emergence of HDLs, typical HDL-based design flow, why Verilog HDL?, trends in HDLs.

2

Hierarchical Modeling Concepts

Top-down and bottom-up design methodology, differences between modules and module instances, parts of a simulation, design block, stimulus block.

3

Basic Concepts

Lexical conventions, data types, system tasks, compiler directives.

4

Modules and Ports

Module definition, port declaration, connecting ports, hierarchical name referencing.

5

Gate-Level Modeling

Modeling using basic Verilog gate primitives, description of **and/or** and **buf/not** type gates, rise, fall and turn-off delays, min, max, and typical delays.

6

Dataflow Modeling

Continuous assignments, delay specification, expressions, operators, operands, operator types.

7

Behavioral Modeling

Structured procedures, **initial** and **always**, blocking and nonblocking statements, delay control, event control, conditional statements, multiway branching, loops, sequential and parallel blocks.

8

Tasks and Functions

Differences between tasks and functions, declaration, invocation.

9

Useful Modeling Techniques

Procedural continuous assignments, overriding parameters, conditional compilation and execution, useful system tasks.

Overview of Digital Design with Verilog® HDL

1

1.1 Evolution of Computer Aided Digital Design

Digital circuit design has evolved rapidly over the last 25 years. The earliest digital circuits were designed with vacuum tubes and transistors. Integrated circuits were then invented where logic gates were placed on a single chip. The first integrated circuit (IC) chips were SSI (*Small Scale Integration*) chips where the gate count was very small. As technologies became sophisticated, designers were able to place circuits with hundreds of gates on a chip. These chips were called MSI (*Medium Scale Integration*) chips. With the advent of LSI (*Large Scale Integration*), designers could put thousands of gates on a single chip. At this point, design processes started getting very complicated, and designers felt the need to automate these processes. *Computer Aided Design* (CAD)[1] techniques began to evolve. Chip designers began to use circuit and logic simulation techniques to verify the functionality of building blocks of the order of about 100 transistors. The circuits were still tested on the breadboard, and the layout was done on paper or by hand on a graphic computer terminal.

With the advent of VLSI (*Very Large Scale Integration*) technology, designers could design single chips with more than 100,000 transistors. Because of the complexity of these circuits, it was not possible to verify these circuits on a breadboard. Computer-aided techniques became critical for verification and design of VLSI digital circuits. Computer programs to do automatic placement and routing of circuit layouts also became popular. The designers were now building gate-level digital circuits manually on graphic terminals. They would build small building blocks and then derive higher-level blocks from them. This process would

1. Technically, the term *Computer-Aided Design (CAD) tools* refers to back-end tools that perform functions related to place and route, and layout of the chip . The term *Computer-Aided Engineering (CAE) tools* refers to tools that are used for front-end processes such HDL simulation, logic synthesis and timing analysis. However, designers use the term CAD and CAE interchangeably. For the sake of simplicity, in this book, we will refer to all design tools as *CAD tools*.

continue until they had built the top-level block. Logic simulators came into existence to verify the functionality of these circuits before they were fabricated on chip.

As designs got larger and more complex, logic simulation assumed an important role in the design process. Designers could iron out functional bugs in the architecture before the chip was designed further.

1.2 Emergence of HDLs

For a long time, programming languages such as FORTRAN, Pascal, and C were being used to describe computer programs that were sequential in nature. Similarly, in the digital design field, designers felt the need for a standard language to describe digital circuits. Thus, *Hardware Description Languages* (HDLs) came into existence. HDLs allowed the designers to model the concurrency of processes found in hardware elements. Hardware description languages such as *Verilog HDL* and VHDL became popular. Verilog HDL originated in 1983 at Gateway Design Automation. Later, VHDL was developed under contract from DARPA. Both Verilog® and VHDL simulators to simulate large digital circuits quickly gained acceptance from designers.

Even though HDLs were popular for logic verification, designers had to manually translate the HDL-based design into a schematic circuit with interconnections between gates. The advent of logic synthesis in the late 1980s changed the design methodology radically. Digital circuits could be described at a *register transfer level (RTL)* by use of an HDL. Thus, the designer had to specify how the data flows between registers and how the design processes the data. The details of gates and their interconnections to implement the circuit were automatically extracted by logic synthesis tools from the RTL description.

Thus, logic synthesis pushed the HDLs into the forefront of digital design. Designers no longer had to manually place gates to build digital circuits. They could describe complex circuits at an abstract level in terms of functionality and data flow by designing those circuits in HDLs. Logic synthesis tools would implement the specified functionality in terms of gates and gate interconnections.

HDLs also began to be used for system-level design. HDLs were used for simulation of system boards, interconnect buses, FPGAs (Field Programmable Gate Arrays), and PALs (Programmable Array Logic). A common approach is to design each IC chip, using an HDL, and then verify system functionality via simulation.

1.3 Typical Design Flow

A typical design flow for designing VLSI IC circuits is shown in Figure 1-1. Unshaded blocks show the level of design representation; shaded blocks show processes in the design flow.



Figure 1-1 Typical Design Flow

The design flow shown in Figure 1-1 is typically used by designers who use HDLs. In any design, specifications are written first. Specifications describe abstractly the functionality, interface, and overall architecture of the digital circuit to be designed. At this point, the architects do not need to think about how they will implement this circuit. A behavioral description is then created to analyze the design in terms of functionality, performance, compliance to standards, and other high-level issues. Behavioral descriptions can be written with HDLs.

The behavioral description is manually converted to an RTL description in an HDL. The designer has to describe the data flow that will implement the desired digital circuit. From this point onward, the design process is done with the assistance of Computer-Aided Design (CAD) tools.

Logic synthesis tools convert the RTL description to a gate-level netlist. A gate-level netlist is a description of the circuit in terms of gates and connections between them. The gate-level netlist is input to an Automatic Place and Route tool, which creates a layout. The layout is verified and then fabricated on chip.

Thus, most digital design activity is concentrated on manually optimizing the RTL description of the circuit. After the RTL description is frozen, CAD tools are available to assist the designer in further processes. Designing at RTL level has shrunk design cycle times from years to a few months. It is also possible to do many design iterations in a short period of time.

Behavioral synthesis tools have begun to emerge recently. These tools can create RTL descriptions from a behavioral or algorithmic description of the circuit. As these tools mature, digital circuit design will become similar to high-level computer programming. Designers will simply implement the algorithm in an HDL at a very abstract level. CAD tools will help the designer convert the behavioral description to a final IC chip.

It is important to note that although CAD tools are available to automate the processes and cut design cycle times, the designer is still the person who controls how the tool will perform. CAD tools are also susceptible to the *"GIGO : Garbage In Garbage Out"* phenomenon. If used improperly, CAD tools will lead to inefficient designs. Thus, the designer still needs to understand the nuances of design methodologies, using CAD tools to obtain an optimized design.

1.4 Importance of HDLs

HDLs have many advantages compared to traditional schematic-based design.

- Designs can be described at a very abstract level by use of HDLs. Designers can write their RTL description without choosing a specific fabrication technology. Logic synthesis tools can automatically convert the design to

any fabrication technology. If a new technology emerges, designers do not need to redesign their circuit. They simply input the RTL description to the logic synthesis tool and create a new gate-level netlist, using the new fabrication technology. The logic synthesis tool will optimize the circuit in area and timing for the new technology.

- By describing designs in HDLs, functional verification of the design can be done early in the design cycle. Since designers work at the RTL level, they can optimize and modify the RTL description until it meets the desired functionality. Most design bugs are eliminated at this point. This cuts down design cycle time significantly because the probability of hitting a functional bug at a later time in the gate-level netlist or physical layout is minimized.
- Designing with HDLs is analogous to computer programming. A textual description with comments is an easier way to develop and debug circuits. This also provides a concise representation of the design, compared to gate-level schematics. Gate-level schematics are almost incomprehensible for very complex designs.

HDLs are most certainly a trend of the future. With rapidly increasing complexities of digital circuits and increasingly sophisticated CAD tools, HDLs will probably be the only method for large digital designs. No digital circuit designer can afford to ignore HDL-based design.

1.5 Popularity of Verilog HDL

Verilog HDL has evolved as a standard hardware description language. Verilog HDL offers many useful features for hardware design.

- Verilog HDL is a general-purpose hardware description language that is easy to learn and easy to use. It is similar in syntax to the C programming language. Designers with C programming experience will find it easy to learn Verilog HDL.
- Verilog HDL allows different levels of abstraction to be mixed in the same model. Thus, a designer can define a hardware model in terms of switches, gates, RTL, or behavioral code. Also, a designer needs to learn only one language for stimulus and hierarchical design.
- Most popular logic synthesis tools support Verilog HDL. This makes it the language of choice for designers.

- All fabrication vendors provide Verilog HDL libraries for postlogic synthesis simulation. Thus, designing a chip in Verilog HDL allows the widest choice of vendors.
- The Programming Language Interface (PLI) is a powerful feature that allows the user to write custom C code to interact with the internal data structures of Verilog. Designers can customize a Verilog HDL simulator to their needs with the PLI.

1.6 Trends in HDLs

The speed and complexity of digital circuits has increased rapidly. Designers have responded by designing at higher levels of abstraction. Designers have to think only in terms of functionality. CAD tools take care of the implementation details. With designer assistance, CAD tools have become sophisticated enough to do a close-to-optimum implementation.

The most popular trend currently is to design in HDL at an RTL level, because logic synthesis tools can create gate-level netlists from RTL level design. Behavioral synthesis has recently emerged. As these tools improve, designers will be able to design directly in terms of algorithms and the behavior of the circuit, and then use CAD tools to do the translation and optimization in each phase of the design. Behavioral modeling will be used more and more as behavioral synthesis matures. Until then, RTL design will remain very popular.

Formal verification techniques are also appearing on the horizon. Formal verification applies formal mathematical techniques to verify the correctness of Verilog HDL descriptions and to establish equivalency between RTL and gate-level netlists. However, the need to describe a design in Verilog HDL will not go away.

For very high speed and timing-critical circuits like microprocessors, the gate-level netlist provided by logic synthesis tools is not optimal. In such cases, designers often mix gate-level description directly into the RTL description to achieve optimum results. This practice is opposite to the high-level design paradigm, yet it is frequently used for high-speed designs because designers need to squeeze the last bit of timing out of circuits and CAD tools sometimes prove to be insufficient to achieve the desired results.

A trend that is emerging for system-level design is a mixed bottom-up methodology where the designers use either existing Verilog HDL modules, basic building blocks, or vendor-supplied core blocks to quickly bring up their system simulation. This is done to reduce development costs and compress design schedules. For example, consider a system that has a CPU, graphics chip, I/O

chip, and a system bus. The CPU designers would build the next-generation CPU themselves at an RTL level, but they would use behavioral models for the graphics chip and the I/O chip and would buy a vendor-supplied model for the system bus. Thus, the system-level simulation for the CPU could be up and running very quickly and long before the RTL descriptions for the graphics chip and the I/O chip are completed.

Hierarchical Modeling Concepts

2

Before we discuss the details of the Verilog language, we must first understand basic hierarchical modeling concepts in digital design. The designer must use a "good" design methodology to do efficient Verilog HDL-based design. In this chapter, we discuss typical design methodologies and illustrate how these concepts are translated to Verilog. A digital simulation is made up of various components. We talk about the components and their interconnections.

Learning Objectives

- Understand top-down and bottom-up design methodologies for digital design.
- Explain differences between modules and module instances i*n Verilog.
- Describe four levels of abstraction—behavioral, data flow, gate level, and switch level—to represent the same module.
- Describe components required for the simulation of a digital design. Define a stimulus block and a design block. Explain two methods of applying stimulus.

2.1 Design Methodologies

There are two basic types of digital design methodologies: a *top-down* design methodology and a *bottom-up* design methodology. In a top-down design methodology, we define the top-level block and identify the sub-blocks necessary to build the top-level block. We further subdivide the sub-blocks until we come to leaf cells, which are the cells that cannot further be divided. Figure 2-1 shows the top-down design process.



Figure 2-1 Top-down Design Methodology

In a bottom-up design methodology, we first identify the building blocks that are available to us. We build bigger cells, using these building blocks. These cells are then used for higher-level blocks until we build the top-level block in the design. Figure 2-2 shows the bottom-up design process.



Figure 2-2 Bottom-up Design Methodology

Typically, a *combination* of top-down and bottom-up flows is used. Design architects define the specifications of the top-level block. Logic designers decide how the design should be structured by breaking up the functionality into blocks and sub-blocks. At the same time, circuit designers are designing optimized circuits for leaf-level cells. They build higher-level cells by using these leaf cells.

The flow meets at an intermediate point where the switch-level circuit designers have created a library of leaf cells by using switches, and the logic level designers have designed from top-down until all modules are defined in terms of leaf cells.

To illustrate these hierarchical modeling concepts, let us consider the design of a negative edge-triggered 4-bit ripple carry counter described in Section 2.2, *4-bit Ripple Carry Counter*.

2.2 4-bit Ripple Carry Counter



Figure 2-3 Ripple Carry Counter

The ripple carry counter shown in Figure 2-3 is made up of negative edge-triggered toggle flip-flops (*T_FF*). Each of the *T_FFs* can be made up from negative edge-triggered D-flipflops (*D_FF*) and *inverters* (assuming *q_bar* output is not available on the *D_FF*), as shown in Figure 2-4.

reset	q_n	q_{n+1}
1	1	0
1	0	0
0	0	1
0	1	0
0	0	0



Figure 2-4 T-flipflop

Thus, the ripple carry counter is built in a hierarchical fashion by using building blocks. The diagram for the design hierarchy is shown in Figure 2-5.



Figure 2-5 Design Hierarchy

In a top-down design methodology, we first have to specify the functionality of the ripple carry counter, which is the top-level block. Then, we implement the counter with *T_FF*s. We build the *T_FF*s from the *D_FF* and an additional inverter gate. Thus, we break bigger blocks into smaller building sub-blocks until we decide that we cannot break up the blocks any further. A bottom-up methodology flows in the opposite direction. We combine small building blocks and build

bigger blocks; e.g., we could build *D_FF* from **and** and **or** gates, or we could build a custom *D_FF* from transistors. Thus, the bottom-up flow meets the top-down flow at the level of the *D_FF*.

2.3 Modules

We now relate these hierarchical modeling concepts to Verilog. Verilog provides the concept of a *module*. A module is the basic building block in Verilog. A module can be an element or a collection of lower-level design blocks. Typically, elements are grouped into modules to provide common functionality that is used at many places in the design. A module provides the necessary functionality to the higher-level block through its port interface (inputs and outputs), but hides the internal implementation. This allows the designer to modify module internals without affecting the rest of the design.

In Figure 2-5, *ripple carry counter, T_FF, D_FF* are examples of modules. In Verilog, a module is declared by the keyword **module**. A corresponding keyword **endmodule** must appear at the end of the module definition. Each module must have a *module_name*, which is the identifier for the module, and a *module_terminal_list*, which describes the input and output terminals of the module.

```
module <module_name> (<module_terminal_list>);

...
<module internals>
...
...
endmodule
```

Specifically, the T-flipflop could be defined as a module as follows:

```
module T_FF (q, clock, reset);
.
.
<functionality of T-flipflop>
.
.
endmodule
```

Verilog is both a behavioral and a structural language. Internals of each module can be defined at *four* levels of abstraction, depending on the needs of the design. The module behaves identically with the external environment irrespective of the level of abstraction at which the module is described. The internals of the module are hidden from the environment. Thus, the level of abstraction to describe a module can be changed without any change in the environment. These levels will be studied in detail in separate chapters later in the book. The levels are defined below.

- **Behavioral or algorithmic level**
  This is the highest level of abstraction provided by Verilog HDL. A module can be implemented in terms of the desired design algorithm without concern for the hardware implementation details. Designing at this level is very similar to C programming.
- **Dataflow level**
  At this level the module is designed by specifying the data flow. The designer is aware of how data flows between hardware registers and how the data is processed in the design.
- **Gate level**
  The module is implemented in terms of logic gates and interconnections between these gates. Design at this level is similar to describing a design in terms of a gate-level logic diagram.
- **Switch level**
  This is the lowest level of abstraction provided by Verilog. A module can be implemented in terms of switches, storage nodes, and the interconnections between them. Design at this level requires knowledge of switch-level implementation details.

Verilog allows the designer to mix and match all four levels of abstractions in a design. In the digital design community, the term *register transfer level (RTL)* is frequently used for a Verilog description that uses a combination of behavioral and dataflow constructs and is acceptable to logic synthesis tools.

If a design contains four modules, Verilog allows each of the modules to be written at a different level of abstraction. As the design matures, most modules are replaced with gate-level implementations.

Normally, the higher the level of abstraction, the more flexible and technology independent the design. As one goes lower toward switch-level design, the design becomes technology dependent and inflexible. A small modification can cause a significant number of changes in the design. Consider the analogy with C programming and assembly language programming. It is easier to program in a

higher-level language such as C. The program can be easily ported to any machine. However, if you design at the assembly level, the program is specific for that machine and cannot be easily ported to another machine.

2.4 Instances

A module provides a template from which you can create actual objects. When a module is invoked, Verilog creates a unique object from the template. Each object has its own name, variables, parameters and I/O interface. The process of creating objects from a module template is called *instantiation*, and the objects are called *instances*. In Example 2-1, the top-level block creates four instances from the T-flipflop (*T_FF*) template. Each *T_FF* instantiates a *D_FF* and an inverter gate. Each instance must be given a unique name. Note that `//` is used to denote single-line comments.

Example 2-1 Module Instantiation

```
// Define the top-level module called ripple carry
// counter. It instantiates 4 T-flipflops. Interconnections are
// shown in Section 2.2, 4-bit Ripple Carry Counter.
module ripple_carry_counter(q, clk, reset);

output [3:0] q; //I/O signals and vector declarations
                //will be explained later.
input clk, reset; //I/O signals will be explained later.

//Four instances of the module T_FF are created. Each has a unique
//name.Each instance is passed a set of signals. Notice, that
//each instance is a copy of the module T_FF.
T_FF tff0(q[0],clk, reset);
T_FF tff1(q[1],q[0], reset);
T_FF tff2(q[2],q[1], reset);
T_FF tff3(q[3],q[2], reset);

endmodule

// Define the module T_FF. It instantiates a D-flipflop. We assumed
// that module D-flipflop is defined elsewhere in the design. Refer
// to Figure 2-4 for interconnections.
module T_FF(q, clk, reset);

//Declarations to be explained later
output q;
```

Example 2-1 Module Instantiation (Continued)

```
input clk, reset;
wire d;

D_FF dff0(q, d, clk, reset); // Instantiate D_FF. Call it dff0.
not n1(d, q); // not gate is a Verilog primitive. Explained later.

endmodule
```

In Verilog, it is illegal to *nest* modules. One module definition cannot contain another module definition within the **module** and **endmodule** statements. Instead, a module definition can incorporate copies of other modules by instantiating them. It is important not to confuse module definitions and instances of a module. Module definitions simply specify how the module will work, its internals, and its interface. Modules must be instantiated for use in the design.

Example 2-2 shows an illegal module nesting where the module *T_FF* is defined inside the module definition of the ripple carry counter.

Example 2-2 Illegal Module Nesting

```
// Define the top-level module called ripple carry counter.
// It is illegal to define the module T_FF inside this module.
module ripple_carry_counter(q, clk, reset);
output [3:0] q;
input clk, reset;

   module T_FF(q, clock, reset);// ILLEGAL MODULE NESTING
   …
   <module T_FF internals>
   …
   endmodule // END OF ILLEGAL MODULE NESTING

endmodule
```

2.5 Components of a Simulation

Once a design block is completed, it must be tested. The functionality of the design block can be tested by applying stimulus and checking results. We call such a block the *stimulus* block. It is good practice to keep the stimulus and design blocks separate. The stimulus block can be written in Verilog. A separate

language is not required to describe stimulus. The stimulus block is also commonly called a *test bench*. Different test benches can be used to thoroughly test the design block.

Two styles of stimulus application are possible. In the first style, the stimulus block instantiates the design block and directly drives the signals in the design block. In Figure 2-6, the stimulus block becomes the top-level block. It manipulates signals *clk* and *reset*, and it checks and displays output signal *q*.



Figure 2-6 Stimulus Block Instantiates Design Block

The second style of applying stimulus is to instantiate both the stimulus and design blocks in a top-level dummy module. The stimulus block interacts with the design block only through the interface. This style of applying stimulus is shown in Figure 2-7. The stimulus module drives the signals *d_clk* and *d_reset*, which are connected to the signals *clk* and *reset* in the design block. It also checks and displays signal *c_q*, which is connected to the signal *q* in the design block. The function of top-level block is simply to instantiate the design and stimulus blocks.



Figure 2-7 Stimulus and Design Blocks Instantiated in a Dummy Top-Level Module

Either stimulus style can be used effectively.

2.6 Example

To illustrate the concepts discussed in the previous sections, let us build the complete simulation of a ripple carry counter. We will define the design block and the stimulus block. We will apply stimulus to the design block and monitor the outputs. As we develop the Verilog models, you do not need to understand the exact syntax of each construct at this stage. At this point, you should simply try to understand the design process. We discuss the syntax in much greater detail in the later chapters.

2.6.1 Design Block

We use a top-down design methodology. First, we write the Verilog description of the top-level design block (Example 2-3), which is the *ripple carry counter (see* Section 2.2, *4-bit Ripple Carry Counter*).

Example 2-3 Ripple Carry Counter Top Block

```
module ripple_carry_counter(q, clk, reset);

output [3:0] q;
input clk, reset;
```

Example 2-3 *Ripple Carry Counter Top Block (Continued)*

```
T_FF tff0(q[0],clk, reset);
T_FF tff1(q[1],q[0], reset);
T_FF tff2(q[2],q[1], reset);
T_FF tff3(q[3],q[2], reset);

endmodule
```

In the above module, four instances of the module *T_FF* (T-flipflop) are used. Therefore, we must now define (Example 2-4) the internals of the module *T_FF,* which was shown in Figure 2-4.

Example 2-4 *Flip-flop T_FF*

```
module T_FF(q, clk, reset);

output q;
input clk, reset;
wire d;

D_FF dff0(q, d, clk, reset);
not n1(d, q); // not is a Verilog-provided primitive. case sensitive
endmodule
```

Since *T_FF* instantiates *D_FF,* we must now define (Example 2-5) the internals of module *D_FF.* We assume asynchronous reset for the *D_FF.*F

Example 2-5 *Flip-flop D_F*

```
// module D_FF with synchronous reset
module D_FF(q, d, clk, reset);

output q;
input d, clk, reset;
reg q;

// Lots of new constructs. Ignore the functionality of the constructs.
// Concentrate on how the design block is built in a top-down fashion.
always @(posedge reset or negedge clk)
if (reset)
    q = 1'b0;
```

Example 2-5 Flip-flop D_F (Continued)

```
// module D_FF with synchronous reset
else
    q = d;

endmodule
```

All modules have been defined down to the lowest-level leaf cells in the design methodology. The design block is now complete.

2.6.2 Stimulus Block

We must now write the stimulus block to check if the ripple carry counter design is functioning correctly. In this case, we must control the signals *clk* and *reset* so that the regular function of the ripple carry counter and the asynchronous reset mechanism are both tested. We use the waveforms shown in Figure 2-8 to test the design. Waveforms for *clk*, *reset*, and 4-bit output *q* are shown. The cycle time for *clk* is 10 units; the *reset* signal stays up from time 0 to 15 and then goes up again from time 195 to 205. Output *q* counts from 0 to 15.



Figure 2-8 Stimulus and Output Waveforms

We are now ready to write the stimulus block (see Example 2-4) that will create the above waveforms. We will use the stimulus style shown in Figure 2-6. Do not worry about the Verilog syntax at this point. Simply concentrate on how the design block is instantiated in the stimulus block.

Example 2-4 Stimulus Block

```
module stimulus;

reg clk;
reg reset;
wire[3:0] q;

// instantiate the design block
ripple_carry_counter r1(q, clk, reset);

// Control the clk signal that drives the design block. Cycle time = 10
initial
    clk = 1'b0; //set clk to 0
always
    #5 clk = ~clk; //toggle clk every 5 time units

// Control the reset signal that drives the design block
// reset is asserted from 0 to 20 and from 200 to 220.
initial
begin
    reset = 1'b1;
    #15 reset = 1'b0;
    #180 reset = 1'b1;
    #10 reset = 1'b0;
    #20 $finish; //terminate the simulation
end

// Monitor the outputs
initial
    $monitor($time, " Output q = %d",  q);

endmodule
```

Once the stimulus block is completed, we are ready to run the simulation and verify the functional correctness of the design block. The output obtained when stimulus and design blocks are simulated is shown in Example 2-6.

Example 2-6 Output of the Simulation

```
  0 Output q =  0
 20 Output q =  1
 30 Output q =  2
 40 Output q =  3
 50 Output q =  4
 60 Output q =  5
 70 Output q =  6
 80 Output q =  7
 90 Output q =  8
100 Output q =  9
110 Output q = 10
120 Output q = 11
130 Output q = 12
140 Output q = 13
150 Output q = 14
160 Output q = 15
170 Output q =  0
180 Output q =  1
190 Output q =  2
195 Output q =  0
210 Output q =  1
220 Output q =  2
```

2.7 Summary

In this chapter we discussed the following concepts.

- Two kinds of design methodologies are used for digital design: top-down and bottom-up. A combination of these two methodologies is used in today's digital designs. As designs become very complex, it is important to follow these structured approaches to manage the design process.

- Modules are the basic building blocks in Verilog. Modules are used in a design by instantiation. An instance of a module has a unique identity and is different from other instances of the same module. Each instance has an independent copy of the internals of the module. It is important to understand the difference between modules and instances.
- There are two distinct components in a simulation: a design block and a stimulus block. A stimulus block is used to test the design block. The stimulus block is usually the top-level block. There are two different styles of applying stimulus to a design block.
- The example of the ripple carry counter explains the step-by-step process of building all the blocks required in a simulation.

This chapter is intended to give an understanding of the design process and how Verilog fits into the design process. The details of Verilog syntax are not important at this stage and will be dealt with in later chapters.

2.8 Exercises

1. An interconnect switch (*IS*) contains the following components, a shared memory (*MEM*), a system controller (*SC*) and a data crossbar (*Xbar*).
   a. Define the modules *MEM, SC,* and *Xbar,* using the **`module/endmodule`** keywords. You do not need to define the internals. Assume that the modules have no terminal lists.
   b. Define the module *IS*, using the **`module/endmodule`** keywords. Instantiate the modules *MEM, SC, Xbar* and call the instances *mem1, sc1,* and *xbar1,* respectively. You do not need to define the internals. Assume that the module *IS* has no terminals.
   c. Define a stimulus block (Top), using the **`module/endmodule`** keywords. Instantiate the design block *IS* and call the instance *is1*. This is the final step in building the simulation environment.
2. A 4-bit ripple carry adder (*Ripple_Add*) contains four 1-bit full adders (*FA*).
   a. Define the module *FA*. Do not define the internals or the terminal list.
   b. Define the module *Ripple_Add*. Do not define the internals or the terminal list. Instantiate four full adders of the type *FA* in the module *Ripple_Add* and call them *fa0, fa1, fa2,* and *fa3*.

Basic Concepts 3

In this chapter, we discuss the basic constructs and conventions in Verilog. These conventions and constructs are used throughout the later chapters. These conventions provide the necessary framework for Verilog HDL. Data types in Verilog model actual data storage and switch elements in hardware very closely. This chapter may seem dry, but understanding these concepts is a necessary foundation for the successive chapters.

Learning Objectives

- Understand lexical conventions for operators, comments, whitespace, numbers, strings, and identifiers.
- Define the logic value set and data types such as nets, registers, vectors, numbers, simulation time, arrays, parameters, memories, and strings.
- Identify useful system tasks for displaying and monitoring information, and for stopping and finishing the simulation.
- Learn basic compiler directives to define macros and include files.

3.1 Lexical Conventions

The basic lexical conventions used by Verilog HDL are similar to those in the C programming language. Verilog contains a stream of tokens. Tokens can be comments, delimiters, numbers, strings, identifiers, and keywords. Verilog HDL is a case-sensitive language. All keywords are in lowercase.

3.1.1 Whitespace

Blank spaces (\b) , tabs (\t) and newlines (\n) comprise the whitespace. Whitespace is ignored by Verilog except when it separates tokens. Whitespace is not ignored in strings.

3.1.2 Comments

Comments can be inserted in the code for readability and documentation. There are two ways to write comments. A one-line comment starts with "`//`". Verilog skips from that point to the end of line. A multiple-line comment starts with "`/*`" and ends with "`*/`". Multiple-line comments cannot be nested.

```
a = b && c; // This is a one-line comment

/* This is a multiple line
   comment */

/* This is /* an illegal */ comment */
```

3.1.3 Operators

Operators are of three types, *unary, binary,* and *ternary.* Unary operators precede the operand. Binary operators appear between two operands. Ternary operators have two separate operators that separate three operands.

```
a = ~ b; // ~ is a unary operator. b is the operand
a = b && c; // && is a binary operator. b and c are operands
a = b ? c : d; // ?: is a ternary operator. b, c and d are operands
```

3.1.4 Number Specification

There are two types of number specification in Verilog: sized and unsized.

Sized numbers

Sized numbers are represented as *<size>* '*<base format>* *<number>*.

<size> is written only in decimal and specifies the number of bits in the number. Legal base formats are decimal (*'d* or *'D*), hexadecimal (*'h* or *'H*), binary (*'b* or *'B*) and octal (*'o* or *'O*). The number is specified as consecutive digits from *0, 1, 2, 3, 4, 5, 6, 7, 8, 9, a, b, c, d, e, f.* Only a subset of these digits is legal for a particular base. Uppercase letters are legal for number specification.

```
4'b1111 // This is a 4-bit   binary number
12'habc // This is a 12-bit  hexadecimal number
16'd255 // This is a 16-bit  decimal number.
```

Unsized numbers

Numbers that are specified without a *<base format>* specification are decimal numbers by default. Numbers that are written without a *<size>* specification have a default number of bits that is simulator- and machine-specific (must be at least 32).

```
23456 // This is a 32-bit   decimal number by default
'hc3 // This is a 32-bit    hexadecimal number
'o21 // This is a 32-bit    octal number
```

X or Z values

Verilog has two symbols for unknown and high impedance values. These values are very important for modeling real circuits. An unknown value is denoted by an **x**. A high impedance value is denoted by **z**.

```
12'h13x // This is a 12-bit hex number; 4 least significant bits unknown
6'hx // This is a 6-bit hex number
32'bz // This is a 32-bit high impedance number
```

An **x** or **z** sets four bits for a number in the hexadecimal base, three bits for a number in the octal base, and one bit for a number in the binary base. If the most significant bit of a number is **0**, **x**, or **z**, the number is automatically extended to fill the most significant bits, respectively, with **0**, **x**, or **z**. This makes it easy to assign **x** or **z** to whole vector. If the most significant digit is **1**, then it is also zero extended.

Negative numbers

Negative numbers can be specified by putting a minus sign before the size for a constant number. Size constants are always positive. It is illegal to have a minus sign between <*base format*> and <*number*>.

```
-6'd3 // 8-bit   negative number stored as 2's complement of 3
4'd-2 // Illegal specification
```

Underscore characters and question marks

An underscore character "_" is allowed anywhere in a number except the first character. Underscore characters are allowed only to improve readability of numbers and are ignored by Verilog.

A question mark "**?**" is the Verilog HDL alternative for **z** in the context of numbers. The **?** is used to enhance readability in the **casex** and **casez** statements discussed in Chapter 7, *Behavioral Modeling*, where the high impedance value is a don't care condition. (Note that **?** has a different meaning in the context of user-defined primitives, which are discussed in Chapter 12, *User-Defined Primitives*.)

```
12'b1111_0000_1010 // Use of underline characters for readability
4'b10?? // Equivalent of a 4'b10zz
```

3.1.5 Strings

A string is a sequence of characters that are enclosed by double quotes. The restriction on a string is that it must be contained on a single line, that is, without a carriage return. It cannot be on multiple lines. Strings are treated as a sequence of one-byte ASCII values.

```
"Hello Verilog World" // is a string
"a / b" // is a string
```

3.1.6 Identifiers and Keywords

Keywords are special identifiers reserved to define the language constructs. Keywords are in lowercase. A list of all keywords in Verilog is contained in Appendix C, *List of Keywords, System Tasks, and Compiler Directives*.

Identifiers are names given to objects so that they can be referenced in the design. Identifiers are made up of alphanumeric characters, the underscore (_) and the dollar sign ($) and are case sensitive. Identifiers start with an alphabetic character or an underscore. They cannot start with a number or a $ sign (The $ sign as the first character is reserved for system tasks, which are explained later in the book).

```
reg value; // reg is a keyword; value is an identifier
input clk; // input is a keyword, clk is an identifier
```

3.1.7 Escaped Identifiers

Escaped identifiers begin with the backslash (\ ) character and end with whitespace (space, tab, or newline). All characters between backslash and whitespace are processed literally. Any printable ASCII character can be included in escaped identifiers. The backslash or whitespace is not considered a part of the identifier.

```
\a+b-c
\**my_name**
```

3.2 Data Types

This section discusses the data types used in Verilog.

3.2.1 Value Set

Verilog supports four values and eight strengths to model the functionality of real hardware. The four value levels are listed in Table 3-1.

Table 3-1 Value Levels

Value Level	Condition in Hardware Circuits
0	Logic zero, false condition
1	Logic one, true condition
x	Unknown value
z	High impedance, floating state

In addition to logic values, strength levels are often used to resolve conflicts between drivers of different strengths in digital circuits. Value levels **0** and **1** can have the strength levels listed in Table 3-2.

Table 3-2 Strength Levels

Strength Level	Type	Degree
supply	Driving	strongest
strong	Driving	
pull	Driving	
large	Storage	
weak	Driving	
medium	Storage	
small	Storage	
highz	High Impedance	weakest

If two signals of unequal strengths are driven on a wire, the stronger signal prevails. For example, if two signals of strength **strong1** and **weak0** contend, the result is resolved as a **strong1**. If two signals of equal strengths are driven on a wire, the result is unknown. If two signals of strength **strong1** and **strong0** conflict, the result is an **x**. Strength levels are particularly useful for accurate modeling of signal contention, MOS devices, dynamic MOS, and other low-level devices. Only **trireg** nets can have storage strengths **large**, **medium**, and **small**. Detailed information about strength modeling is provided in Appendix A, *Strength Modeling and Advanced Net Definitions.*

3.2.2 Nets

Nets represent connections between hardware elements. Just as in real circuits, nets have values continuously driven on them by the outputs of devices that they are connected to. In Figure 3-1 net *a* is connected to the output of **and** gate *g1*. Net *a* will continuously assume the value computed at the output of gate *g1*, which is *b* & *c*.



Figure 3-1 Example of Nets

Nets are declared primarily with the keyword **wire**. Nets are one-bit values by default unless they are declared explicitly as vectors. The terms **wire** and net are often used interchangeably. The default value of a net is **z** (except the **trireg** net, which defaults to **x**). Nets get the output value of their drivers. If a net has no driver, it gets the value **z**.

```
wire a; // Declare net a for the above circuit
wire b,c; // Declare two wires b,c for the above circuit
wire d = 1'b0; // Net d is fixed to logic value 0 at declaration.
```

Note that *net* is not a keyword but represents a class of data types such as **wire**, **wand**, **wor**, **tri**, **triand**, **trior**, **trireg**, etc. The **wire** declaration is used most frequently. Other net declarations are discussed in Appendix A, *Strength Modeling and Advanced Net Definitions*.

3.2.3 Registers

Registers represent data storage elements. Registers retain value until another value is placed onto them. Do not confuse the term *registers* in Verilog with hardware registers built from edge-triggered flip-flops in real circuits. In Verilog, the term *register* merely means a variable that can hold a value. Unlike a net, a register does not need a driver. Verilog registers do not need a clock as hardware registers do. Values of registers can be changed anytime in a simulation by assigning a new value to the register.

Register data types are commonly declared by the keyword **reg**. The default value for a **reg** data type is **x**. An example of how registers are used is shown Example 3-1.

Example 3-1 Example of Register

```
reg reset; // declare a variable reset that can hold its value
initial // this construct will be discussed later
begin
  reset = 1'b1; //initialize reset to 1 to reset the digital circuit.
   #100 reset = 1'b0; // after 100 time units reset is deasserted.
end
```

3.2.4 Vectors

Nets or **reg** data types can be declared as vectors (multiple bit widths). If bit width is not specified, the default is scalar (1-bit).

```
wire a; // scalar net variable, default
wire [7:0] bus; // 8-bit   bus
wire [31:0] busA,busB,busC; // 3 buses of 32-bit width.
reg clock; // scalar register, default
reg [0:40] virtual_addr; // Vector register,virtual address 41 bits wide
```

Vectors can be declared at *[high# : low#]* or *[low# : high#]*, but the left number in the squared brackets is always the most significant bit of the vector. In the example shown above, bit 0 is the most significant bit of vector *virtual_addr.*

For the vector declarations shown above, it is possible to address bits or parts of vectors.

```
busA[7] // bit # 7 of vector busA
bus[2:0] // Three least significant bits of vector bus,
 // using bus[0:2] is illegal because the significant bit should
 // always be on the left of a range specification
virtual_addr[0:1] // Two most significant bits of vector virtual_addr
```

3.2.5 Integer, Real, and Time Register Data Types

integer, **real**, and **time** register data types are supported in Verilog.

Integer

An integer is a general purpose register data type used for manipulating quantities. Integers are declared by the keyword **integer**. Although it is possible to use **reg** as a general-purpose variable, it is more convenient to declare an **integer** variable for purposes such as counting. The default width for an integer is the host-machine word size, which is implementation specific but is at least 32 bits. Registers declared as data type **reg** store values as *unsigned* quantities, whereas integers store values as *signed* quantities.

```
integer counter; // general purpose variable used as a counter.
initial
   counter = -1; // A negative one is stored in the counter
```

Real

Real number constants and real register data types are declared with the keyword **real**. They can be specified in *decimal* notation (e.g., 3.14) or in *scientific* notation (e.g., 3e6, which is 3 x 10^6). Real numbers cannot have a range declaration, and their default value is 0. When a real value is assigned to an integer, the real number is rounded off to the nearest integer.

```
real delta; // Define a real variable called delta
initial
begin
   delta = 4e10; // delta is assigned in scientific notation
   delta = 2.13; // delta is assigned a value 2.13
end
integer i; // Define an integer i
initial
   i = delta; // i gets the value 2 (rounded value of 2.13)
```

Time

Verilog simulation is done with respect to *simulation time*. A special time register data type is used in Verilog to store simulation time. A time variable is declared with the keyword **time**. The width for time register data types is implementation specific but is at least 64 bits.The system function **$time** is invoked to get the current simulation time.

```
time save_sim_time; // Define a time variable save_sim_time
initial
   save_sim_time = $time; // Save the current simulation time
```

Simulation time is measured in terms of *simulation seconds*. The unit is denoted by *s*, the same as real time. However, the relationship between real time in the digital circuit and simulation time is left to the user. This is discussed in detail in Section 9.4, *Time Scales*.

3.2.6 Arrays

Arrays are allowed in Verilog for **reg**, **integer**, **time**, and *vector* register data types. Arrays are not allowed for real variables. Arrays are accessed by <*array_name*>**[**<*subscript*>**]**. Multidimensional arrays are not permitted in Verilog.

```
integer count[0:7]; // An array of 8 count variables
reg bool[31:0]; // Array of 32 one-bit boolean register variables
time chk_point[1:100]; // Array of 100 time checkpoint variables
reg [4:0] port_id[0:7]; // Array of 8 port_ids; each port_id is 5 bits wide
integer matrix[4:0][4:0]; // Illegal declaration. Multidimensional
array

count[5] // 5th element of array of count variables
chk_point[100] // 100th time check point value
port_id[3] // 3rd element of port_id array. This is a 5-bit value.
```

It is important not to confuse arrays with net or register vectors. A vector is a single element that is *n*-bits wide. On the other hand, arrays are multiple elements that are *1*-bit or *n*-bits wide.

3.2.7 Memories

In digital simulation, one often needs to model register files, RAMs, and ROMs. Memories are modeled in Verilog simply as an array of registers. Each element of the array is known as a word. Each word can be one or more bits. It is important to differentiate between *n* 1-bit registers and one *n*-bit register. A particular word in memory is obtained by using the address as a memory array subscript.

```
reg mem1bit[0:1023]; // Memory mem1bit with 1K 1-bit words
reg [7:0] membyte[0:1023];// Memory membyte with 1K 8-bit words (bytes)
membyte[511] // Fetches 1 byte word whose address is 511.
```

3.2.8 Parameters

Verilog allows constants to be defined in a module by the keyword **parameter**. Parameters cannot be used as variables. Parameter values for each module instance can be overridden individually at compile time. This allows the module instances to be customized. This aspect is discussed later.

```
parameter port_id = 5; // Defines a constant port_id
parameter cache_line_width = 256; // Constant defines width of cache line
```

Module definitions may be written in terms of parameters. Hardcoded numbers should be avoided. Parameters can be changed at module instantiation or by using the **defparam** statement, which is discussed in detail in Chapter 9, *Useful Modeling Techniques*. Thus, use of parameters makes the module definition flexible. Module behavior can be altered simply by changing the value of a parameter.

3.2.9 Strings

Strings can be stored in **reg**. The width of the register variables must be large enough to hold the string. Each character in the string takes up 8 bits (1 byte). If the width of the register is greater than the size of the string, Verilog fills bits to the left of the string with zeros. If the register width is smaller than the string width, Verilog truncates the leftmost bits of the string. It is always safe to declare a string that is slightly wider than necessary.

```
reg [8*18:1] string_value; // Declare a variable that is 18 bytes
wide
initial
   string_value = "Hello Verilog World"; // String can be stored
                                         // in variable
```

Special characters serve a special purpose in displaying strings, such as newline, tabs and displaying argument values. Special characters can be displayed in strings only when they are preceded by escape characters, as shown in Table 3-3.

Table 3-3 Special Characters

Escaped Characters	Character Displayed
\n	newline
\t	tab
%%	%
\\	\
\"	"
\ooo	Character written in 1–3 octal digits

3.3 System Tasks and Compiler Directives

In this section we introduce two special concepts used in Verilog: system tasks and compiler directives.

3.3.1 System Tasks

Verilog provides standard system tasks to do certain routine operations. All system tasks appear in the form **$<keyword>**. Operations such as displaying on the screen, monitoring values of nets, stopping, and finishing are done by system tasks. We will discuss only the most useful system tasks. Other tasks are listed in Verilog manuals provided by your simulator vendor or in the *Verilog HDL Language Reference Manual.*

Displaying information

$display is the main system task for displaying values of variables or strings or expressions. This is one of the most useful tasks in Verilog.

Usage: **$display**(*p1, p2, p3,....., pn);*

p1, p2, p3,..., pn can be quoted strings or variables or expressions. The format of **$display** is very similar to `printf` in C. A **$display** inserts a newline at the end of the string by default. A **$display** without any arguments produces a newline.

Strings can be formatted by using the format specifications listed in Table 3-4. For more detailed format specifications, see *Verilog HDL Language Reference Manual.*

Table 3-4 String Format Specifications

Format	Display
%d or %D	Display variable in decimal
%b or %B	Display variable in binary
%s or %S	Display string
%h or %H	Display variable in hex
%c or %C	Display ASCII character
%m or %M	Display hierarchical name (no argument required)
%v or %V	Display strength
%o or %O	Display variable in octal
%t or %T	Display in current time format
%e or %E	Display real number in scientific format (e.g., 3e10)
%f or %F	Display real number in decimal format (e.g., 2.13)
%g or %G	Display real number in scientific or decimal, whichever is shorter

Example 3-2 shows some examples of the **$display** task. If variables contain **x** or **z** values they are printed in the displayed string as **x** or **z**.

Example 3-2 $display Task

```
//Display the string in quotes
$display("Hello Verilog World");
-- Hello Verilog World

//Display value of current simulation time 230
$display($time);
-- 230

//Display value of 41-bit virtual address 1fe0000001c and time 200
reg [0:40] virtual_addr;
$display("At time %d virtual address is %h", $time, virtual_addr);
-- At time 200 virtual address is 1fe0000001c

//Display value of port_id 5 in binary
reg [4:0] port_id;
```

Example 3-2 $display Task (Continued)

```
$display("ID of the port is %b", port_id);
-- ID of the port is 00101

//Display x characters
//Display value of 4-bit bus 10xx (signal contention) in binary
reg [3:0] bus;
$display("Bus value is %b", bus);
-- Bus value is 10xx

//Display the hierarchical name of instance p1 instantiated under
//the highest-level module called top. No argument is required. This
//is a useful feature)
$display("This string is displayed from %m level of hierarchy");
-- This string is displayed from top.p1 level of hierarchy
```

Special characters are discussed in Section 3.2.9, *Strings*. Examples of displaying special characters in strings as discussed are shown in Example 3-3.

Example 3-3 Special Characters

```
//Display special characters, newline and %
$display("This is a \n multiline string with a %% sign");
-- This is a
-- multiline string with a % sign

//Display other special characters
```

Monitoring information

Verilog provides a mechanism to monitor a signal when its value changes. This facility is provided by the **$monitor** task.

Usage: **$monitor***(p1,p2,p3,....,pn);*

The parameters *p1, p2, ... , pn* can be variables, signal names, or quoted strings. A format similar to the **$display** task is used in the **$monitor** task. **$monitor** continuously monitors the values of the variables or signals specified in the parameter list and displays all parameters in the list whenever the value of any one variable or signal changes. Unlike **$display, $monitor** needs to be invoked only once.

Only one monitoring list can be active at a time. If there is more than one **$monitor** statement in your simulation, the last **$monitor** statement will be the active statement. The earlier **$monitor** statements will be overridden.

Two tasks are used to switch monitoring on and off.

Usage: **$monitoron;**

$monitoroff;

The **$monitoron** tasks enables monitoring, and the **$monitoroff** task disables monitoring during a simulation. Monitoring is turned on by default at the beginning of the simulation and can be controlled during the simulation with the **$monitoron** and **$monitoroff** tasks. Examples of monitoring statements are given in Example 3-4. Note the use of **$time** in the **$monitor** statement.

Example 3-4 Monitor Statement

```
//Monitor time and value of the signals clock and reset
//Clock toggles every 5 time units and reset goes down at 10 time units
initial
begin
   $monitor($time,
            " Value of signals clock = %b reset = %b", clock,reset);
end

Partial output of the monitor statement:
-- 0 Value of signals clock = 0 reset = 1
-- 5 Value of signals clock = 1 reset = 1
-- 10 Value of signals clock = 0 reset = 0
```

Stopping and finishing in a simulation

The task **$stop** is provided to stop during a simulation.

Usage: **$stop;**

The **$stop** task puts the simulation in an interactive mode. The designer can then debug the design from the interactive mode. The **$stop** task is used whenever the designer wants to suspend the simulation and examine the values of signals in the design.

The **$finish** task terminates the simulation.

Usage: **$finish;**

Examples of **$stop** and **$finish** are shown in Example 3-5.

Example 3-5 Stop and Finish Tasks

```
// Stop at time 100 in the simulation and examine the results
// Finish the simulation at time.
initial // to be explained later. time = 0
begin
clock = 0;
reset = 1;
#100 $stop; // This will suspend the simulation at time = 100
#900 $finish; // This will terminate the simulation at time = 1000
end
```

3.3.2 Compiler Directives

Compiler directives are provided in Verilog. All compiler directives are defined by using the **`<keyword>** construct. We deal with the two most useful compiler directives.

`define

The **`define** directive is used to define text macros in Verilog (see Example 3-6). This is similar to the *#define* construct in C. The defined constants or text macros are used in the Verilog code by preceding them with a ` (back tick). The Verilog compiler substitutes the text of the macro wherever it encounters a *`<macro_name>*.

Example 3-6 `define Directive

```
//define a text macro that defines default word size
//Used as `WORD_SIZE in the code
`define WORD_SIZE 32

//define an alias. A $stop will be substituted wherever `S appears
`define S $stop;

//define a frequently used text string
`define WORD_REG reg [31:0]
// you can then define a 32-bit register as `WORD_REG reg32;
```

`include

The **`include** directive allows you to include entire contents of a Verilog source file in another Verilog file during compilation. This works similarly to the *#include* in the C programming language. This directive is typically used to include header files, which typically contain global or commonly used definitions (see Example 3-7).

Example 3-7 `include Directive

```
// Include the file header.v, which contains declarations in the
// main verilog file design.v.
`include header.v
...
...
<Verilog code in file design.v>
...
...
```

Two other directives, **`ifdef** and **`timescale**, are used frequently. They are discussed in Chapter 9, *Useful Modeling Techniques*.

3.4 Summary

We discussed the basic concepts of Verilog in this chapter. These concepts lay the foundation for the material discussed in the further chapters.

- Verilog is similar in syntax to the C programming language . Hardware designers with previous C programming experience will find Verilog easy to learn.
- Lexical conventions for operators, comments, whitespace, numbers, strings, and identifiers were discussed.
- Various data types are available in Verilog. There are four logic values, each with different strength levels. Available data types include nets, registers, vectors, numbers, simulation time, arrays, memories, parameters, and strings. Data types represent actual hardware elements very closely.
- Verilog provides useful system tasks to do functions like displaying, monitoring, suspending, and finishing a simulation.

- Compiler directive `**define** is used to define text macros, and `**include** is used to include other Verilog files.

3.5 Exercises

1. Practice writing the following numbers
   a. Decimal number 123 as a sized 8-bit number in binary. Use _ for readability.
   b. A 16-bit hexadecimal unknown number with all **x**'s.
   c. A 4-bit negative 2 in decimal . Write the 2's complement form for this number.
   d. An unsized hex number *1234*.

2. Are the following legal strings? If not, write the correct strings.
   a. "This is a string displaying the % sign"
   b. "out = in1 + in2"
   c. "Please ring a bell \007"
   d. "This is a backslash \ character\n"
3. Are these legal identifiers?
   a. system1
   b. 1reg
   c. $latch
   d. exec$
4. Declare the following variables in Verilog.
   a. An 8-bit vector net called *a_in*.
   b. A 32-bit storage register called *address*. Bit 31 must be the most significant bit. Set the value of the register to a 32-bit decimal number equal to 3.
   c. An integer called *count*.
   d. A time variable called *snap_shot*.
   e. An array called *delays*. Array contains 20 elements of the type **integer**.
   f. A memory *MEM* containing 256 words of 64 bits each.
   g. A parameter *cache_size* equal to 512.

5. What would be the output/effect of the following statements?

   a. *latch = 4'd12;*
   $display(*"The current value of latch = %b\n", latch);*

   b. *in_reg = 3'd2;*
   **$monitor**(*$time, " In register value = %b\n", in_reg[2:0]);*

   c. **`define** *MEM_SIZE 1024*
   $display(*"The maximum memory size is %h", 'MEM_SIZE);*

Modules and Ports 4

In the previous chapters, we acquired an understanding of the fundamental hierarchical modeling concepts, basic conventions, and Verilog constructs. In this chapter, we take a closer look at modules and ports from the Verilog language point of view.

Learning Objectives

- Identify the components of a Verilog module definition, such as module names, port lists, parameters, variable declarations, dataflow statements, behavioral statements, instantiation of other modules, and tasks or functions.
- Understand how to define the port list for a module and declare it in Verilog.
- Describe the port connection rules in a module instantiation.
- Understand how to connect ports to external signals, by ordered list, and by name.
- Explain hierarchical name referencing of Verilog identifiers.

4.1 Modules

We discussed how a module is a basic building block in Chapter 2, *Hierarchical Modeling Concepts*. We ignored the internals of modules and concentrated on how modules are defined and instantiated. In this section we analyze the internals of the module in greater detail.

A module in Verilog consists of distinct parts, as shown in Figure 4-1.



Figure 4-1 Components of a Verilog Module

A module definition always begins with the keyword **module**. The *module name, port list, port declarations,* and optional *parameters* must come first in a module definition. *Port list* and *port declarations* are present only if the module has any ports to interact with the external environment.The five components within a module are - *variable declarations, dataflow statements, instantiation of lower modules, behavioral blocks,* and *tasks* or *functions*. These components can be in any order and at any place in the module definition. The **endmodule** statement must always come last in a module definition. All components except **module**, *module name,* and **endmodule** are optional and can be mixed and matched as per design needs. Verilog allows multiple modules to be defined in a single file. The modules can be defined in any order in the file.

To understand the components of a module shown above, let us consider a simple example of an *SR latch,* as shown in Figure 4-2.



Figure 4-2 SR Latch

The *SR latch* has *S* and *R* as the input ports and *Q* and *Qbar* as the output ports. The *SR latch* and its stimulus can be modeled as shown in Example 4-1.

Example 4-1 Components of SR Latch

```
// This example illustrates the different components of a module

// Module name and port list
// SR_latch module
module SR_latch(Q, Qbar, Sbar, Rbar);

//Port declarations
output Q, Qbar;
input Sbar, Rbar;

// Instantiate lower-level modules
// In this case, instantiate Verilog primitive nand gates
// Note, how the wires are connected in a cross-coupled fashion.
nand n1(Q, Sbar, Qbar);
nand n2(Qbar, Rbar, Q);

// endmodule statement
endmodule

// Module name and port list
// Stimulus module
module Top;

// Declarations of wire, reg, and other variables
```

Example 4-1 Components of SR Latch (Continued)

```
wire q, qbar;
reg set, reset;

// Instantiate lower-level modules
// In this case, instantiate SR_latch
// Feed inverted set and reset signals to the SR latch
SR_latch m1(q, qbar, ~set, ~reset);

// Behavioral block, initial
initial
begin
  $monitor($time, " set = %b, reset= %b, q= %b\n",set,reset,q);
  set = 0; reset = 0;
  #5 reset = 1;
  #5 reset = 0;
  #5 set = 1;
end

// endmodule statement
endmodule
```

Notice the following characteristics about the modules defined above.

- In the *SR latch* definition above , notice that all components described in Figure 4-1 need not be present in a module. We do not find *variable declarations, dataflow* (**assign**) *statements,* or *behavioral blocks* (**always** or **initial**).
- However, the stimulus block for the SR latch contains *module name, wire, reg,* and *variable declarations, instantiation of lower level modules, behavioral block* (**initial**), and **endmodule** statement but does not contain *port list, port declarations,* and *data flow* (**assign**) *statements.*
- Thus, all parts except **module**, *module name,* and **endmodule** are optional and can be mixed and matched as per design needs.

4.2 Ports

Ports provide the interface by which a module can communicate with its environment. For example, the *input/output* pins of an IC chip are its ports. The environment can interact with the module only through its ports. The internals of the module are not visible to the environment. This provides a very powerful flexibility to the designer. The internals of the module can be changed without affecting the environment as long as the interface is not modified. Ports are also referred to as *terminals*.

4.2.1 List of Ports

A module definition contains an optional list of ports. If the module does not exchange any signals with the environment, there are no ports in the list. Consider a 4-bit full adder that is instantiated inside a top-level module *Top*. The diagram for the *input/ output* ports is shown in Figure 4-3.



Figure 4-3 I/O Ports for Top and Full Adder

Notice that in the above figure, the module *Top* is a top-level module. The module *fulladd4* is instantiated below *Top*. The module *fulladd4* takes input on ports *a*, *b*, and *c_in* and produces an output on ports *sum* and *c_out*. Thus, module *fulladd4* performs an addition for its environment. The module *Top* is a top-level module in the simulation and does not need to pass signals to or receive signals from the environment. Thus, it does not have a list of ports. The module names and port lists for both module declarations in Verilog are as shown in Example 4-2.

Example 4-2 List of Ports

```
module fulladd4(sum, c_out, a, b, c_in); //Module with a list of ports
module Top; // No list of ports, top-level module in simulation
```

4.2.2 Port Declaration

All ports in the list of ports must be declared in the module. Ports can be declared as follows:

Verilog Keyword	Type of Port
input	Input port
output	Output port
inout	Bidirectional port

Each port in the port list is defined as **input**, **output**, or **inout**, based on the direction of the port signal. Thus, for the example of the *fulladd4* in Example 4-2, the port declarations will be as shown in Example 4-3.

Example 4-3 Port Declarations

```
module fulladd4(sum, c_out, a, b, c_in);

//Begin port declarations section
output[3:0] sum;
output c_cout;

input [3:0] a, b;
input c_in;
//End port declarations section
...
<module internals>
...
endmodule
```

Note that all port declarations are implicitly declared as **wire** in Verilog. Thus, if a port is intended to be a **wire**, it is sufficient to declare it as **output**, **input**, or **inout**. **Input** or **inout** ports are normally declared as **wire**s. However, if **output** ports hold their value, they must be declared as **reg**. For example, in the definition of *DFF*, in Example 2-5, we wanted the output *q* to retain its value until the next clock edge. The port declarations for *DFF* will look as shown in Example 4-4.

Example 4-4 *Port Declarations for DFF*

```
module DFF(q, d, clk, reset);
output q;
reg q; // Output port q holds value; therefore it is declared as reg.
input d, clk, reset;
...
...
endmodule
```

Ports of the type **input** and **inout** cannot be declared as **reg** because **reg** variables store values and input ports should not store values but simply reflect the changes in the external signals they are connected to.

4.2.3 Port Connection Rules

One can visualize a port as consisting of two units, one unit that is *internal* to the module another that is *external* to the module. The internal and external units are connected. There are rules governing port connections when modules are instantiated within other modules. The Verilog simulator complains if any port connection rules are violated. These rules are summarized in Figure 4-4.



Figure 4-4 *Port Connection Rules*

Inputs

Internally, input ports must always be of the type *net*. Externally, the inputs can be connected to a variable which is a **reg** or a *net*.

Outputs

Internally, outputs ports can be of the type **reg** or *net*. Externally, outputs must always be connected to a *net*. They cannot be connected to a **reg**.

Inouts

Internally, inout ports must always be of the type *net*. Externally, inout ports must always be connected to a *net*.

Width matching

It is legal to connect internal and external items of different sizes when making inter-module port connections. However, a warning is typically issued that the widths do not match.

Unconnected ports

Verilog allows ports to remain unconnected. For example, certain output ports might be simply for debugging, and you might not be interested in connecting them to the external signals. You can let a port remain unconnected by instantiating a module as shown below.

```
fulladd4 fa0(SUM, , A, B, C_IN); // Output port c_out is unconnected
```

Example of illegal port connection

To illustrate port connection rules, assume that the module *fulladd4* in Example 4-3 is instantiated in the stimulus block *Top*. An example of an illegal port connection is shown in Example 4-5.

Example 4-5 Illegal Port Connection

```
module Top;

//Declare connection variables
reg [3:0]A,B;
reg C_IN;
reg [3:0] SUM;
wire C_OUT;

    //Instantiate fulladd4, call it fa0
    fulladd4 fa0(SUM, C_OUT, A, B, C_IN);
    //Illegal connection because output port sum in module fulladd4
    //is connected to a register variable SUM in module Top.
```

Example 4-5 Illegal Port Connection (Continued)

```
    .
    .
    <stimulus>
    .
    .
endmodule
```

This problem is rectified if the variable *SUM* is declared as a *net* (**wire**). A similar problem would occur if an input port were declared as a **reg**.

4.2.4 Connecting Ports to External Signals

There are two methods of making connections between signals specified in the module instantiation and the ports in a module definition. The two methods cannot be mixed.

Connecting by ordered list

Connecting by ordered list is the most intuitive method for most beginners. The signals to be connected must appear in the module instantiation in the same order as the ports in the port list in the module definition. Once again, consider the module *fulladd4* defined in Example 4-3. To connect signals in module *Top* by ordered list, the Verilog code is shown in Example 4-6. Notice that the external signals *SUM, C_OUT, A, B,* and *C_IN* appear in exactly the same order as the ports *sum, c_out, a, b,* and *c_in* in module definition of *fulladd4*.

Example 4-6 Connection by Ordered List

```
module Top;

//Declare connection variables
reg [3:0]A,B;
reg C_IN;
wire [3:0] SUM;
wire C_OUT;

    //Instantiate fulladd4, call it fa_ordered.
    //Signals are connected to ports in order (by position)
    fulladd4 fa_ordered(SUM, C_OUT, A, B, C_IN);
    …
    <stimulus>
```

Example 4-6 Connection by Ordered List (Continued)

```
    ...
endmodule


module fulladd4(sum, c_out, a, b, c_in);
output[3:0] sum;
output c_cout;
input [3:0] a, b;
input c_in;
    ...
    <module internals>
    ...
endmodule
```

Connecting ports by name

For large designs where modules have, say, 50 ports, remembering the order of the ports in the module definition is impractical and error prone. Verilog provides the capability to connect external signals to ports by the port names, rather than by position. We could connect the ports by name in Example 4-6 above by instantiating the module *fulladd4*, as follows. Note that you can specify the port connections in any order as long as the port name in the module definition correctly matches the external signal.

```
// Instantiate module fa_byname and connect signals to ports by name
fulladd4 fa_byname(.c_out(C_OUT), .sum(SUM), .b(B), .c_in(C_IN),
.a(A),);
```

Note that only those ports that are to be connected to external signals must be specified in port connection by name. Unconnected ports can be dropped. For example, if the port *c_out* were to be kept unconnected, the instantiation of *fulladd4* would look as follows. The port *c_out* is simply dropped from the port list.

```
// Instantiate module fa_byname and connect signals to ports by name
fulladd4 fa_byname(.sum(SUM), .b(B), .c_in(C_IN), .a(A),);
```

Another advantage of connecting ports by name is that as long as the port name is not changed, the order of ports in the port list of a module can be rearranged without changing the port connections in module instantiations.

4.3 Hierarchical Names

We described earlier that Verilog supports a hierarchical design methodology. Every module instance, signal, or variable is defined with an *identifier*. A particular identifier has a unique place in the design hierarchy. *Hierarchical name referencing* allows us to denote every identifier in the design hierarchy with a unique name. A *hierarchical name* is a list of identifiers separated by dots (".") for each level of hierarchy. Thus, any identifier can be addressed from any place in the design by simply specifying the complete hierarchical name of that identifier.

The top-level module is called the *root* module because it is not instantiated anywhere. It is the starting point. To assign a unique name to an identifier, start from the top-level module and trace the path along the design hierarchy to the desired identifier. To clarify this process, let us consider the simulation of SR latch in Example 4-1. The design hierarchy is shown in Figure 4-5.



Figure 4-5 Design Hierarchy for SR Latch Simulation

For this simulation, *stimulus* is the top-level module. Since the top-level module is not instantiated anywhere, it is called the *root* module. The identifiers defined in this module are *q*, *qbar*, *set*, and *reset*. The root module instantiates *m1*, which is a module of type *SR_latch*. The module *m1* instantiates **nand** gates *n1* and *n2*. *Q*, *Qbar*, *S*, and *R* are port signals in instance *m1*. Hierarchical name referencing

assigns a unique name to each identifier. To assign hierarchical names, use the module name for root module and instance names for all module instances below the root module. Example 4-7 shows hierarchical names for all identifiers in the above simulation. Notice that there is a dot (.) for each level of hierarchy from the *root* module to the desired identifier.

Example 4-7 Hierarchical Names

```
stimulus                                stimulus.q
stimulus.qbar                           stimulus.set
stimulus.reset                          stimulus.m1
stimulus.m1.Q                           stimulus.m1.Qbar
stimulus.m1.S                           stimulus.m1.R
stimulus.n1                             stimulus.n2
```

Each identifier in the design is uniquely specified by its hierarchical path name. To display the level of hierarchy, use the special character *%m* in the **$display** task. See Table 3-4, *String Format Specifications*, for details.

4.4 Summary

In this chapter we discussed the following aspects of Verilog

- Module definitions contain various components. Keywords **module** and **endmodule** are mandatory. Other components—*port list, port declarations, variable* and *signal declarations, dataflow statements, behavioral blocks, lower-level module instantiations,* and *tasks* or *functions*—are optional and can be added as needed.
- Ports provide the module with a means to communicate with other modules or its environment. A module can have a port list. Ports in the port list must be declared as **input**, **output**, or **inout**. When instantiating a module, port connection rules are enforced by the Verilog simulator.
- Ports can be connected by *name* or by *ordered list.*
- Each identifier in the design has a unique hierarchical name. Hierarchical names allow us to address any identifier in the design from any other level of hierarchy in the design.

4.5 Exercises

1. What are the basic components of a module? Which components are mandatory?
2. Does a module that does not interact with its environment have any I/O ports? Does it have a port list in the module definition?
3. A 4-bit parallel shift register has I/O pins as shown in the figure below. Write the module definition for this module *shift_reg*. Include the list of ports and port declarations. You do not need to show the internals.



4. Declare a top-level module *stimulus*. Define *REG_IN* (4 bit) and *CLK* (1 bit) as **reg** register variables and *REG_OUT* (4 bit) as **wire**. Instantiate the module *shift_reg* and call it *sr1*. Connect the ports by ordered list.
5. Connect the ports in Step 4 by name.
6. Write the hierarchical names for variables *REG_IN, CLK,* and *REG_OUT.*
7. Write the hierarchical name for the instance *sr1*. Write the hierarchical names for its ports *clock* and *reg_in*.

Gate-Level Modeling 5

In the earlier chapters, we laid the foundations of Verilog design by discussing design methodologies, basic conventions and constructs, modules and port interfaces. In this chapter, we get into modeling actual hardware circuits in Verilog.

We discussed the four levels of abstraction used to describe hardware. In this chapter, we discuss a design at a low level of abstraction—*gate* level. Most digital design is now done at gate level or higher levels of abstraction. At gate level, the circuit is described in terms of gates (e.g., **and**, **nand**). Hardware design at this level is intuitive for a user with a basic knowledge of digital logic design because it is possible to see a one-to-one correspondence between the logic circuit diagram and the Verilog description. Hence, in this book, we chose to start with gate-level modeling and move to higher levels of abstraction in the succeeding chapters.

Actually, the lowest level of abstraction is *switch-* (transistor-) level modeling. However, with designs getting very complex, very few hardware designers work at switch level. Therefore, we will defer switch-level modeling to Chapter 11, *Switch-Level Modeling*, in Part 2 of this book.

Learning Objectives

- Identify logic gate primitives provided in Verilog.
- Understand instantiation of gates, gate symbols and truth tables for *and/or* and *buf/not* type gates.
- Understand how to construct a Verilog description from the logic diagram of the circuit.
- Describe *rise*, *fall*, and *turn-off* delays in the gate-level design.
- Explain *min*, *max*, and *typ* delays in the gate-level design.

5.1 Gate Types

A logic circuit can be designed by use of logic gates. Verilog supports basic logic gates as predefined *primitives*. These primitives are instantiated like modules except that they are predefined in Verilog and do not need a module definition. All logic circuits can be designed by using basic gates. There are two classes of basic gates: *and/or* gates and *buf/not* gates.

5.1.1 And/Or Gates

And/or gates have *one* scalar output and *multiple* scalar inputs. The first terminal in the list of gate terminals is an output and the other terminals are inputs. The output of a gate is evaluated as soon as one of the inputs changes. The **and/or** gates available in Verilog are shown below.

```
and      or      xor
nand     nor     xnor
```

The corresponding logic symbols for these gates are shown in Figure 5-1. We consider gates with two inputs. The output terminal is denoted by *out*. Input terminals are denoted by *i1* and *i2*.



Figure 5-1 Basic Gates

These gates are instantiated to build logic circuits in Verilog. Examples of gate instantiations are shown below. In Example 5-1, for all instances, *OUT* is connected to the output *out*, and *IN1* and *IN2* are connected to the two inputs *i1* and *i2* of the gate primitives. Note that the instance name does not need to be specified for primitives. This lets the designer instantiate hundreds of gates without giving them a name.

More than two inputs can be specified in a gate instantiation. Gates with more than two inputs are instantiated by simply adding more input ports in the gate instantiation (see Example 5-1). Verilog automatically instantiates the appropriate gate.

Example 5-1 Gate Instantiation of And/Or Gates

```
wire OUT, IN1, IN2;

// basic gate instantiations.
and a1(OUT, IN1, IN2);
nand na1(OUT, IN1, IN2);
or or1(OUT, IN1, IN2);
nor nor1(OUT, IN1, IN2);
xor x1(OUT, IN1, IN2);
xnor nx1(OUT, IN1, IN2);

// More than two inputs; 3 input nand gate
nand na1_3inp(OUT, IN1, IN2, IN3);

// gate instantiation without instance name
and (OUT, IN1, IN2); // legal gate instantiation
```

The truth tables for these gates define how outputs for the gates are computed from the inputs. Truth tables are defined assuming two inputs. The truth tables for these gates are shown in Table 5-1. Outputs of gates with more than two inputs are computed by applying the truth table iteratively.

Table 5-1 Truth Tables for And/Or Gates

and	i1=0	i1=1	i1=x	i1=z
i2=0	0	0	0	0
i2=1	0	1	x	x
i2=x	0	x	x	x
i2=z	0	x	x	x

nand	i1=0	i1=1	i1=x	i1=z
i2=0	1	1	1	1
i2=1	1	0	x	x
i2=x	1	x	x	x
i2=z	1	x	x	x

or	i1=0	i1=1	i1=x	i1=z
i2=0	0	1	x	x
i2=1	1	1	1	1
i2=x	x	1	x	x
i2=z	x	1	x	x

nor	i1=0	i1=1	i1=x	i1=z
i2=0	1	0	x	x
i2=1	0	0	0	0
i2=x	x	0	x	x
i2=z	x	0	x	x

xor	i1=0	i1=1	i1=x	i1=z
i2=0	0	1	x	x
i2=1	1	0	x	x
i2=x	x	x	x	x
i2=z	x	x	x	x

xnor	i1=0	i1=1	i1=x	i1=z
i2=0	1	0	x	x
i2=1	0	1	x	x
i2=x	x	x	x	x
i2=z	x	x	x	x

5.1.2 Buf/Not Gates

Buf/not gates have one scalar input and one or more scalar outputs. The last terminal in the port list is connected to the input. Other terminals are connected to the outputs. We will discuss gates that have one input and one output.

Two basic *buf/not* gate primitives are provided in Verilog.

```
buf          not
```

The symbols for these logic gates are shown in Figure 5-2.



Figure 5-2 Buf and Not gates

These gates are instantiated in Verilog as shown Example 5-2. Notice that these gates can have multiple outputs but exactly one input, which is the last terminal in the port list.

Example 5-2 Gate Instantiations of Buf/Not Gates

```
// basic gate instantiations.
buf b1(OUT1, IN);
not n1(OUT1, IN);

// More than two outputs
buf b1_2out(OUT1, OUT2, IN);

// gate instantiation without instance name
not (OUT1, IN); // legal gate instantiation
```

The truth tables for these gates are very simple. Truth tables for gates with one input and one output are shown in Table 5-2.

Table 5-2 Truth Tables for Buf/Not gates

buf	in	out
	0	0
	1	1
	x	x
	z	x

not	in	out
	0	1
	1	0
	x	x
	z	x

Bufif/notif

Gates with an additional control signal on **buf** and **not** gates are also available.

```
bufif1        notif1
bufif0        notif0
```

These gates propagate only if their control signal is asserted. They propagate **z** if their control signal is deasserted. Symbols for *bufif/notif* are shown in Figure 5-3.



Figure 5-3 Gates Bufif and Notif

The truth tables for these gates are shown in Table 5-3.

Table 5-3 Truth Tables for Bufif/Notif Gates

	bufif1 \ ctrl	0	1	x	z
in	0	z	0	L	L
	1	z	1	H	H
	x	z	x	x	x
	z	z	x	x	x

	bufif0 \ ctrl	0	1	x	z
in	0	0	z	L	L
	1	1	z	H	H
	x	x	z	x	x
	z	x	z	x	x

	notif1 \ ctrl	0	1	x	z
in	0	z	1	H	H
	1	z	0	L	L
	x	z	x	x	x
	z	z	x	x	x

	notif0 \ ctrl	0	1	x	z
in	0	1	z	H	H
	1	0	z	L	L
	x	x	z	x	x
	z	x	z	x	x

These gates are used when a signal is to be driven only when the control signal is asserted. Such a situation is applicable when multiple drivers drive the signal. These drivers are designed to drive the signal on mutually exclusive control signals. Example 5-3 shows examples of instantiation of bufif and notif gates.

Example 5-3 Gate Instantiations of Bufif/Notif Gates

```
//Instantiation of bufif gates.
bufif1 b1 (out, in, ctrl);
bufif0 b0 (out, in, ctrl);

//Instantiation of notif gates
notif1 n1 (out, in, ctrl);
notif0 n0 (out, in, ctrl);
```

5.1.3 Examples

Having understood the various types of gates available in Verilog, we will discuss a real example that illustrates design of gate-level digital circuits.

Gate-level multiplexer

We will design a *4-to-1 multiplexer* with 2 select signals. Multiplexers serve a useful purpose in logic design. They can connect two or more sources to a single destination. They can also be used to implement boolean functions. We will assume for this example that signals *s1* and *s0* do not get the value `x` or `z`. The I/O diagram and the truth table for the multiplexer are shown in Figure 5-4. The I/O diagram will be useful in setting up the port list for the multiplexer.



s1	s0	out
0	0	I0
0	1	I1
1	0	I2
1	1	I3

Figure 5-4 4-to-1 Multiplexer

We will implement the logic for the multiplexer using basic logic gates. The logic diagram for the multiplexer is shown in Figure 5-5.



Figure 5-5 Logic Diagram for Multiplexer

The logic diagram has a one-to-one correspondence with the Verilog description. The Verilog description for the multiplexer is shown in Example 5-4. Two intermediate nets, *s0n* and *s1n*, are created; they are complements of input signals *s1* and *s0*. Internal nets *y0, y1, y2, y3* are also required. Note that instance names are not specified for primitive gates, **not**, **and**, and **or**. Instance names are optional for Verilog primitives but are mandatory for instances of user-defined modules.

Example 5-4 Verilog Description of Multiplexer

```
// Module 4-to-1 multiplexer. Port list is taken exactly from
// the I/O diagram.
module mux4_to_1 (out, i0, i1, i2, i3, s1, s0);

// Port declarations from the I/O diagram
output out;
input i0, i1, i2, i3;
input s1, s0;
```

Example 5-4 Verilog Description of Multiplexer (Continued)

```
// Internal wire declarations
wire s1n, s0n;
wire y0, y1, y2, y3;

// Gate instantiations

// Create s1n and s0n signals.
not (s1n, s1);
not (s0n, s0);

// 3-input and gates instantiated
and (y0, i0, s1n, s0n);
and (y1, i1, s1n, s0);
and (y2, i2, s1, s0n);
and (y3, i3, s1, s0);

// 4-input or gate instantiated
or (out, y0, y1, y2, y3);

endmodule
```

This multiplexer can be tested with the stimulus shown in Example 5-5. The stimulus checks that each combination of select signals connects the appropriate input to the output. The signal *OUTPUT* is displayed one time unit after it changes. System task **$monitor** could also be used to display the signals when they change values.

Example 5-5 Stimulus for Multiplexer

```
// Define the stimulus module (no ports)
module stimulus;

// Declare variables to be connected
// to inputs
reg IN0, IN1, IN2, IN3;
reg S1, S0;

// Declare output wire
wire OUTPUT;

// Instantiate the multiplexer
mux4_to_1 mymux(OUTPUT, IN0, IN1, IN2, IN3, S1, S0);
```

Example 5-5 Stimulus for Multiplexer (Continued)

```
// Define the stimulus module (no ports)

// Stimulate the inputs
initial
begin
  // set input lines
  IN0 = 1; IN1 = 0; IN2 = 1; IN3 = 0;
  #1 $display("IN0= %b, IN1= %b, IN2= %b, IN3=
%b\n",IN0,IN1,IN2,IN3);

  // choose IN0
  S1 = 0; S0 = 0;
  #1 $display("S1 = %b, S0 = %b, OUTPUT = %b \n", S1, S0, OUTPUT);

  // choose IN1
  S1 = 0; S0 = 1;
  #1 $display("S1 = %b, S0 = %b, OUTPUT = %b \n", S1, S0, OUTPUT);

  // choose IN2
  S1 = 1; S0 = 0;
  #1 $display("S1 = %b, S0 = %b, OUTPUT = %b \n", S1, S0, OUTPUT);

  // choose IN3
  S1 = 1; S0 = 1;
  #1 $display("S1 = %b, S0 = %b, OUTPUT = %b \n", S1, S0, OUTPUT);
end

endmodule
```

The output of the simulation is shown below. Each combination of the select signals is tested.

```
IN0= 1, IN1= 0, IN2= 1, IN3= 0

S1 = 0, S0 = 0, OUTPUT = 1

S1 = 0, S0 = 1, OUTPUT = 0

S1 = 1, S0 = 0, OUTPUT = 1

S1 = 1, S0 = 1, OUTPUT = 0
```

4-bit full adder

In this example, we design a 4-bit full adder whose port list was defined in Section 4.2.1, *List of Ports*. We use primitive logic gates, and we apply stimulus to the 4-bit full adder to check functionality . For the sake of simplicity, we will implement a ripple carry adder. The basic building block is a 1-bit full adder. The mathematical equations for a 1-bit full adder are shown below.

$$sum = (a \oplus b \oplus cin)$$

$$cout = (a \cdot b) + cin \cdot (a \oplus b)$$

The logic diagram for a 1-bit full adder is shown in Figure 5-6.



Figure 5-6 1-bit Full Adder

This logic diagram for the 1-bit full adder is converted to a Verilog description, shown in Example 5-6.

Example 5-6 Verilog Description for 1-bit Full Adder

```
// Define a 1-bit full adder
module fulladd(sum, c_out, a, b, c_in);

// I/O port declarations
output sum, c_out;
input a, b, c_in;

// Internal nets
wire s1, c1, c2;
```

Example 5-6 Verilog Description for 1-bit Full Adder (Continued)

```
// Instantiate logic gate primitives
xor (s1, a, b);
and (c1, a, b);

xor (sum, s1, c_in);
and (c2, s1, c_in);

or  (c_out, c2, c1);

endmodule
```

A 4-bit ripple carry full adder can be constructed from four 1-bit full adders, as shown in Figure 5-7. Notice that *fa0, fa1, fa2,* and *fa3* are instances of the module *fulladd* (1-bit full adder).



Figure 5-7 4-bit Full Adder

This structure can be translated to Verilog as shown in Example 5-7. Note that the port names used in a 1-bit full adder and a 4-bit full adder are the same but they represent different elements. The element *sum* in a 1-bit adder is a scalar quantity and the element *sum* in the 4-bit full adder is a 4-bit vector quantity. Verilog keeps names local to a module. Names are not visible outside the module unless full-path, hierarchical name referencing is used. Also note that instance names must be specified when defined modules are instantiated, but when instantiating Verilog primitives, the instance names are optional.

Example 5-7 *Verilog Description for 4-bit Full Adder*

```
// Define a 4-bit full adder
module fulladd4(sum, c_out, a, b, c_in);

// I/O port declarations
output [3:0] sum;
output c_out;
input[3:0] a, b;
input c_in;

// Internal nets
wire c1, c2, c3;

// Instantiate four 1-bit full adders.
fulladd fa0(sum[0], c1, a[0], b[0], c_in);
fulladd fa1(sum[1], c2, a[1], b[1], c1);
fulladd fa2(sum[2], c3, a[2], b[2], c2);
fulladd fa3(sum[3], c_out, a[3], b[3], c3);

endmodule
```

Finally, the design must be checked by applying stimulus, as shown in Example 5-8. The module *stimulus* stimulates the 4-bit full adder by applying a few input combinations and monitors the results.

Example 5-8 *Stimulus for 4-bit Full Adder*

```
// Define the stimulus (top level module)
module stimulus;

// Set up variables
reg [3:0] A, B;
reg C_IN;
wire [3:0] SUM;
wire C_OUT;

// Instantiate the 4-bit full adder. call it FA1_4
fulladd4 FA1_4(SUM, C_OUT, A, B, C_IN);


// Setup the monitoring for the signal values
```

Example 5-8 *Stimulus for 4-bit Full Adder (Continued)*

```
initial
begin
  $monitor($time," A= %b, B=%b, C_IN= %b, --- C_OUT= %b, SUM= %b\n",
                              A, B, C_IN, C_OUT, SUM);
end

// Stimulate inputs
initial
begin
  A = 4'd0; B = 4'd0; C_IN = 1'b0;

  #5 A = 4'd3; B = 4'd4;

  #5 A = 4'd2; B = 4'd5;

  #5 A = 4'd9; B = 4'd9;

  #5 A = 4'd10; B = 4'd15;

  #5 A = 4'd10; B = 4'd5; C_IN = 1'b1;
end

endmodule
```

The output of the simulation is shown below.

```
0 A= 0000, B=0000, C_IN= 0, --- C_OUT= 0, SUM= 0000

5 A= 0011, B=0100, C_IN= 0, --- C_OUT= 0, SUM= 0111

10 A= 0010, B=0101, C_IN= 0, --- C_OUT= 0, SUM= 0111

15 A= 1001, B=1001, C_IN= 0, --- C_OUT= 1, SUM= 0010

20 A= 1010, B=1111, C_IN= 0, --- C_OUT= 1, SUM= 1001

25 A= 1010, B=0101, C_IN= 1,, C_OUT= 1, SUM= 0000
```

5.2 Gate Delays

Until now, we described circuits without any delays (i.e., zero delay). In real circuits, logic gates have delays associated with them. Gate delays allow the Verilog user to specify delays through the logic circuits. Pin-to-pin delays can also be specified in Verilog. They are discussed in Chapter 10, *Timing and Delays.*

5.2.1 Rise, Fall, and Turn-off Delays

There are three types of delays from the inputs to the output of a primitive gate.

Rise delay

The rise delay is associated with a gate output transition to a `1` from another value.



Fall delay

The fall delay is associated with a gate output transition to a `0` from another value.



Turn-off delay

The turn-off delay is associated with a gate output transition to the high impedance value (`z`) from another value.

If the value changes to `x`, the minimum of the three delays is considered.

Three types of delay specifications are allowed. If only *one* delay is specified, this value is used for all transitions. If *two* delays are specified, they refer to the rise and fall delay values. The turn-off delay is the minimum of the two delays. If all *three* delays are specified, they refer to rise, fall, and turn-off delay values. If no delays are specified, the default value is zero. Examples of delay specification are shown in Example 5-9.

Example 5-9 Types of Delay Specification

```
// Delay of delay_time for all transitions
and #(delay_time) a1(out, i1, i2);

// Rise and Fall Delay Specification.
and #(rise_val, fall_val) a2(out, i1, i2);

// Rise, Fall, and Turn-off Delay Specification
bufif0 #(rise_val, fall_val, turnoff_val) b1 (out, in, control);
```

Examples of delay specification are shown below.

```
and #(5) a1(out, i1, i2); //Delay of 5 for all transitions
and #(4,6) a2(out, i1, i2); // Rise = 4, Fall = 6
bufif0 #(3,4,5) b1 (out, in, control); //Rise = 3, Fall = 4, Turn-off = 5
```

5.2.2 Min/Typ/Max Values

Verilog provides an additional level of control for each type of delay mentioned above. For each type of delay—rise, fall, and turn-off—three values, *min*, *typ*, and *max*, can be specified. Any one value can be chosen at the start of the simulation. Min/typ/max values are used to model devices whose delays vary within a minimum and maximum range because of the IC fabrication process variations.

Min value

The min value is the minimum delay value that the designer expects the gate to have.

Typ val

The typ value is the typical delay value that the designer expects the gate to have.

Max value

The max value is the maximum delay value that the designer expects the gate to have.

Min, typ, or max values can be chosen at Verilog run time. Method of choosing a min/typ/max value may vary for different simulators or operating systems. (For Verilog-XL™, the values are chosen by specifying options **+maxdelays**, **+typdelay**, and **+mindelays** at run time. If no option is specified, the typical delay value is the default). This allows the designers the flexibility of building three delay values for each transition into their design. The designer can experiment with delay values without modifying the design.

Examples of min, typ, and max value specification for Verilog-XL are shown in Example 5-10.

Example 5-10 Min, Max and Typical Delay Values

```
// One delay
// if +mindelays, delay= 4
// if +typdelays, delay= 5
// if +maxdelays, delay= 6
and #(4:5:6) a1(out, i1, i2);

// Two delays
// if +mindelays, rise= 3, fall= 5, turn-off = min(3,5)
// if +typdelays, rise= 4, fall= 6, turn-off = min(4,6)
// if +maxdelays, rise= 5, fall= 7, turn-off = min(5,7)
and #(3:4:5, 5:6:7) a2(out, i1, i2);

// Three delays
// if +mindelays, rise= 2 fall= 3 turn-off = 4
// if +typdelays, rise= 3 fall= 4 turn-off = 5
// if +maxdelays, rise= 4 fall= 5 turn-off = 6
and #(2:3:4, 3:4:5, 4:5:6) a3(out, i1,i2);
```

Examples of invoking the Verilog-XL simulator with the command-line options are shown below. Assume that the module with delays is declared in the file *test.v*.

```
//invoke simulation with maximum delay
> verilog test.v +maxdelays

//invoke simulation with minimum delay
```

```
> verilog test.v +mindelays

//invoke simulation with typical delay
> verilog test.v +typdelays
```

5.2.3 Delay Example

Let us consider a simple example to illustrate the use of gate delays to model timing in the logic circuits. A simple module called *D* implements the following logic equations:

$$out = (a \cdot b) + c$$

The gate-level implementation is shown in *Module D* (Figure 5-8). The module contains two gates with delays of 5 and 4 time units.



Figure 5-8 Module D

The module *D* is defined in Verilog as shown in Example 5-11.

Example 5-11 Verilog Definition for Module D with Delay

```
// Define a simple combination module called D
module D (out, a, b, c);

// I/O port declarations
output out;
input a,b,c;

// Internal nets
wire e;

// Instantiate primitive gates to build the circuit
and #(5) a1(e, a, b); //Delay of 5 on gate a1
```

Example 5-11 Verilog Definition for Module D with Delay (Continued)

```
or  #(4) o1(out, e,c); //Delay of 4 on gate o1

endmodule
```

This module is tested by the stimulus file shown in Example 5-12.

Example 5-12 Stimulus for Module D with Delay

```
// Stimulus (top-level module)
module stimulus;

// Declare variables
reg A, B, C;
wire OUT;

// Instantiate the module D
D d1( OUT, A, B, C);

// Stimulate the inputs. Finish the simulation at 40 time units.
initial
begin
  A= 1'b0; B= 1'b0; C= 1'b0;

  #10 A= 1'b1; B= 1'b1; C= 1'b1;

  #10 A= 1'b1; B= 1'b0; C= 1'b0;

  #20 $finish;
end

endmodule
```

The waveforms from the simulation are shown in Figure 5-9 to illustrate the effect of specifying delays on gates. The waveforms are not drawn to scale. However, simulation time at each transition is specified below the transition.



Figure 5-9 Waveforms for Delay Simulation

1. The outputs *E* and *OUT* are initially unknown.
2. At time 10, after *A*, *B*, and *C* all transition to `1`, *OUT* transitions to `1` after a delay of 4 time units and E changes value to `1` after 5 time units.
3. At time 20, *B* and *C* transition to `0`. *E* changes value to `0` after 5 time units, and *OUT* transitions to `0`, 4 time units after *E* changes.

It is a useful exercise to understand how the timing for each transition in the above waveform corresponds to the gate delays shown in *Module D*.

5.3 Summary

In this chapter we discussed how to model gate-level logic in Verilog. We discussed different aspects of gate-level design.

- Basic types of gates are **`and`**, **`or`**, **`xor`**, **`buf`**, and **`not`**. Each gate has a logic symbol, truth table, and a corresponding Verilog primitive. Primitives are instantiated like modules except that they are predefined in Verilog. Output of a gate is evaluated as soon as one of its inputs changes.

- For gate-level design, start with the logic diagram, write the Verilog description for the logic by using gate primitives, provide stimulus, and look at the output. Two design examples, a 4-to-1 multiplexer and a 4-bit full adder, were discussed. Each step of the design process was explained.
- Three types of delays are associated with gates, *rise, fall,* and *turn-off.* Verilog allows specification of one, two, or three delays for each gate. Values of rise, fall, and turn-off delays are computed by Verilog, based on the *one, two,* or *three* delays specified.
- For each type of delay, a *minimum, typical,* and *maximum* value can be specified. The user can choose which value to apply at simulation time. This provides the flexibility to experiment with three delay values without changing the Verilog code.
- The effect of propagation delay on waveforms was explained by the simple, two-gate logic example. For each gate with a delay of t, the output changes t time units after any of the inputs change.

5.4 Exercises

1. Create your own 2-input Verilog gates called *my-or, my-and* and *my-not* from 2-input **nand** gates. Check the functionality of these gates with a stimulus module.
2. A 2-input **xor** gate can be built from *my_and, my_or* and *my_not* gates. Construct an **xor** module in Verilog that realizes the logic function, $z = xy' + x'y$. Inputs are x and y, and z is the output. Write a stimulus module that exercises all four combinations of x and y inputs.
3. The 1-bit full adder described in the chapter can be expressed in a sum of products form.

   $sum = a.b.c\_in + a'.b.c\_in' + a'.b'.c\_in + a.b'.c\_in'$

   $c\_out = a.b + b.c\_in + a.c\_in$

   Assuming *a, b, c_in* are the inputs and *sum* and *c_out* are the outputs, design a logic circuit to implement the 1-bit full adder, using only **and**, **not**, and **or** gates. Write the Verilog description for the circuit. You may use up to 4-input Verilog primitive **and** and **or** gates. Write the stimulus for the full adder and check the functionality for all input combinations.

4. The logic diagram for an *RS latch* with delay is shown below.



Write the Verilog description for the RS latch. Include delays of 1 unit when instantiating the **nor** gates. Write the stimulus module for the RS latch, using the following table, and verify the outputs.

set	reset	q_{n+1}
0	0	q_n
0	1	0
1	0	1
1	1	?

5. Design a 2-to-1 multiplexer using **`bufif0`** and **`bufif1`** gates as shown below.



The delay specification for gates *b1* and *b2* are as follows.

	Min	Typ	Max
Rise	1	2	3
Fall	3	4	5
Turnoff	5	6	7

Apply stimulus and test the output values.

Dataflow Modeling

For small circuits, the gate-level modeling approach works very well because the number of gates is limited and the designer can instantiate and connect every gate individually. Also, gate-level modeling is very intuitive to a designer with a basic knowledge of digital logic design. However, in complex designs the number of gates is very large. Thus, designers can design more effectively if they concentrate on implementing the function at a level of abstraction higher than gate level. Dataflow modeling provides a powerful way to implement a design. Verilog allows a circuit to be designed in terms of the data flow between registers and how a design processes data rather than instantiation of individual gates. Later in this chapter, the benefits of dataflow modeling will become more apparent.

With gate densities on chips increasing rapidly, dataflow modeling has assumed great importance. No longer can companies devote engineering resources to handcrafting entire designs with gates. Currently, automated tools are used to create a gate-level circuit from a dataflow design description. This process is called *logic synthesis*. Dataflow modeling has become a popular design approach as logic synthesis tools have become sophisticated. This approach allows the designer to concentrate on optimizing the circuit in terms of data flow. For maximum flexibility in the design process, designers typically use a Verilog description style that combines the concepts of gate-level, data flow, and behavioral design. In the digital design community , the term RTL (Register Transfer Level) design is commonly used for a combination of dataflow modeling and behavioral modeling.

Learning Objectives

- Describe the continuous assignment (**`assign`**) statement, restrictions on the **`assign`** statement, and the implicit continuous assignment statement.
- Explain assignment delay, implicit assignment delay, and net declaration delay for continuous assignment statements.
- Define expressions, operators, and operands.

- List operator types for all possible operations—arithmetic, logical, relational, equality, bitwise, reduction, shift, concatenation, and conditional.
- Use dataflow constructs to model practical digital circuits in Verilog.

6.1 Continuous Assignments

A *continuous assignment* is the most basic statement in dataflow modeling, used to drive a value onto a net. A continuous assignment replaces gates in the description of the circuit and describes the circuit at a higher level of abstraction. A continuous assignment statement starts with the keyword **assign**. The syntax of an **assign** statement is as follows.

```
//Syntax of assign statement in the simplest form
<continuous_assign>
       ::= assign <drive_strength>?<delay>? <list_of_assignments>;
```

Notice that drive strength is optional and can be specified in terms of strength levels discussed in Section 3.2.1, *Value Set*. We will not discuss drive strength specification in this chapter. The default value for drive strength is **strong1** and **strong0**. The delay value is also optional and can be used to specify delay on the **assign** statement. This is like specifying delays for gates. Delay specification is discussed in this chapter. Continuous assignments have the following characteristics.

1. The left hand side of an assignment must always be a scalar or vector net or a concatenation of scalar and vector nets. It cannot be a scalar or vector register. Concatenations are discussed in Section 6.4.8, *Concatenation Operator*.
2. Continuous assignments are always active. The assignment expression is evaluated as soon as one of the right-hand-side operands changes and the value is assigned to the left-hand-side net.
3. The operands on the right-hand side can be registers or nets or function calls. Registers or nets can be scalars or vectors.
4. Delay values can be specified for assignments in terms of time units. Delay values are used to control the time when a net is assigned the evaluated value. This feature is similar to specifying delays for gates. It is very useful in modeling timing behavior in real circuits.

Examples of continuous assignments are shown below. Operators such as &, ^, |, {, } and + used in the examples are explained in Section 6.4, *Operator Types*. At this point, concentrate on how the **assign** statements are specified.

Example 6-1 *Examples of Continuous Assignment*

```
// Continuous assign. out is a net. i1 and i2 are nets.
assign out = i1 & i2;

// Continuous assign for vector nets. addr is a 16-bit vector net
// addr1 and addr2 are 16-bit vector registers.
assign addr[15:0] = addr1_bits[15:0] ^ addr2_bits[15:0];

// Concatenation. Left-hand side is a concatenation of a scalar
// net and a vector net.
assign {c_out, sum[3:0]} = a[3:0] + b[3:0] + c_in;
```

We now discuss a shorthand method of placing a continuous assignment on a net.

6.1.1 Implicit Continuous Assignment

Instead of declaring a net and then writing a continuous assignment on the net, Verilog provides a shortcut by which a continuous assignment can be placed on a net when it is declared. There can be only one implicit declaration assignment per net because a net is declared only once.

In the example below, an implicit continuous assignment is contrasted with a regular continuous assignment.

```
//Regular continuous assignment
wire out;
assign out = in1 & in2;

//Same effect is achieved by an implicit continuous assignment
wire out = in1 & in2;
```

6.2 Delays

Delay values control the time between the change in a right-hand-side operand and when the new value is assigned to the left-hand side. Three ways of specifying delays in continuous assignment statements are *regular assignment delay, implicit continuous assignment delay,* and *net declaration delay.*

6.2.1 Regular Assignment Delay

The first method is to assign a delay value in a continuous assignment statement. The delay value is specified after the keyword **assign**. Any change in values of *in1* or *in2* will result in a delay of 10 time units before recomputation of the expression *in1* & *in2*, and the result will be assigned to *out*. If *in1* or *in2* changes value again before 10 time units when the result propagates to *out*, the values of *in1* and *in2* at the time of recomputation are considered. This property is called *inertial delay*. An input pulse that is shorter than the delay of the assignment statement does not propagate to the output.

```
assign #10 out = in1 & in2; // Delay in a continuous assign
```

The waveform in Figure 6-1 is generated by simulating the above **assign** statement. It shows the delay on signal *out*. Note the following changes.

1. When signals *in1* and *in2* go high at time 20, *out* goes to a high 10 time units later (time = 30).
2. When *in1* goes low at 60, *out* changes to low at 70.
3. However, *in1* changes to high at 80, but it goes down to low before 10 time units have elapsed.
4. Hence, at the time of recomputation, 10 units after time 80, *in1* is 0. Thus, *out* gets the value 0. A pulse of width less than the specified assignment delay is not propagated to the output.



Figure 6-1 Delays

Inertial delays also apply to gate delays, discussed in Chapter 5, *Gate-Level Modeling.*

6.2.2 Implicit Continuous Assignment Delay

An equivalent method is to use an implicit continuous assignment to specify both a delay and an assignment on the net.

```
//implicit continuous assignment delay
wire #10 out = in1 & in2;

//same as
wire out;
assign #10 out = in1 & in2;
```

The declaration above has the same effect as defining a *wire out* and declaring a continuous assignment on *out*.

6.2.3 Net Declaration Delay

A delay can be specified on a net when it is declared without putting a continuous assignment on the net. If a delay is specified on a net *out*, then any value change applied to the net *out* is delayed accordingly. Net declaration delays can also be used in gate-level modeling.

```
//Net Delays
wire # 10 out;
assign out = in1 & in2;

//The above statement has the same effect as the following.
wire out;
assign #10 out = in1 & in2;
```

Having discussed continuous assignments and delays, let us take a closer look at expressions, operators, and operands that are used inside continuous assignments.

6.3 Expressions, Operators, and Operands

Dataflow modeling describes the design in terms of expressions instead of primitive gates. *Expressions, operators,* and *operands* form the basis of dataflow modeling.

6.3.1 Expressions

Expressions are constructs that combine operators and operands to produce a result.

```
// Examples of expressions. Combines operands and operators
a ^ b
addr1[20:17] + addr2[20:17]
in1 | in2
```

6.3.2 Operands

Operands can be any one of the data types defined in Section 3.2, *Data Types*. Some constructs will take only certain types of operands. Operands can be *constants, integers, real numbers, nets, registers, times, bit-select* (one bit of vector net or a vector register), *part-select* (selected bits of the vector net or register vector), *memories* or *function calls* (functions are discussed later).

```
integer count, final_count;
final_count = count + 1;//count is an integer operand

real a, b, c;
c = a - b; //a and b are real operands

reg [15:0] reg1, reg2;
reg [3:0] reg_out;
reg_out = reg1[3:0] ^ reg2[3:0];//reg1[3:0] and reg2[3:0] are
                                //part-select register operands

reg ret_value;
ret_value = calculate_parity(A, B);//calculate_parity is a
                                //function type operand
```

6.3.3 Operators

Operators act on the operands to produce desired results. Verilog provides various types of operators. Operator types are discussed in detail in Section 6.4, *Operator Types*.

```
d1 && d2 // && is an operator on operands d1 and d2
!a[0] // ! is an operator on operand a[0]
B >> 1 // >> is an operator on operands B and 1
```

6.4 Operator Types

Verilog provides many different operator types. Operators can be *arithmetic, logical, relational, equality, bitwise, reduction, shift, concatenation,* or *conditional.* Some of these operators are similar to the operators used in the C programming language. Each operator type is denoted by a symbol. Table 6-1 shows the complete listing of operator symbols classified by category.

Table 6-1 Operator Types and Symbols

Operator Type	Operator Symbol	Operation Performed	Number of Operands
Arithmetic	*	multiply	two
	/	divide	two
	+	add	two
	-	subtract	two
	%	modulus	two
Logical	!	logical negation	one
	&&	logical and	two
	||	logical or	two
Relational	>	greater than	two
	<	less than	two
	>=	greater than or equal	two
	<=	less than or equal	two
Equality	==	equality	two
	!=	inequality	two
	===	case equality	two
	!==	case inequality	two
Bitwise	~	bitwise negation	one
	&	bitwise and	two
	|	bitwise or	two
	^	bitwise xor	two
	^~ or ~^	bitwise xnor	two
Reduction	&	reduction and	one
	~&	reduction nand	one
	|	reduction or	one
	~|	reduction nor	one
	^	reduction xor	one
	^~ or ~^	reduction xnor	one

Table 6-1 Operator Types and Symbols (Continued)

Operator Type	Operator Symbol	Operation Performed	Number of Operands
Shift	>> <<	Right shift Left shift	two two
Concatenation	{ }	Concatenation	any number
Replication	{ { } }	Replication	any number
Conditional	?:	Conditional	three

Let us now discuss each operator type in detail.

6.4.1 Arithmetic Operators

There are two types of arithmetic operators: binary and unary.

Binary operators

Binary arithmetic operators are *multiply (*), divide (/), add (+), subtract (-)* and *modulus (%)*. Binary operators take two operands.

```
A = 4'b0011; B = 4'b0100; // A and B are register vectors
D = 6; E = 4; // D and E are integers

A * B // Multiply A and B. Evaluates to 4'b1100
D / E // Divide D by E. Evaluates to 1. Truncates any fractional part.
A + B // Add A and B. Evaluates to 4'b0111
B - A // Subtract A from B. Evaluates to 4'b0001
```

If any operand bit has a value **x**, then the result of the entire expression is **x**. This seems intuitive because if an operand value is not known precisely, the result should be an unknown.

```
in1 = 4'b101x;
in2 = 4'b1010;
sum = in1 + in2; // sum will be evaluated to the value 4'bx
```

Modulus operators produce the *remainder* from the division of two numbers. They operate similarly to the modulus operator in the C programming language.

```
13 % 3 // Evaluates to 1
16 % 4 // Evaluates to 0
-7 % 2 // Evaluates to -1, takes sign of the first operand
7 % -2 // Evaluates to +1, takes sign of the first operand
```

Unary operators

The operators + and - can also work as *unary* operators. They are used to specify the positive or negative sign of the operand. Unary + or - operators have higher precedence than the binary + or - operators.

```
-4 // Negative 4
+5 // Positive 5
```

Negative numbers are represented as 2's complement internally in Verilog. It is advisable to use negative numbers only of the type integer or real in expressions. Designers should avoid negative numbers of the type *<sss> '<base> <nnn>* in expressions because they are converted to unsigned 2's complement numbers and hence yield unexpected results.

```
//Advisable to use integer or real numbers
-10 / 5// Evaluates to -2

//Do not use numbers of type <sss> '<base> <nnn>
-'d10 / 5// Is equivalent (2's complement of 10)/5 = (2^32 - 10)/5
// where 32 is the default machine word width.
// This evaluates to an incorrect and unexpected result
```

6.4.2 Logical Operators

Logical operators are *logical-and* (**&&**), *logical-or* (**||**) and *logical-not* (**!**). Operators **&&** and **||** are binary operators. Operator **!** is a unary operator. Logical operators follow these conditions:

1. Logical operators always evaluate to a 1-bit value, **0** (false), **1** (true), or **x** (ambiguous).
2. If an operand is not equal to zero, it is equivalent to a logical **1** (true condition). If it is equal to zero, it is equivalent to a logical **0** (false condition). If any operand bit is **x** or **z**, it is equivalent to **x** (ambiguous condition) and is normally treated by simulators as a false condition.
3. Logical operators take variables or expressions as operands.

Use of parentheses to group logical operations is highly recommended to improve readability. Also, the user does not have to remember the precedence of operators.

```
// Logical operations
A = 3; B = 0;
A && B // Evaluates to 0. Equivalent to (logical-1 && logical-0)
A || B // Evaluates to 1. Equivalent to (logical-1 || logical-0)
!A// Evaluates to 0. Equivalent to not(logical-1)
!B// Evaluates to 1. Equivalent to not(logical-0)

// Unknowns
A = 2'b0x; B = 2'b10;
A && B // Evaluates to x. Equivalent to (x && logical 1)

// Expressions
(a == 2) && (b == 3) // Evaluates to 1 if both a == 2 and b == 3
are true.
 // Evaluates to 0 if either is false.
```

6.4.3 Relational Operators

Relational operators are *greater-than* (**>**), *less-than* (**<**), *greater-than-or-equal-to* (**>=**), and *less-than-or-equal-to* (**<=**). If relational operators are used in an expression, the expression returns a logical value of **1** if the expression is true and **0** if the expression is false. If there are any unknown or **z** bits in the operands, the expression takes a value **x**. These operators function exactly as the corresponding operators in the C programming language.

```
// A = 4, B = 3
// X = 4'b1010, Y = 4'b1101, Z = 4'b1xxx

A <= B // Evaluates to a logical 0
A > B // Evaluates to a logical 1
```

```
Y >= X // Evaluates to a logical 1
Y < Z // Evaluates to an x
```

6.4.4 Equality Operators

Equality operators are *logical equality* (==), *logical inequality* (!=), *case equality* (===), and *case inequality* (!==). When used in an expression, equality operators return logical value 1 if true, 0 if false. These operators compare the two operands bit by bit, with zero filling if the operands are of unequal length. Table 6-2 lists the operators.

Table 6-2 Equality Operators

Expression	Description	Possible Logical Value
a == b	a equal to b, result unknown if x or z in a or b	0, 1, x
a != b	a not equal to b, result unknown if x or z in a or b	0, 1, x
a === b	a equal to b, including x and z	0, 1
a !== b	a not equal to b, including x and z	0, 1

It is important to note the difference between the logical equality operators (==, !=) and case equality operators (===, !==). The logical equality operators (==, !=) will yield an x if either operand has x or z in its bits. However, the case equality operators (===, !==) compare both operands bit by bit and compare all bits, including x and z. The result is 1 if the operands match exactly, including x and z bits. The result is 0 if the operands do not match exactly. Case equality operators never result in an x.

```
// A = 4, B = 3
// X = 4'b1010, Y = 4'b1101
// Z = 4'b1xxz, M = 4'b1xxz, N = 4'b1xxx

A == B // Results in logical 0
X != Y // Results in logical 1
X == Z // Results in x
Z === M //Results in logical 1 (all bits match, including x and z)
Z === N //Results in logical 0 (least significant bit does not match)
M !== N // Results in logical 1
```

6.4.5 Bitwise Operators

Bitwise operators are *negation* (~), *and*(&), *or* (|), *xor* (^), *xnor* (^~, ~^). Bitwise operators perform a bit-by-bit operation on two operands. They take each bit in one operand and perform the operation with the corresponding bit in the other operand. If one operand is shorter than the other, it will be bit extended with zeros to match the length of the longer operand. Logic tables for the bit-by-bit computation are shown in Table 6-3. A **z** is treated as an **x** in a bitwise operation. The exception is the unary negation operator (~), which takes only one operand and operates on the bits of the single operand.

Table 6-3 Truth Tables for Bitwise Operators

bitwise and	0	1	x
0	0	0	0
1	0	1	x
x	0	x	x

bitwise or	0	1	x
0	0	1	x
1	1	1	1
x	x	1	x

bitwise xor	0	1	x
0	0	1	x
1	1	0	x
x	x	x	x

bitwise xnor	0	1	x
0	1	0	x
1	0	1	x
x	x	x	x

bitwise negation	Result
0	1
1	0
x	x

Examples of bitwise operators are shown below.

```
// X = 4'b1010, Y = 4'b1101
// Z = 4'b10x1

~X     // Negation. Result is 4'b0101
X & Y  // Bitwise and. Result is 4'b1000
X | Y  // Bitwise or. Result is 4'b1111
X ^ Y  // Bitwise xor. Result is 4'b0111
```

```
X ^~ Y // Bitwise xnor. Result is 4'b1000
X & Z  // Result is 4'b10x0
```

It is important to distinguish bitwise operators **~**, **&**, and **|** from logical operators **!**, **&&**, **||**. Logical operators always yield a logical value **0**, **1**, **x**, whereas bitwise operators yield a bit-by-bit value. Logical operators perform a logical operation, not a bit-by-bit operation.

```
// X = 4'b1010, Y = 4'b0000

X | Y // bitwise operation. Result is 4'b1010
X || Y // logical operation. Equivalent to 1 || 0. Result is 1.
```

6.4.6 Reduction Operators

Reduction operators are *and* (**&**), *nand* (**~&**), *or* (**|**), *nor* (**~|**), *xor* (**^**), and *xnor* (**~^**, **^~**). Reduction operators take only one operand. Reduction operators perform a bitwise operation on a single vector operand and yield a 1-bit result. The logic tables for the operators are the same as shown in Section 6.4.5, *Bitwise Operators*. The difference is that bitwise operations are on bits from two different operands, whereas reduction operations are on the bits of the same operand. Reduction operators work bit by bit from right to left. *Reduction nand, reduction nor,* and *reduction xnor* are computed by inverting the result of the *reduction and, reduction or,* and *reduction xor,* respectively.

```
// X = 4'b1010

&X //Equivalent to 1 & 0 & 1 & 0. Results in 1'b0
|X//Equivalent to 1 | 0 | 1 | 0. Results in 1'b1
^X//Equivalent to 1 ^ 0 ^ 1 ^ 0. Results in 1'b0
//A reduction xor or xnor can be used for even or odd parity
//generation of a vector.
```

The use of a similar set of symbols for logical (**!**, **&&**, **||**), bitwise (**~**, **&**, **|**, **^**), and reduction operators (**&**, **|**, **^**) is somewhat confusing initially. The difference lies in the number of operands each operator takes and also the value of result computed.

6.4.7 Shift Operators

Shift operators are *right shift* (>>) and *left shift* (<<). These operators shift a vector operand to the right or the left by a specified number of bits. The operands are the vector and the number of bits to shift. When the bits are shifted, the vacant bit positions are filled with zeros. Shift operations do not wrap around.

```
// X = 4'b1100

Y = X >> 1; //Y is 4'b0110. Shift right 1 bit. 0 filled in MSB position.
Y = X << 1; //Y is 4'b1000. Shift left 1 bit. 0 filled in LSB position.
Y = X << 2; //Y is 4'b0000. Shift left 2 bits.
```

Shift operators are useful because they allow the designer to model shift operations, *shift-and-add* algorithms for multiplication, and other useful operations.

6.4.8 Concatenation Operator

The *concatenation* operator ({, }) provides a mechanism to append multiple operands. The operands must be sized. Unsized operands are not allowed because the size of each operand must be known for computation of the size of the result.

Concatenations are expressed as operands within braces, with commas separating the operands. Operands can be scalar nets or registers, vector nets or registers, bit-select, part-select, or sized constants.

```
// A = 1'b1, B = 2'b00, C = 2'b10, D = 3'b110

Y = {B , C} // Result Y is 4'b0010
Y = {A , B , C , D , 3'b001} // Result Y is 11'b10010110001
Y = {A , B[0], C[1]} // Result Y is 3'b101
```

6.4.9 Replication Operator

Repetitive concatenation of the same number can be expressed by using a replication constant. A replication constant specifies how many times to replicate the number inside the brackets ({ }).

```
reg A;
reg [1:0] B, C;
reg [2:0] D;
A = 1'b1; B = 2'b00; C = 2'b10; D = 3'b110;

Y = { 4{A} } // Result Y is 4'b1111
Y = { 4{A} , 2{B} } // Result Y is 8'b11110000
Y = { 4{A} , 2{B} , C } // Result Y is 8'b1111000010
```

6.4.10 Conditional Operator

The *conditional* operator(`?:`) takes three operands.

Usage: condition_expr ? true_expr : false_expr ;

The condition expression (*condition_expr*) is first evaluated. If the result is true (logical `1`), then the *true_expr* is evaluated. If the result is false (logical `0`), then the *false_expr* is evaluated. If the result is `x` (ambiguous), then both *true_expr* and *false_expr* are evaluated and their results are compared, bit by bit, to return for each bit position an `x` if the bits are different and the value of the bits if they are the same.

The action of a conditional operator is similar to a multiplexer. Alternately, it can be compared to the *if-else* expression.



Conditional operators are frequently used in dataflow modeling to model conditional assignments. The conditional expression acts as a switching control.

```
//model functionality of a tristate buffer
assign addr_bus = drive_enable ? addr_out : 36'bz;

//model functionality of a 2-to-1 mux
assign out = control ? in1 : in0;
```

Conditional operations can be nested. Each *true_expr* or *false_expr* can itself be a conditional operation. In the example that follows, convince yourself that (*A*==*3*) and control are the two select signals of 4-to-1 multiplexer with *n, m, y, x* as the inputs and *out* as the output signal.

```
assign out = (A == 3) ? ( control ? x : y ): ( control ? m : n) ;
```

6.4.11 Operator Precedence

Having discussed the operators, it is now important to discuss operator precedence. If no parentheses are used to separate parts of expressions, Verilog enforces the following precedence. Operators listed in Table 6-4 are in order from highest precedence to lowest precedence. It is recommended that parentheses be used to separate expressions except in case of unary operators or when there is no ambiguity.

Table 6-4 Operator Precedence

Operators	Operator Symbols	Precedence
Unary	+ - ! ~	Highest precedence
Multiply, Divide, Modulus	* / %	
Add, Subtract	+ -	
Shift	<< >>	
Relational	< <= > >=	
Equality	== != === !==	

Table 6-4 Operator Precedence

Operators	Operator Symbols	Precedence
Reduction	&, ~& ^ ^~ |, ~|	
Logical	&& ||	
Conditional	?:	Lowest precedence

6.5 Examples

A design can be represented in terms of gates, data flow, or a behavioral description. In this section we consider the 4-to-1 multiplexer and 4-bit full adder described in Section 5.1.3, *Examples*. Previously, these designs were directly translated from the logic diagram into a gate-level Verilog description. Here, we describe the same designs in terms of data flow. We also discuss two additional examples: a *4-bit full adder using carry lookahead* and a *4-bit counter using negative edge-triggered D-flipflops*.

6.5.1 4-to-1 Multiplexer

Gate-level modeling of a 4-to-1 multiplexer is discussed in Section 5.1.3, *Examples*. The logic diagram for the multiplexer is given in Figure 5-5 on page 69 and the gate-level Verilog description is shown in Example 5-4 on page 69. We describe the multiplexer, using dataflow statements. Compare it with the gate-level description. We show two methods to model the multiplexer by using dataflow statements.

Method 1: logic equation

We can use assignment statements instead of gates to model the logic equations of the multiplexer (see Example 6-2). Notice that everything is same as the gate-level Verilog description except that computation of *out* is done by specifying one logic equation by using operators instead of individual gate instantiations. I/O ports

remain the same. This is important so that the interface with the environment does not change. Only the internals of the module change. Notice how concise the description is compared to the gate-level description.

Example 6-2 *4-to-1 Multiplexer, Using Logic Equations*

```
// Module 4-to-1 multiplexer using data flow. logic equation
// Compare to gate-level model
module mux4_to_1 (out, i0, i1, i2, i3, s1, s0);

// Port declarations from the I/O diagram
output out;
input i0, i1, i2, i3;
input s1, s0;

//Logic equation for out
assign out =   (~s1 & ~s0 & i0)|
               (~s1 & s0 & i1) |
               (s1 & ~s0 & i2) |
               (s1 & s0 & i3) ;

endmodule
```

Method 2: conditional operator

There is a more concise way to specify the 4-to-1 multiplexers. In Section 6.4.10, *Conditional Operator*, we described how a conditional statement corresponds to a multiplexer operation. We will use this operator to write a 4-to-1 multiplexer. Convince yourself that this description (Example 6-3) correctly models a multiplexer.

Example 6-3 *4-to-1 Multiplexer, Using Conditional Operators*

```
// Module 4-to-1 multiplexer using data flow. Conditional operator.
// Compare to gate-level model
module multiplexer4_to_1 (out, i0, i1, i2, i3, s1, s0);

// Port declarations from the I/O diagram
output out;
input i0, i1, i2, i3;
input s1, s0;

// Use nested conditional operator
```

Example 6-3 4-to-1 Multiplexer, Using Conditional Operators

```
assign out = s1 ? ( s0 ? i3 : i2) : (s0 ? i1 : i0) ;

endmodule
```

In the simulation of the multiplexer, the gate-level module in Example 5-4 on page 69 can be substituted with the dataflow multiplexer modules described above. The stimulus module will not change. The simulation results will be identical. By encapsulating functionality inside a module, we can replace the gate-level module with a dataflow module without affecting the other modules in the simulation. This is a very powerful feature of Verilog.

6.5.2 4-bit Full Adder

The 4-bit full adder in Section 5.1.3, *Examples*, was designed by using gates; the logic diagram is shown in Figure 5-7 on page 73 and Figure 5-6 on page 72. In this section, we write the dataflow description for the 4-bit adder. Compare it with the gate-level description in Figure 5-7. In gates, we had to first describe a 1-bit full adder. Then we built a 4-bit full ripple carry adder. We again illustrate two methods to describe a 4-bit full adder by means of dataflow statements.

Method 1: dataflow operators

A concise description of the adder (Example 6-4) is defined with the + and { } operators.

Example 6-4 4-bit Full Adder, Using Dataflow Operators

```
// Define a 4-bit full adder by using dataflow statements.
module fulladd4(sum, c_out, a, b, c_in);

// I/O port declarations
output [3:0] sum;
output c_out;
input[3:0] a, b;
input c_in;

// Specify the function of a full adder
assign {c_out, sum} = a + b + c_in;

endmodule
```

If we substitute the gate-level 4-bit full adder with the dataflow 4-bit full adder, the rest of the modules will not change. The simulation results will be identical.

Method 2: full adder with carry lookahead

In ripple carry adders, the carry must propagate through the gate levels before the sum is available at the output terminals. An *n*-bit ripple carry adder will have *2n* gate levels. The propagation time can be a limiting factor on the speed of the circuit. One of the most popular methods to reduce delay is to use a *carry lookahead* mechanism. Logic equations for implementing the carry lookahead mechanism can be found in any logic design book. The propagation delay is reduced to *four gate levels*, irrespective of the number of bits in the adder. The Verilog description for a carry lookahead adder is shown in Example 6-5. This module can be substituted in place of the full adder modules described before without changing any other component of the simulation. The simulation results will be unchanged.

Example 6-5 *4-bit Full Adder With Carry Lookahead*

```
module fulladd4(sum, c_out, a, b, c_in);
// Inputs and outputs
output [3:0] sum;
output c_out;
input [3:0] a,b;
input c_in;

// Internal wires
wire p0,g0, p1,g1, p2,g2, p3,g3;
wire c4, c3, c2, c1;

// compute the p for each stage
assign p0 = a[0] ^ b[0],
       p1 = a[1] ^ b[1],
       p2 = a[2] ^ b[2],
       p3 = a[3] ^ b[3];

// compute the g for each stage
assign g0 = a[0] & b[0],
       g1 = a[1] & b[1],
       g2 = a[2] & b[2],
       g3 = a[3] & b[3];

// compute the carry for each stage
// Note that c_in is equivalent c0 in the arithmetic equation for
```

Example 6-5 4-bit Full Adder With Carry Lookahead (Continued)

```
// carry lookahead computation
assign c1 = g0 | (p0 & c_in),
       c2 = g1 | (p1 & g0) | (p1 & p0 & c_in),
       c3 = g2 | (p2 & g1) | (p2 & p1 & g0) | (p2 & p1 & p0 & c_in),
       c4 = g3 | (p3 & g2) | (p3 & p2 & g1) | (p3 & p2 & p1 & g0) |
                                  (p3 & p2 & p1 & p0 & c_in);
// Compute Sum
assign sum[0] = p0 ^ c_in,
       sum[1] = p1 ^ c1,
       sum[2] = p2 ^ c2,
       sum[3] = p3 ^ c3;

// Assign carry output
assign c_out = c4;

endmodule
```

6.5.3 Ripple Counter

We now discuss an additional example that was not discussed in the gate-level modeling chapter. We design a 4-bit ripple counter by using negative edge-triggered flip-flops. This example was discussed at a very abstract level in Chapter 2, *Hierarchical Modeling Concepts*. We design it using Verilog dataflow statements and test it with a stimulus module. The diagrams for the 4-bit ripple carry counter modules are shown below.

Figure 6-2 shows the counter being built with *four T-flipflops.*



Figure 6-2 4-bit Ripple Carry Counter

Figure 6-3 shows that the *T-flipflop* is built with one *D-flipflop* and an inverter gate.



Figure 6-3 T-flipflop

Finally, Figure 6-4 shows the *D-flipflop* constructed from basic logic gates.



Figure 6-4 Negative Edge-Triggered D-flipflop with clear

Given the above diagrams, we write the corresponding Verilog, using dataflow statements in a top-down fashion. First we design the module *counter*. The code is shown in Figure 6-6. The code contains instantiation of four *T_FF* modules.

Example 6-6 Verilog Code for Ripple Counter

```
// Ripple counter
module counter(Q , clock, clear);

// I/O ports
output [3:0] Q;
input clock, clear;

// Instantiate the T flipflops
T_FF tff0(Q[0], clock, clear);
T_FF tff1(Q[1], Q[0], clear);
T_FF tff2(Q[2], Q[1], clear);
T_FF tff3(Q[3], Q[2], clear);

endmodule
```

Next, we write the Verilog description for *T_FF* (Example 6-7). Notice that instead of the **not** gate, a dataflow operator ~ negates the signal *q*, which is fed back.

Example 6-7 *Verilog Code for T-flipflop*

```
// Edge-triggered T-flipflop. Toggles every clock
// cycle.
module T_FF(q, clk, clear);

// I/O ports
output q;
input clk, clear;

// Instantiate the edge-triggered DFF
// Complement of output q is fed back.
// Notice qbar not needed. Unconnected port.
edge_dff ff1(q, ,~q, clk, clear);

endmodule
```

Finally, we define the lowest level module *D_FF* (*edge_dff*), using dataflow statements (Example 6-8). The dataflow statements correspond to the logic diagram shown in Figure 6-4. The nets in the logic diagram correspond exactly to the declared nets.

Example 6-8 *Verilog Code for Edge-Triggered D-flipflop*

```
// Edge-triggered D flipflop
module edge_dff(q, qbar, d, clk, clear);

// Inputs and outputs
output q,qbar;
input d, clk, clear;

// Internal variables
wire s, sbar, r, rbar,cbar;

// dataflow statements
//Create a complement of signal clear
assign cbar = ~clear;

// Input latches; A latch is level sensitive. An edge-sensitive
// flip-flop is implemented by using 3 SR latches.
```

Example 6-8 Verilog Code for Edge-Triggered D-flipflop (Continued)

```
assign  sbar = ~(rbar & s),
        s = ~(sbar & cbar & ~clk),
        r = ~(rbar & ~clk & s),
        rbar = ~(r & cbar & d);

// Output latch
assign  q = ~(s & qbar),
        qbar = ~(q & r & cbar);

endmodule
```

The design block is now ready. Now we must instantiate the design block inside the stimulus block to test the design. The stimulus block is shown in Example 6-9. The clock has a time period of 20 with a 50% duty cycle.

Example 6-9 Stimulus Module for Ripple Counter

```
// Top level stimulus module
module stimulus;

// Declare variables for stimulating input
reg CLOCK, CLEAR;
wire [3:0] Q;

initial
        $monitor($time, " Count Q = %b Clear= %b",  Q[3:0],CLEAR);

// Instantiate the design block counter
counter c1(Q, CLOCK, CLEAR);

// Stimulate the Clear Signal
initial
begin
        CLEAR = 1'b1;
        #34 CLEAR = 1'b0;
        #200 CLEAR = 1'b1;
        #50 CLEAR = 1'b0;
end
// Set up the clock to toggle every 10 time units
initial
begin
```

Example 6-9 Stimulus Module for Ripple Counter (Continued)

```
        CLOCK = 1'b0;
        forever #10 CLOCK = ~CLOCK;
end

// Finish the simulation at time 400
initial
begin
        #400 $finish;
end

endmodule
```

The output of the simulation is shown below. Note that the clear signal resets the count to zero.

```
  0 Count Q = 0000 Clear= 1
 34 Count Q = 0000 Clear= 0
 40 Count Q = 0001 Clear= 0
 60 Count Q = 0010 Clear= 0
 80 Count Q = 0011 Clear= 0
100 Count Q = 0100 Clear= 0
120 Count Q = 0101 Clear= 0
140 Count Q = 0110 Clear= 0
160 Count Q = 0111 Clear= 0
180 Count Q = 1000 Clear= 0
200 Count Q = 1001 Clear= 0
220 Count Q = 1010 Clear= 0
234 Count Q = 0000 Clear= 1
284 Count Q = 0000 Clear= 0
300 Count Q = 0001 Clear= 0
320 Count Q = 0010 Clear= 0
340 Count Q = 0011 Clear= 0
360 Count Q = 0100 Clear= 0
380 Count Q = 0101 Clear= 0
```

6.6 Summary

- *Continuous assignment* is one of the main constructs used in dataflow modeling. A continuous assignment is always active and the assignment expression is evaluated as soon as one of the right-hand-side variables changes. The left-hand side of a continuous assignment must be a net. Any logic function can be realized with continuous assignments.
- Delay values control the time between the change in a right-hand-side variable and when the new value is assigned to the left-hand side. Delays on a net can be defined in the **`assign`** statement, implicit continuous assignment, or net declaration.
- Assignment statements contain expressions, operators, and operands.
- The operator types are *arithmetic, logical, relational, equality, bitwise, reduction, shift, concatenation, replication,* and *conditional.* Unary operators require one operand, binary operators require two operands, and ternary require three operands. The concatenation operator can take any number of operands.
- The *conditional operator* behaves like a multiplexer in hardware or like the if-then-else statement in programming languages.
- Dataflow description of a circuit is more concise than a gate-level description. The 4-to-1 multiplexer and the 4-bit full adder discussed in the gate-level modeling chapter can also be designed by use of dataflow statements. Two dataflow implementations for both circuits were discussed. A 4-bit ripple counter using negative edge-triggered D-flipflops was designed.

6.7 Exercises

1. A *full subtractor* has three 1-bit inputs x, y, and z (previous borrow) and two 1-bit outputs D (difference) and B (borrow). The logic equations for D and B are as follows:

   $D = x'.y'.z + x'.y.z' + x.y'.z' + x.y.z$

   $B = x'.y + x'.z + y.z$

Write the full Verilog description for the full subtractor module, including I/O ports (Remember that + in logic equations corresponds to a logical or operator (| |) in dataflow). Instantiate the subtractor inside a stimulus block and test all eight possible combinations of *x*, *y*, and z given in the following truth table.

x	y	z	B	D
0	0	0	0	0
0	0	1	1	1
0	1	0	1	1
0	1	1	1	0
1	0	0	0	1
1	0	1	0	0
1	1	0	0	0
1	1	1	1	1

2. A *magnitude comparator* checks if one number is greater than or equal to or less than another number. A 4-bit magnitude comparator takes two 4-bit numbers, *A* and *B*, as input. We write the bits in *A* and *B* as follows. Leftmost bit is the most significant bit.

   A = A(3) A(2) A(1) A(0)

   B = B(3) B(2) B(1) B(0)

   The magnitude can be compared by comparing the numbers bit by bit, starting with the most significant bit. If any bit mismatches, the number with bit 0 is the lower number. To realize this functionality in logic equations, let us define an intermediate variable. Notice that the function below is an **xnor** function.

   x(i) = A(i).B(i) + A(i)'.B(i)'

   The three outputs of the magnitude comparator are *A_gt_B*, *A_lt_B*, *A_eq_B*. They are defined with the following logic equations.

   A_gt_B = A(3).B(3)' + x(3).A(2).B(2)' + x(3).x(2).A(1).B(1)' + x(3).x(2).x(1).A(0).B(0)'

   A_lt_B = A(3)'.B(3) + x(3).A(2)'.B(2) + x(3).x(2).A(1)'.B(1) + x(3).x(2).x(1).A(0)'.B(0)

   A_eq_B = x(3).x(2).x(1).x(0)

Write the Verilog description of the module *magnitude_comparator*. Instantiate the magnitude comparator inside the stimulus module and try out a few combinations of *A* and *B*.

3. A *synchronous counter* can be designed by using *master-slave JK flip-flops*. Design a 4-bit synchronous counter. Circuit diagrams for the synchronous counter and the JK flip-flop are given below. *Clear* signal is active low. Data gets latched on the positive edge of *clock*, and the output of the flip-flop appears on the negative edge of *clock*. Counting is disabled when *count_enable* signal is low. Write the dataflow description for the synchronous counter. Write a stimulus file that exercises *clear* and *count_enable*. Display the output count *Q[3:0]*.



Figure 6-5 Master-Slave JK-flipflop



Figure 6-6 4-bit Synchronous Counter with clear and count_enable

Behavioral Modeling 7

With the increasing complexity of digital design, it has become vitally important to make wise design decisions early in a project. Designers need to be able to evaluate the trade-offs of various architectures and algorithms before they decide on the optimum architecture and algorithm to implement in hardware. Thus, architectural evaluation takes place at an algorithmic level where the designers do not necessarily think in terms of logic gates or data flow but in terms of the algorithm they wish to implement in hardware. They are more concerned about the behavior of the algorithm and its performance. Only after the high-level architecture and algorithm are finalized, do designers start focusing on building the digital circuit to implement the algorithm.

Verilog provides designers the ability to describe design functionality in an algorithmic manner. In other words, the designer describes the *behavior* of the circuit. Thus, behavioral modeling represents the circuit at a very high level of abstraction. Design at this level resembles C programming more than it resembles digital circuit design. Behavioral Verilog constructs are similar to C language constructs in many ways. Verilog is rich in behavioral constructs that provide the designer with a great amount of flexibility.

Learning Objectives

- Explain the significance of structured procedures **`always`** and **`initial`** in behavioral modeling.
- Define *blocking* and *nonblocking* procedural assignments.
- Understand delay-based timing control mechanism in behavioral modeling. Use *regular delays, intra-assignment delays,* and *zero delays.*
- Describe event-based timing control mechanism in behavioral modeling. Use *regular event control, named event control,* and *event OR control*
- Use level-sensitive timing control mechanism in behavioral modeling.
- Explain conditional statements using **`if`** and **`else`**.

- Describe multiway branching, using **case**, **casex**, and **casez** statements.
- Understand looping statements such as **while**, **for**, **repeat**, and **forever**.
- Define *sequential* and *parallel* blocks.
- Understand *naming of blocks* and *disabling* of named blocks.
- Use behavioral modeling statements in practical examples.

7.1 Structured Procedures

There are two structured procedure statements in Verilog: **always** and **initial**. These statements are the two most basic statements in behavioral modeling. All other behavioral statements can appear only inside these structured procedure statements.

Verilog is a concurrent programming language unlike the C programming language, which is sequential in nature. Activity flows in Verilog run in parallel rather than in sequence. Each **always** and **initial** statement represents a separate activity flow in Verilog. Each activity flow starts at simulation time 0. The statements **always** and **initial** cannot be nested. The fundamental difference between the two statements is explained in the following sections.

7.1.1 initial Statement

All statements inside an **initial** statement constitute an **initial** block. An **initial** block starts at time 0, executes exactly once during a simulation, and then does not execute again. If there are multiple **initial** blocks, each block starts to execute concurrently at time 0. Each block finishes execution independently of other blocks. Multiple behavioral statements must be grouped, typically using the keywords **begin** and **end**. If there is only one behavioral statement, grouping is not necessary. This is similar to the *begin-end* blocks in Pascal programming language or the { } grouping in the C programming language. Example 7-1 illustrates the use of the **initial** statement.

Example 7-1 initial Statement

```
module stimulus;

reg x,y, a,b, m;

initial
```

Example 7-1 *initial Statement*

```
    m = 1'b0; //single statement; does not need to be grouped

initial
begin
    #5 a = 1'b1; //multiple statements; need to be grouped
    #25 b = 1'b0;
end

initial
begin
    #10 x = 1'b0;
    #25 y = 1'b1;
end

initial
    #50 $finish;

endmodule
```

In the above example, the three initial statements start to execute in parallel at time 0. If a delay #<*delay*> is seen before a statement, the statement is executed <*delay*> time units after the current simulation time. Thus, the execution sequence of the statements inside the **`initial`** blocks will be as follows.

```
time            statement executed
0               m = 1'b0;
5               a = 1'b1;
10              x = 1'b0;
30              b = 1'b0;
35              y = 1'b1;
50              $finish;
```

The **`initial`** blocks are typically used for initialization, monitoring, waveforms and other processes that must be executed only once during the entire simulation run.

7.1.2 always Statement

All behavioral statements inside an **always** statement constitute an **always** block. The **always** statement starts at time 0 and executes the statements in the **always** block continuously in a looping fashion. This statement is used to model a block of activity that is repeated continuously in a digital circuit. An example is a clock generator module that toggles the clock signal every half cycle. In real circuits, the clock generator is active from time 0 to as long as the circuit is powered on. Example 7-2 illustrates one method to model a clock generator in Verilog.

Example 7-2 always Statement

```
module clock_gen;

reg clock;

//Initialize clock at time zero
initial
        clock = 1'b0;

//Toggle clock every half-cycle (time period = 20)
always
        #10 clock = ~clock;

initial
        #1000 $finish;

endmodule
```

In Example 7-2, the **always** statement starts at time 0 and executes the statement *clock* = *~clock* every 10 time units. Notice that the initialization of *clock* has to be done inside a separate **initial** statement. If we put the initialization of clock inside the **always** block, *clock* will be initialized every time the **always** is entered. Also, the simulation must be halted inside an **initial** statement. If there is no **$stop** or **$finish** statement to halt the simulation, the clock generator will run forever.

C programmers might draw an analogy between the **always** block and an infinite loop. But hardware designers tend to view it as a continuously repeated activity in a digital circuit starting from power on. The activity is stopped only by power off (**$finish**) or by an interrupt (**$stop**).

7.2 Procedural Assignments

Procedural assignments update values of **reg**, **integer**, **real**, or **time** variables. The value placed on a variable will remain unchanged until another procedural assignment updates the variable with a different value. These are unlike continuous assignments discussed in Chapter 6, *Dataflow Modeling*, where one assignment statement can cause the value of the right-hand-side expression to be continuously placed onto the left-hand-side net. The syntax for the simplest form of procedural assignment is shown below.

```
<assignment>
   ::= <lvalue> = <expression>
```

The left-hand side of a procedural assignment <*lvalue*> can be one of the following:

- A **reg**, integer, real, or time register variable or a memory element
- A bit select of these variables (e.g., *addr[0]*)
- A part select of these variables (e.g., *addr[31:16]*)
- A concatenation of any of the above

The right-hand side can be any expression that evaluates to a value. In behavioral modeling all operators listed in Table 6-1 on page 92 can be used in behavioral expressions.

There are two types of procedural assignment statements: *blocking* and *nonblocking*.

7.2.1 Blocking assignments

Blocking assignment statements are executed in the order they are specified in a sequential block. A blocking assignment will not block execution of statements that follow in a parallel block. Both parallel and sequential blocks are discussed in Section 7.7, *Sequential and Parallel Blocks*, The = operator is used to specify blocking assignments.

Example 7-3 *Blocking Statements*

```
reg x, y, z;
reg [15:0] reg_a, reg_b;
integer count;
```

Example 7-3 Blocking Statements (Continued)

```
//All behavioral statements must be inside an initial or always block
initial
begin
        x = 0; y = 1; z = 1; //Scalar assignments
        count = 0; //Assignment to integer variables
        reg_a = 16'b0; reg_b = reg_a; //initialize vectors

        #15 reg_a[2] = 1'b1; //Bit select assignment with delay
      #10 reg_b[15:13] = {x, y, z} //Assign result of concatenation to
                                    // part select of a vector
        count = count + 1; //Assignment to an integer (increment)
end
```

In Example 7-3, the statement *y = 1* is executed only after *x = 0* is executed. The behavior in a particular block is sequential in a *begin-end* block if blocking statements are used, because the statements can execute only in sequence. The statement *count = count + 1* is executed last. The simulation times at which the statements are executed are as follows:

- All statements *x = 0* through *reg_b = reg_a* are executed at time 0
- Statement *reg_a[2] = 0* at time = 15
- Statement *reg_b[15:13] = {x, y, z}* at time = 25
- Statement *count = count + 1* at time = 25
- Since there is a delay of 15 and 10 in the preceding statements, *count = count + 1* will be executed at time = 25 units

Note that for procedural assignments to registers, if the right-hand side has more bits than the register variable, the right-hand side is truncated to match the width of the register variable. The least significant bits are selected and the most significant bits are discarded. If right-hand side has fewer bits, zeros are filled in the most significant bits of the register variable.

7.2.2 Nonblocking Assignments

Nonblocking assignments allow scheduling of assignments without blocking execution of the statements that follow in a sequential block. A **<=** operator is used to specify nonblocking assignments. Note that this operator has the same

symbol as a relational operator, *less_than_equal_to*. The operator <= is interpreted as a relational operator in an expression and as an assignment operator in the context of a nonblocking assignment. To illustrate the behavior of nonblocking statements and its difference from blocking statements, let us consider Example 7-4, convert some blocking assignments to nonblocking assignments, and observe the behavior.

Example 7-4 *Nonblocking Assignments*

```
reg x, y, z;
reg [15:0] reg_a, reg_b;
integer count;


//All behavioral statements must be inside an initial or always block
initial
begin
        x = 0; y = 1; z = 1; //Scalar assignments
        count = 0; //Assignment to integer variables
        reg_a = 16'b0; reg_b = reg_a; //Initialize vectors

        reg_a[2] <= #15 1'b1; //Bit select assignment with delay
      reg_b[15:13] <= #10 {x, y, z}; //Assign result of concatenation
                                     //to part select of a vector
        count <= count + 1; //Assignment to an integer (increment)
end
```

In this example the statements *x = 0* through *reg_b = reg_a* are executed sequentially at time 0. Then, the three nonblocking assignments are processed at the same simulation time.

1. *reg_a[2] = 0* is scheduled to execute after 15 units (i.e., time = 15)
2. *reg_b[15:13] = {x, y, z}* is scheduled to execute after 10 time units (i.e., time = 10)
3. *count = count + 1* is scheduled to be executed without any delay (i.e., time = 0)

Thus, the simulator schedules a *nonblocking* assignment statement to execute and continues to the next statement in the block without waiting for the nonblocking statement to complete execution. Typically, nonblocking assignment statements are executed last in the time step in which they are scheduled, that is, after all the blocking assignments in that time step are executed.

Application of nonblocking assignments

Having described the behavior of nonblocking assignments, it is important to understand why they are used in digital design. They are used as a method to model several concurrent data transfers that take place after a common event. Consider the following example where three concurrent data transfers take place at the positive edge of clock.

```
always @(posedge clock)
begin
    reg1 <= #1 in1;
    reg2 <= @(negedge clock) in2 ^ in3;
    reg3 <= #1 reg1; //The old value of reg1
end
```

At each positive edge of clock, the following sequence takes place for the nonblocking assignments.

1. A *read* operation is performed on each right-hand-side variable, *in1*, *in2*, *in3*, and *reg1*, at the positive edge of clock. The right-hand-side expressions are evaluated, and the results are stored internally in the simulator.
2. The *write* operations to the left-hand-side variables are scheduled to be executed at the time specified by the intra-assignment delay in each assignment, that is, schedule *"write"* to *reg1* after 1 time unit, to *reg2* at the next negative edge of clock, and to *reg3* after 1 time unit.
3. The *write* operations are executed at the scheduled time steps. The order in which the write operations are executed is not important because the internally stored right-hand-side expression values are used to assign to the left-hand-side values. For example, note that *reg3* is assigned the old value of *reg1* that was stored after the read operation, even if the write operation wrote a new value to *reg1* before the write operation to *reg3* was executed.

Thus, the final values of *reg1*, *reg2*, and *reg3* are not dependent on the order in which the assignments are processed.

To understand the read and write operations further, consider Example 7-5, which is intended to swap the values of registers *a* and *b* at each positive edge of clock, using two concurrent **always** blocks.

Example 7-5 *Nonblocking Statements to Eliminate Race Conditions*

```
//Illustration 1: Two concurrent always blocks with blocking
//statements
always @(posedge clock)
          a = b;

always @(posedge clock)
          b = a;

//Illustration 2: Two concurrent always blocks with nonblocking
//statements
always @(posedge clock)
          a <= b;

always @(posedge clock)
          b <= a;
```

In Example 7-5, in *illustration 1*, there is a race condition when blocking statements are used. Either $a = b$ would be executed before $b = a$, or vice versa, depending on the simulator implementation. Thus, values of registers a and b will not be swapped. Instead, both registers will get the same value (previous value of a or b), based on the Verilog simulator implementation.

However, nonblocking statements used in *illustration 2* eliminate the race condition. At the positive edge of clock, the values of all right-hand-side variables are "*read*," and the right-hand-side expressions are evaluated and stored in *temporary* variables. During the write operation, the values stored in the temporary variables are assigned to the left-hand-side variables. Separating the read and write operations ensures that the values of registers a and b are swapped correctly, regardless of the order in which the write operations are performed. Example 7-6 shows how nonblocking assignments in *illustration 2* might be processed by a simulator.

Example 7-6 *Processing of Nonblocking Assignments*

```
//Process nonblocking assignments by using temporary variables
always @(posedge clock)
begin
   //Read operation
   //store values of right-hand-side expressions in temporary variables
   temp_a = a;
   temp_b = b;
```

Example 7-6 Processing of Nonblocking Assignments (Continued)

```
    //Write operation
    //Assign values of temporary variables to left-hand-side variables
    a = temp_b;
    b = temp_a;
end
```

For digital design, use of nonblocking assignments in place of blocking assignments is highly recommended in places where concurrent data transfers take place after a common event. In such cases, blocking assignments can potentially cause race conditions because the final result depends on the order in which the assignments are evaluated. Nonblocking assignments can be used effectively to model concurrent data transfers because the final result is not dependent on the order in which the assignments are evaluated. Typical applications of nonblocking assignments include pipeline modeling and modeling of several mutually exclusive data transfers. On the downside, nonblocking assignments can potentially cause a degradation in the simulator performance and increase in memory usage.

7.3 Timing Controls

Various behavioral timing control constructs are available in Verilog. In Verilog, if there are no timing control statements, the simulation time does not advance. Timing controls provide a way to specify the simulation time at which procedural statements will execute. There are three methods of timing control: *delay-based timing control, event-based timing control,* and *level-sensitive timing control.*

7.3.1 Delay-Based Timing Control

Delay-based timing control in an expression specifies the time duration between when the statement is encountered and when it is executed. We used delay-based timing control statements when writing few modules in the preceding chapters but did not explain them in detail. In this section we will discuss delay-based timing control statements. Delays are specified by the symbol **#**. Syntax for the delay-based timing control statement is shown below.

```
<delay>
    ::= #<NUMBER>
    ||= #<identifier>
    ||= #(<mintypmax_expression> <,<mintypmax_expression>>*)
```

Delay-based timing control can be specified by a *number, identifier,* or a *mintypmax_expression.* There are three types of delay control for procedural assignments: *regular delay control, intra-assignment delay control,* and *zero delay control.*

Regular delay control

Regular delay control is used when a non-zero delay is specified to the left of a procedural assignment. Usage of regular delay control is shown in Example 7-7.

Example 7-7 Regular Delay Control

```
//define parameters
parameter latency = 20;
parameter delta = 2;
//define register variables
reg x, y, z, p, q;

initial
begin
        x = 0; // no delay control
        #10 y = 1; // delay control with a number. Delay execution of
                                    // y = 1 by 10 units

      #latency z = 0; //Delay control with identifier.Delay of 20 units
        #(latency + delta) p = 1; // Delay control with expression

       #y x = x + 1; // Delay control with identifier. Take value of y.

        #(4:5:6) q = 0; // Minimum, typical and maximum delay values.
                    //Discussed in gate-level modeling chapter.
end
```

In Example 7-7, the execution of a procedural assignment is delayed by the number specified by the delay control. For begin-end groups, delay is always relative to time when the statement is encountered. Thus, *y =1* is executed 10 units after it is encountered in the activity flow.

Intra-assignment delay control

Instead of specifying delay control to the left of the assignment, it is possible to assign a delay to the right of the assignment operator. Such delay specification alters the flow of activity in a different manner. Example 7-8 shows the contrast between intra-assignment delays and regular delays.

Example 7-8 Intra-assignment Delays

```
//define register variables
reg x, y, z;

//intra assignment delays
initial
begin
        x = 0; z = 0;
        y = #5 x + z; //Take value of x and z at the time=0, evaluate
                      //x + z and then wait 5 time units to assign value
                     //to y.

end

//Equivalent method with temporary variables and regular delay control
initial
begin
        x = 0; z = 0;
        temp_xz = x + z;
       #5 y = temp_xz; //Take value of x + z at the current time and
              //store it in a temporary variable. Even though x and z
                 //might change between 0 and 5,
                 //the value assigned to y at time 5 is unaffected.
end
```

Note the difference between intra-assignment delays and regular delays. Regular delays defer the execution of the entire assignment. Intra-assignment delays compute the right-hand-side expression at the current time and defer the assignment of the computed value to the left-hand-side variable. Intra-assignment delays are like using regular delays with a temporary variable to store the current value of a right-hand-side expression.

Zero delay control

Procedural statements in different *always-initial* blocks may be evaluated at the same simulation time. The order of execution of these statements in different *always-initial* blocks is nondeterministic. Zero delay control is a method to ensure that a statement is executed last, after all other statements in that simulation time are executed. This is used to eliminate race conditions. However, if there are multiple zero delay statements, the order between them is nondeterministic. Example 7-9 illustrates zero delay control.

Example 7-9 Zero Delay Control

```
initial
begin
    x = 0;
    y = 0;
end

initial
begin
    #0 x = 1; //zero delay control
    #0 y = 1;
end
```

In Example 7-9, four statements—$x = 0$, $y = 0$, $x = 1$, $y = 1$—are to be executed at simulation time 0. However, since $x = 1$ and $y = 1$ have *#0*, they will be executed last. Thus, at the end of time 0, x will have value `1` and y will have value `1`. The order in which $x = 1$ and $y = 1$ are executed is not deterministic.

The above example was used as an illustration. The practice of assigning two different values to a variable in a single time step is generally not recommended and may cause race conditions in the design. However, *#0* provides a useful mechanism to control the order of execution of statements in a simulation.

7.3.2 Event-Based Timing Control

An *event* is the change in the value on a register or a net. Events can be utilized to trigger execution of a statement or a block of statements. There are four types of event-based timing control: *regular event control, named event control, event OR control,* and *level-sensitive timing control.*

Regular event control

The **@** symbol is used to specify an event control. Statements can be executed on changes in signal value or at a *positive* or *negative* transition of the signal value. The keyword **posedge** is used for a negative transition, as shown in Example 7-10.

Example 7-10 Regular Event Control

```
@(clock) q = d; //q = d is executed whenever signal clock changes value
@(posedge clock) q = d; //q = d is executed whenever signal clock does
                        //a positive transition ( 0 to 1,x or z,
                        // x to 1, z to 1 )
@(negedge clock) q = d; //q = d is executed whenever signal clock does
                        //a negative transition ( 1 to 0,x or z,
                        //x to 0, z to 0)
q = @(posedge clock) d; //d is evaluated immediately and assigned
                        //to q at the positive edge of clock
```

Named event control

Verilog provides the capability to *declare* an event and then *trigger* and *recognize* the occurrence of that event (see Example 7-11). The event does not hold any data. A named event is declared by the keyword **event**. An event is triggered by the symbol **->**. The triggering of the event is recognized by the symbol **@**.

Example 7-11 Named Event Control

```
//This is an example of a data buffer storing data after the
//last packet of data has arrived.

event received_data; //Define an event called received_data

always @(posedge clock) //check at each positive clock edge
begin
        if(last_data_packet) //If this is the last data packet
                ->received_data; //trigger the event received_data
end

always @(received_data) //Await triggering of event received_data
                        //When event is triggered, store all four
```

Example 7-11 Named Event Control (Continued)

```
                        //packets of received data in data buffer
                    //use concatenation operator { }
    data_buf = {data_pkt[0], data_pkt[1], data_pkt[2], data_pkt[3]};
```

Event OR control

Sometimes a transition on any one of multiple signals or events can trigger the execution of a statement or a block of statements. This is expressed as an OR of events or signals. The list of events or signals expressed as an OR is also known as a *sensitivity* list. The keyword **or** is used to specify multiple triggers, as shown in Example 7-12.

Example 7-12 Event OR Control

```
//A level-sensitive latch with asynchronous reset
always @( reset or clock or d)
                              //Wait for reset or clock or d to change
begin
        if (reset)              //if reset signal is high, set q to 0.
                q = 1'b0;
        else    if(clock)       //if clock is high, latch input
                q = d;
end
```

7.3.3 Level-Sensitive Timing Control

Event control discussed earlier waited for the change of a signal value or the triggering of an event. The symbol **@** provided edge-sensitive control. Verilog also allows level-sensitive timing control, that is, the ability to wait for a certain condition to be true before a statement or a block of statements is executed. The keyword **wait** is used for level-sensitive constructs.

```
always
   wait (count_enable) #20 count = count + 1;
```

In the above example, the value of *count_enable* is monitored continuously. If *count_enable* is **0**, the statement is not entered. If it is logical **1**, the statement *count = count + 1* is executed after 20 time units. If *count_enable* stays at 1, count will be incremented every 20 time units.

7.4 Conditional Statements

Conditional statements are used for making decisions based upon certain conditions. These conditions are used to decide whether or not a statement should be executed. Keywords **if** and **else** are used for conditional statements. There are three types of conditional statements. Usage of conditional statements is shown below. For formal syntax, see Appendix D, *Formal Syntax Definition*.

```
//Type 1 conditional statement. No else statement.
//Statement executes or does not execute.
if (<expression>) true_statement ;

//Type 2 conditional statement. One else statement
//Either true_statement or false_statement is evaluated
if (<expression>) true_statement ; else false_statement ;

//Type 3 conditional statement. Nested if-else-if.
//Choice of multiple statements. Only one is executed.
if (<expression1>) true_statement1 ;
else if (<expression2>) true_statement2 ;
else if (<expression3>) true_statement3 ;
else default_statement ;
```

The *<expression>* is evaluated. If it is true (1 or a non-zero value), the *true_statement* is executed. However, if it is false (zero) or ambiguous (**x** or **z**), the *false_statement* is executed. The *<expression>* can contain any operators mentioned in Table 6-1 on page 92. Each *true_statement* or *false_statement* can be a single statement or a block of multiple statements. A block must be grouped, typically by using keywords **begin** and **end**. A single statement need not be grouped.

Example 7-13 Conditional Statement Examples

```
//Type 1 statements
if(!lock) buffer = data;
if(enable) out = in;
```

Example 7-13 Conditional Statement Examples (Continued)

```
//Type 2 statements
if (number_queued < MAX_Q_DEPTH)
begin
        data_queue = data;
        number_queued = number_queued + 1;
end
else
        $display("Queue Full. Try again");

//Type 3 statements
//Execute statements based on ALU control signal.
if (alu_control == 0)
        y = x + z;
else if(alu_control == 1)
        y = x - z;
else if(alu_control == 2)
        y = x * z;
else
        $display("Invalid ALU control signal");
```

7.5 Multiway Branching

In *type 3* conditional statement in Section 7.4, *Conditional Statements*, there were many alternatives, from which one was chosen. The nested *if-else-if* can become unwieldy if there are too many alternatives. A shortcut to achieve the same result is to use the **case** statement.

7.5.1 case Statement

The keywords **case**, **endcase**, and **default** are used in the *case* statement. .

```
case (expression)
    alternative1: statement1;
    alternative2: statement2;
    alternative3: statement3;
        ...
        ...
    default: default_statement;
endcase
```

Each of *statement1, statement2 ..., default_statement* can be a single statement or a block of multiple statements. A block of multiple statements must be grouped by keywords **begin** and **end**. The expression is compared to the alternatives in the order they are written. For the first alternative that matches, the corresponding statement or block is executed. If none of the alternatives match, the *default_statement* is executed. The *default_statement* is optional. Placing of multiple default statements in one **case** statement is not allowed. The **case** statements can be nested. The following Verilog code implements the type 3 conditional statement in Example 7-13.

```
//Execute statements based on the ALU control signal
reg [1:0] alu_control;
...
...
case (alu_control)
  2'd0 : y = x + z;
  2'd1 : y = x - z;
  2'd2 : y = x * z;
  default : $display("Invalid ALU control signal");
endcase
```

The **case** statement can also act like a many-to-one multiplexer. To understand this, let us model the 4-to-1 multiplexer in Section 6.5, *Examples*, on page 102, using case statements. The I/O ports are unchanged. Notice that an 8-to-1 or 16-to-1 multiplexer can also be easily implemented by **case** statements.

Example 7-14 4-to-1 Multiplexer with case Statement

```
module mux4_to_1 (out, i0, i1, i2, i3, s1, s0);

// Port declarations from the I/O diagram
output out;
input i0, i1, i2, i3;
input s1, s0;
reg out;

always @(s1 or s0 or i0 or i1 or i2 or i3)
case ({s1, s0}) //Switch based on concatenation of control signals
        2'd0 : out = i0;
        2'd1 : out = i1;
        2'd2 : out = i2;
        2'd3 : out = i3;
        default: $display("Invalid control signals");
endcase

endmodule
```

The **case** statement compares **0**, **1**, **x**, and **z** values in the expression and the alternative bit for bit. If the expression and the alternative are of unequal bit width, they are zero filled to match the bit width of the widest of the expression and the alternative. In Example 7-15, we will define a 1-to-4 demultiplexer for which outputs are completely specified, that is, definitive results are provided even for **x** and **z** values on the select signal.

Example 7-15 Case Statement with x and z

```
module demultiplexer1_to_4 (out0, out1, out2, out3, in, s1, s0);

// Port declarations from the I/O diagram
output out0, out1, out2, out3;
reg  out0,  out1,  out2,  out3;
input in;
input s1, s0;
```

Example 7-15 Case Statement with x and z (Continued)

```
always @(s1 or s0 or in)
case ({s1, s0}) //Switch based on control signals
   2'b00 : begin  out0 = in;  out1 = 1'bz;  out2 = 1'bz; out3 = 1'bz; end
   2'b01 : begin  out0 = 1'bz;  out1 = in;  out2 = 1'bz; out3 = 1'bz; end
   2'b10 : begin  out0 = 1'bz;  out1 = 1'bz;  out2 = in; out3 = 1'bz; end
   2'b11 : begin  out0 = 1'bz;  out1 = 1'bz;  out2 = 1'bz; out3 = in; end

    //Account for unknown signals on select. If any select signal is x
    //then outputs are x. If any select signal is z, outputs are z.
    //If one is x and the other is z, x gets higher priority.
    2'bx0, 2'bx1, 2'bxz, 2'bxx, 2'b0x, 2'b1x, 2'bzx :
         begin
              out0 = 1'bx;  out1 = 1'bx;  out2 = 1'bx;  out3 = 1'bx;
         end
    2'bz0, 2'bz1, 2'bzz, 2'b0z, 2'b1z :
         begin
              out0 = 1'bz;  out1 = 1'bz;  out2 = 1'bz;  out3 = 1'bz;
         end
    default: $display("Unspecified control signals");
endcase

endmodule
```

In the demultiplexer shown above, multiple input signal combinations such as *2'bz0, 2'bz1, 2,bzz, 2'b0z,* and *2'b1z* that cause the same block to be executed are put together with a *comma* (**,**) symbol.

7.5.2 casex, casez Keywords

There are two variations of the **case** statement. They are denoted by keywords, **casex** and **casez**.

- **casez** treats all **z** values in the case alternatives or the case expression as don't cares. All bit positions with **z** can also represented by **?** in that position.
- **casex** treats all **x** and **z** values in the case item or the case expression as don't cares.

The use of **casex** and **casez** allows comparison of only non-**x** or -**z** positions in the case expression and the case alternatives. Example 7-16 illustrates the decoding of state bits in a finite state machine using a **casex** statement. The use of **casez** is similar. Only one bit is considered to determine the next state and the other bits are ignored.

Example 7-16 casex Use

```
reg [3:0] encoding;
integer state;

casex (encoding) //logic value x represents a don't care bit.
4'b1xxx : next_state = 3;
4'bx1xx : next_state = 2;
4'bxx1x : next_state = 1;
4'bxxx1 : next_state = 0;
default : next_state = 0;
endcase
```

Thus, an input *encoding = 4'b10xz* would cause *next_state = 3* to be executed.

7.6 Loops

There are four types of looping statements in Verilog: *while, for, repeat,* and *forever.* The syntax of these loops is very similar to the syntax of loops in the C programming language. All looping statements can appear only inside an **initial** or **always** block. Loops may contain delay expressions.

7.6.1 While Loop

The keyword **while** is used to specify this loop. The **while** loop executes until the *while-expression* becomes false. If the loop is entered when the *while-expression* is false, the loop is not executed at all. Each expression can contain the operators in Table 6-1 on page 92. Any logical expression can be specified with these operators. If multiple statements are to be executed in the loop, they must be grouped typically using keywords **begin** and **end**. Example 7-17 illustrates the use of the **while** loop.

Example 7-17 While Loop

```
//Illustration 1: Increment count from 0 to 127. Exit at count 128.
//Display the count variable.
integer count;

initial
begin
        count = 0;

        while (count < 128) //Execute loop till count is 127.
                        //exit at count 128
        begin
                $display("Count = %d", count);
             count = count + 1;
        end
end


//Illustration 2: Find the first bit with a value 1 in flag (vector
variable)
`define TRUE 1'b1;
`define FALSE 1'b0;
reg [15:0] flag;
integer i; //integer to keep count
reg continue;

initial
begin
  flag = 16'b 0010_0000_0000_0000;
  i = 0;
  continue = `TRUE;

  while((i < 16) && continue ) //Multiple conditions using
operators.
  begin
    if (flag[i])
    begin
      $display("Encountered a TRUE bit at element number %d", i);
      continue = `FALSE;
   end
   i = i + 1;
  end
end
```

7.6.2 For Loop

The keyword **for** is used to specify this loop. The **for** loop contains three parts:

- An initial condition
- A check to see if the terminating condition is true
- A procedural assignment to change value of the control variable

The counter described in Example 7-17 can be coded as a **for** loop (Example 7-18). The initialization condition and the incrementing procedural assignment are included in the **for** loop and do not need to be specified separately. Thus, the **for** loop provides a more compact loop structure than the **while** loop. Note, however, that the **while** loop is more general purpose than the **for** loop. The **for** loop cannot be used in place of the while loop in all situations. p

Example 7-18 For Loop

```
integer count;

initial
    for ( count=0; count < 128; count = count + 1)
            $display("Count = %d", count);
```

for loops can also be used to initialize an array or memory, as shown below.

```
//Initialize array elements
`define MAX_STATES 32
integer state [0:`MAX_STATES-1];//Integer array state with elements 0:31
integer i;

initial
begin
    for(i = 0; i < 32; i = i + 2) //initialize all even locations with 0
        state[i] = 0;
    for(i = 1; i < 32; i = i + 2) //initialize all odd locations with 1
        state[i] = 1;
end
```

for loops are generally used when there is a fixed beginning and end to the loop. If the loop is simply looping on a certain condition, it is better to use the **while** loop.

7.6.3 Repeat Loop

The keyword **repeat** is used for this loop. The **repeat** construct executes the loop a fixed number of times. A **repeat** construct cannot be used to loop on a general logical expression. A **while** loop is used for that purpose. A **repeat** construct must contain a number, which can be a constant, a variable or a signal value. However, if the number is a variable or signal value, it is evaluated only when the loop starts and not during the loop execution.

The counter in Example 7-17 can be expressed with the repeat loop, as shown in *Illustration 1* in Example 7-19. *Illustration 2* shows how to model a data buffer that latches data at the positive edge of clock for the next eight cycles after it receives a data start signal.

Example 7-19 Repeat Loop

```
//Illustration 1 : increment and display count from 0 to 127
integer count;

initial
begin
   count = 0;
   repeat(128)
   begin
        $display("Count = %d", count);
        count = count + 1;
   end
end

//Illustration 2 : Data buffer module example
//After it receives a data_start signal.
//Reads data for next 8 cycles.

module data_buffer(data_start, data, clock);

parameter cycles = 8;
input data_start;
input [15:0] data;
input clock;

reg [15:0] buffer [0:7];
integer i;

always @(posedge clock)
begin
```

Example 7-19 Repeat Loop (Continued)

```
  if(data_start) //data start signal is true
  begin
    i = 0;
    repeat(cycles) //Store data at the posedge of next 8 clock
                   //cycles
    begin
      @(posedge clock) buffer[i] = data; //waits till next
                                         // posedge to latch data
      i = i + 1;
    end
  end
end

endmodule
```

7.6.4 Forever loop

The keyword **forever** is used to express this loop. The loop does not contain any expression and executes forever until the **$finish** task is encountered. The loop is equivalent to a **while** loop with an expression that always evaluates to true, e.g., *while (1)*. A forever loop can be exited by use of the **disable** statement.

A **forever** loop is typically used in conjunction with timing control constructs. If timing control constructs are not used, the Verilog simulator would execute this statement infinitely without advancing simulation time and the rest of the design would never be executed. Example 7-20 explains the use of the **forever** statement.

Example 7-20 Forever Loop

```
//Example 1: Clock generation
//Use forever loop instead of always block
reg clock;

initial
begin
        clock = 1'b0;
        forever #10 clock = ~clock; //Clock with period of 20 units
end
```

Example 7-20 Forever Loop

```
//Example 2: Synchronize two register values at every positive edge of
//clock
reg clock;
reg x, y;

initial
        forever @(posedge clock) x = y;
```

7.7 Sequential and Parallel Blocks

Block statements are used to group multiple statements to act together as one. In previous examples we used keywords **begin** and **end** to group multiple statements. Thus, we used sequential blocks where the statements in the block execute one after another. In this section we discuss the block types: *sequential* blocks and *parallel* blocks. We also discuss three special features of blocks: *named blocks*, *disabling* named blocks, and *nested* blocks.

7.7.1 Block Types

There are two types of blocks: *sequential* blocks and *parallel* blocks.

Sequential blocks

The keywords **begin** and **end** are used to group statements into sequential blocks. Sequential blocks have the following characteristics:

- The statements in a sequential block are processed in the order they are specified. A statement is executed only after its preceding statement completes execution (except for nonblocking assignments with intra-assignment timing control).
- If delay or event control is specified, it is relative to the simulation time when the previous statement in the block completed execution.

We have used numerous examples of sequential blocks in this book. Two more examples of sequential blocks are given in Example 7-21. Statements in the sequential block execute in order. In *illustration 1*, the final values are $x = 0$, $y = 1$, $z = 1$, $w = 2$ at simulation time 0. In *illustration 2*, the final values are the same except that the simulation time is 35 at the end of the block.

Example 7-21 Sequential Blocks

```
//Illustration 1: Sequential block without delay
reg x, y;
reg [1:0] z, w;

initial
begin
        x = 1'b0;
        y = 1'b1;
        z = {x, y};
        w = {y, x};
end

//Illustration 2: Sequential blocks with delay.
reg x, y;
reg [1:0] z, w;

initial
begin
        x = 1'b0; //completes at simulation time 0
        #5 y = 1'b1; //completes at simulation time 5
        #10 z = {x, y}; //completes at simulation time 15
        #20 w = {y, x}; //completes at simulation time 35
end
```

Parallel blocks

Parallel blocks, specified by keywords **fork** and **join**, provide interesting simulation features. Parallel blocks have the following characteristics.

- Statements in a parallel block are executed concurrently.
- Ordering of statements is controlled by the delay or event control assigned to each statement.
- If delay or event control is specified, it is relative to the time the block was entered.

Notice the fundamental difference between sequential and parallel blocks. All statements in a parallel block start at the time when the block was entered. Thus, the order in which the statements are written in the block is not important.

Let us consider the sequential block with delay in Example 7-21 and convert it to a parallel block. The converted Verilog code is shown in Example 7-22. The result of simulation remains the same except that all statements start in *parallel* at time 0. Hence, the block finishes at time 20 instead of time 35.

Example 7-22 Parallel Blocks

```
//Example 1: Parallel blocks with delay.
reg x, y;
reg [1:0] z, w;

initial
fork
        x = 1'b0; //completes at simulation time 0
        #5 y = 1'b1; //completes at simulation time 5
        #10 z = {x, y}; //completes at simulation time 10
        #20 w = {y, x}; //completes at simulation time 20
join
```

Parallel blocks provide a mechanism to execute statements in parallel. However, it is important to be careful with parallel blocks because of implicit race conditions that might arise if two statements that affect the same variable complete at the same time. Shown below is the parallel version of *illustration 1* from Example 7-21. Race conditions have been deliberately introduced in this example. All statements start at simulation time 0. The order in which the statements will execute is not known. Variables *z* and *w* will get values *1* and *2* if *x = 1'b0* and *y = 1'b1* execute first. Variables z and *w* will get values *2'bxx* and *2'bxx* if *x = 1'b0* and *y = 1'b1* execute last. Thus, the result of *z* and *w* is nondeterministic and dependent on the simulator implementation. In simulation time, all statements in the fork-join block are executed at once. However, in reality, CPUs running simulations can execute only one statement at a time. Different simulators execute statements in different order. Thus, the race condition is a limitation of today's simulators, not of the *fork-join* block.

```
//Parallel blocks with deliberate race condition
reg x, y;
reg [1:0] z, w;

initial
fork
        x = 1'b0;
        y = 1'b1;
```

```
        z = {x, y};
        w = {y, x};
join
```

The keyword **fork** can be viewed as splitting a single flow into independent flows. The keyword **join** can be seen as joining the independent flows back into a single flow. Independent flows operate concurrently.

7.7.2 Special Features of Blocks

We discuss three special features available with block statements: *nested blocks, named blocks,* and *disabling of named blocks.*

Nested blocks

Blocks can be nested. Sequential and parallel blocks can be mixed, as shown in Example 7-23.

Example 7-23 Nested Blocks

```
//Nested blocks
initial
begin
        x = 1'b0;
        fork
                #5 y = 1'b1;
                #10 z = {x, y};
        join
        #20 w = {y, x};
end
```

Named blocks

Blocks can be given names.

- Local variables can be declared for the named block.
- Named blocks are a part of the design hierarchy. Variables in a named block can be accessed by using hierarchical name referencing.
- Named blocks can be disabled, i.e., their execution can be stopped.

Example 7-24 shows naming of blocks and hierarchical naming of blocks.

Example 7-24 Named Blocks

```
//Named blocks
module top;

initial
begin: block1 //sequential block named block1
integer i; //integer i is static and local to block1
          // can be accessed by hierarchical name, top.block1.i
...
...
end

initial
fork: block2 //parallel block named block2
reg i; // register i is static and local to block2
               // can be accessed by hierarchical name, top.block2.i
...
...
join
```

Disabling named blocks

The keyword **disable** provides a way to terminate the execution of a block. **disable** can be used to get out of loops, handle error conditions, or control execution of pieces of code, based on a control signal. Disabling a block causes the execution control to be passed to the statement immediately succeeding the block. For C programmers, this is very similar to the `break` statement used to exit a loop. The difference is that a `break` statement can break the current loop only, whereas the keyword **disable** allows disabling of any named block in the design.

Consider the illustration in Example 7-17 on page 136, which looks for the first true bit in the flag. The while loop can be recoded, using the disable statement as shown in Example 7-25. The disable statement terminates the **while** loop as soon as a true bit is seen.

Example 7-25 Disabling Named Blocks

```
//Illustration: Find the first bit with a value 1 in flag (vector
//variable)
reg [15:0] flag;
integer i; //integer to keep count
```

Example 7-25 Disabling Named Blocks

```
initial
begin
  flag = 16'b 0010_0000_0000_0000;
  i = 0;
  begin: block1 //The main block inside while is named block1
  while(i < 16)
   begin
      if (flag[i])
      begin
         $display("Encountered a TRUE bit at element number %d", i);
         disable block1; //disable block1 because you found true bit.
      end
      i = i + 1;
   end
  end
end
```

7.8 Examples

In order to illustrate the use of behavioral constructs discussed earlier in this chapter, we consider three examples in this section. The first two, *4-to-1 multiplexer* and *4-bit counter* are taken from Section 6.5, *Examples*. Earlier, these circuits were designed by using dataflow statements. We will model these circuits with behavioral statements. The third example is a new example. We will design a *traffic signal controller*, using behavioral constructs, and simulate it.

7.8.1 4-to-1 Multiplexer

We can define a 4-to-1 multiplexer with the behavioral case statement. This multiplexer was defined, in Section 6.5.1, *4-to-1 Multiplexer*, by dataflow statements. It is described in Example 7-26 by behavioral constructs. The behavioral multiplexer can be substituted for the dataflow multiplexer; the simulation results will be identical.

Example 7-26 Behavioral 4-to-1 Multiplexer

```
// 4-to-1 multiplexer. Port list is taken exactly from
// the I/O diagram.
module mux4_to_1 (out, i0, i1, i2, i3, s1, s0);
```

Example 7-26 Behavioral 4-to-1 Multiplexer (Continued)

```
// Port declarations from the I/O diagram
output out;
input i0, i1, i2, i3;
input s1, s0;
//output declared as register
reg out;

//recompute the signal out if any input signal changes.
//All input signals that cause a recomputation of out to
//occur must go into the always @(...)  sensitivity list.
always @(s1 or s0 or i0 or i1 or i2 or i3)
begin
  case ({s1, s0})
  2'b00: out = i0;
  2'b01: out = i1;
  2'b10: out = i2;
  2'b11: out = i3;
  default: out = 1'bx;
  endcase
end

endmodule
```

7.8.2 4-bit Counter

In Section 6.5.3, *Ripple Counter*, we designed a 4-bit ripple carry counter. We will now design the 4-bit counter by using behavioral statements. At dataflow or gate level, the counter might be designed in hardware as ripple carry, synchronous counter, etc. But, at a behavioral level, we work at a very high level of abstraction and do not care about the underlying hardware implementation. We will only design functionality. The counter can be designed by using behavioral constructs, as shown in Example 7-27. Notice how concise the behavioral counter description is compared to its dataflow counterpart. If we substitute the counter in place of the dataflow counter, the simulation results will be exactly same, assuming that there are no **x** and **z** values on the inputs.

Example 7-27 Behavioral 4-bit Counter Description

```
//4-bit Binary  counter
module counter(Q , clock, clear);

// I/O ports
output [3:0] Q;
input clock, clear;
//output defined as register
reg [3:0] Q;

always @( posedge clear  or negedge clock)
begin
  if (clear)
     Q = 4'd0;
  else
     Q = (Q + 1) % 16;
end

endmodule
```

7.8.3 Traffic Signal Controller

This example is fresh and has not been discussed before in the book. We will design a *traffic signal controller,* using a *finite state machine* approach.

Specification

Consider a controller for traffic at the intersection of a main highway and a country road.



The following specifications must be considered.

- The traffic signal for the main highway gets highest priority because cars are continuously present on the main highway. Thus, the main highway signal remains green by default.
- Occasionally, cars from the country road arrive at the traffic signal. The traffic signal for the country road must turn green only long enough to let the cars on the country road go.
- As soon as there are no cars on the country road, the country road traffic signal turns yellow and then red and the traffic signal on the main highway turns green again.
- There is a sensor to detect cars waiting on the country road. The sensor sends a signal *X* as input to the controller. *X = 1* if there are cars on the country road; otherwise, *X= 0* .
- There are delays on transitions from *S1* to *S2*, from *S2* to *S3*, and from *S4* to *S0*. The delays must be controllable.

The state machine diagram and the state definitions for the traffic signal controller are shown in Figure 7-1.



State	Signals
S0	*Hwy = G Cntry = R*
S1	*Hwy = Y Cntry = R*
S2	*Hwy = R Cntry = R*
S3	*Hwy = R Cntry = G*
S4	*Hwy = R Cntry = Y*

Figure 7-1 FSM for Traffic Signal Controller

Verilog description

The traffic signal controller module can be designed with behavioral Verilog constructs, as shown in Example 7-28.

Example 7-28 Traffic Signal Controller

```
`define TRUE  1'b1
`define FALSE 1'b0
`define RED   2'd0
`define YELLOW  2'd1
`define GREEN 2'd2

//State definition     HWY           CNTRY
`define S0     3'd0  //GREEN         RED
`define S1     3'd1  //YELLOW        RED
`define S2     3'd2  //RED           RED
`define S3     3'd3  //RED           GREEN
`define S4     3'd4  //RED           YELLOW

//Delays
`define Y2RDELAY  3 //Yellow to red delay
`define R2GDELAY  2 //Red to green delay

module sig_control
    (hwy, cntry, X, clock, clear);

//I/O ports
output [1:0] hwy, cntry;
      //2-bit output for 3 states of signal
      //GREEN, YELLOW, RED;
reg [1:0] hwy, cntry;
      //declared output signals are registers

input X;
      //if TRUE, indicates that there is car on
      //the country road, otherwise FALSE

input clock, clear;


//Internal state variables
reg [2:0] state;
reg [2:0] next_state;

//Signal controller starts in S0 state
```

Example 7-28 Traffic Signal Controller (Continued)

```
initial
begin
  state = `S0;
  next_state = `S0;
  hwy = `GREEN;
  cntry = `RED;
end

//state changes only at positive edge of clock
always @(posedge clock)
  state = next_state;

//Compute values of main signal and country signal
always @(state)
begin
  case(state)
    `S0:  begin
            hwy = `GREEN;
            cntry = `RED;
          end
    `S1:  begin
            hwy = `YELLOW;
            cntry = `RED;
          end
    `S2:  begin
            hwy = `RED;
            cntry = `RED;
          end
    `S3:  begin
            hwy = `RED;
            cntry = `GREEN;
          end
    `S4:  begin
            hwy = `RED;
            cntry = `YELLOW;
          end
  endcase
end

//State machine using case statements
always @(state or clear or  X)
begin
  if(clear)
```

Example 7-28 Traffic Signal Controller (Continued)

```
    next_state = `S0;
  else
    case (state)
      `S0: if( X)
             next_state = `S1;
           else
             next_state = `S0;
      `S1: begin //delay some positive edges of clock
             repeat(`Y2RDELAY) @(posedge clock) ;
             next_state = `S2;
           end
      `S2: begin //delay some positive edges of clock
             repeat(`R2GDELAY) @(posedge clock)
             next_state = `S3;
           end
      `S3: if( X)
             next_state = `S3;
           else
             next_state = `S4;
      `S4: begin //delay some positive edges of clock
             repeat(`Y2RDELAY) @(posedge clock) ;
             next_state = `S0;
           end
      default: next_state = `S0;
    endcase
end

endmodule
```

Stimulus

Stimulus can be applied to check if the traffic signal transitions correctly when cars arrive on the country road. The stimulus file in Example 7-29 instantiates the traffic signal controller and checks all possible states of the controller.

Example 7-29 Stimulus for Traffic Signal Controller

```
//Stimulus Module
module stimulus;

wire [1:0] MAIN_SIG, CNTRY_SIG;
reg CAR_ON_CNTRY_RD;
```

Example 7-29 Stimulus for Traffic Signal Controller (Continued)

```
      //if TRUE, indicates that there is car on
      //the country road
reg CLOCK, CLEAR;

//Instantiate signal controller
sig_control SC(MAIN_SIG, CNTRY_SIG, CAR_ON_CNTRY_RD, CLOCK, CLEAR);

//Set up monitor
initial
  $monitor($time, "Main Sig = %b Country Sig = %b Car_on_cntry = %b",
                        MAIN_SIG, CNTRY_SIG, CAR_ON_CNTRY_RD);

//Set up clock
initial
begin
    CLOCK = `FALSE;
    forever #5 CLOCK = ~CLOCK;
end

//control clear signal
initial
begin
    CLEAR = `TRUE;
    repeat (5) @(negedge CLOCK);
    CLEAR = `FALSE;
end

//apply stimulus
initial
begin
    CAR_ON_CNTRY_RD = `FALSE;

    #200 CAR_ON_CNTRY_RD = `TRUE;
    #100 CAR_ON_CNTRY_RD = `FALSE;

    #200 CAR_ON_CNTRY_RD = `TRUE;
    #100 CAR_ON_CNTRY_RD = `FALSE;

    #200 CAR_ON_CNTRY_RD = `TRUE;
    #100 CAR_ON_CNTRY_RD = `FALSE;

    #100 $stop;
```

Example 7-29 Stimulus for Traffic Signal Controller (Continued)

```
end
endmodule
```

Note that we designed only the behavior of the controller without worrying about how it will be implemented in hardware.

7.9 Summary

We discussed digital circuit design with behavioral Verilog constructs.

- A behavioral description expresses a digital circuit in terms of the algorithms it implements. A behavioral description does not necessarily include the hardware implementation details. Behavioral modeling is used in the initial stages of a design process to evaluate various design-related trade-offs. Behavioral modeling is similar to C programming in many ways.
- Structured procedures **initial** and **always** form the basis of behavioral modeling. All other behavioral statements can appear only inside **initial** or **always** blocks. An **initial** block executes once; an **always** block executes continuously until simulation ends.
- Procedural assignments are used in behavioral modeling to assign values to register variables. *Blocking* assignments must complete before the succeeding statement can execute. *Nonblocking* assignments schedule assignments to be executed and continue processing to the succeeding statement.
- *Delay-based timing control, event-based timing control,* and *level-sensitive timing control* are three ways to control timing and execution order of statements in Verilog. *Regular delay, zero delay,* and *intra-assignment delay* are three types of delay-based timing control. *Regular event, named event,* and *event OR* are three types of event-based timing control. The **wait** statement is used to model level-sensitive timing control.
- Conditional statements are modeled in behavioral Verilog with **if** and **else** statements. If there are multiple branches, use of **case** statements is recommended. **casex** and **casez** are special cases of the **case** statement.
- Keywords **while**, **for**, **repeat**, and **forever** are used for four types of looping statements in Verilog.

- *Sequential* and *parallel* are two types of blocks. Sequential blocks are specified by keywords **begin** and **end** . Parallel blocks are expressed by keywords **fork** and **join**. Blocks can be *nested* and *named*. If a block is named, the execution of the block can be disabled from anywhere in the design. Named blocks can be referenced by hierarchical names.

7.10 Exercises

1. Declare a register called *oscillate*. Initialize it to 0 and make it toggle every 30 time units. Do not use **always** statement (Hint: Use the **forever** loop).
2. Design a *clock* with time period = 40 and a duty cycle of 25% by using the **always** and **initial** statements. The value of *clock* at time = 0 should be initialized to 0.
3. Given below is an **initial** block with blocking procedural assignments. At what simulation time is each statement executed? What are the intermediate and final values of *a, b, c, d*?

```
initial
begin
    a = 1'b0;
    b = #10 1'b1;
    c = #5 1'b0;
    d = #20 {a, b, c};
end
```

4. Repeat exercise 3 if *nonblocking* procedural assignments were used.
5. What is the order of execution of statements in the following Verilog code? Is there any ambiguity in the order of execution? What are the final values of *a, b, c, d*?

```
initial
begin
        a = 1'b0;
        #0 c = b;
end
initial
begin
        b = 1'b1;
        #0 d = a;
end
```

6. What is the final value of *d* in the following example. (Hint: See *intra-assignment delays*).

```
initial
begin
        b = 1'b1; c = 1'b0;
        #10 b = 1'b0;
initial
begin
        d = #25 (b | c);
end
```

7. Design a negative edge-triggered D-flipflop (*D_FF*) with synchronous clear, active high (*D_FF* clears only at a negative edge of *clock* when clear is high). Use behavioral statements only. (Hint: Output *q* of *D_FF* must be declared as **reg**). Design a *clock* with a period of 10 units and test the *D_FF*.
8. Design the D-flipflop in exercise 7 with asynchronous clear (*D_FF* clears whenever *clear* goes high. It does not wait for next negative edge). Test the *D_FF*.
9. Using the **wait** statement, design a level-sensitive latch that takes *clock* and *d* as inputs and *q* as output. *q* = *d* whenever *clock* = *1*.
10. Design the 4-to-1 multiplexer in Example 7-14 by using **if** and **else** statements. The port interface must remain the same.
11. Design the traffic signal controller discussed in this chapter by using **if** and **else** statements.
12. Using a **case** statement, design an 8-function ALU that takes 4-bit inputs *a* and *b* and a 3-bit input signal *select*, and gives a 5-bit output *out*. The ALU implements the following functions based on a 3-bit input signal *select*. Ignore any overflow or underflow bits.

Select Signal	Function
3'b000	out = a
3'b001	out = a + b
3'b010	out = a - b
3'b011	out = a / b
3'b100	out = a % b (remainder)
3'b101	out = a << 1
3'b110	out = a >> 1
3'b111	out = (a > b) (magnitude compare)

13. Using a **while** loop, design a clock generator. Initial value of *clock* is 0. Time period for the clock is 10.
14. Using the **for** loop, initialize locations 0 to 1023 of a 4-bit register array *cache_var* to 0.
15. Using a **forever** statement, design a *clock* with time period = 10 and duty cycle = 40%. Initial value of *clock* is 0.
16. Using the **repeat** loop, delay the statement $a = a + 1$ by 20 positive edges of *clock*.
17. Below is a block with nested *sequential* and *parallel* blocks. When does the block finish and what is the order of execution of events? At what simulation times does each statement finish execution?

```
initial
begin
        x = 1'b0;
        #5 y = 1'b1;
        fork
                #20 a = x;
                #15 b = y;
        join
        #40 x = 1'b1;
        fork
                #10 p = x;
                begin
                        #10 a = y;
                        #30 b = x;
                end
                #5 m = y;
        join
end
```

18. Design an 8-bit *counter* by using a **forever** loop, *named block*, and *disabling of named block*. The *counter* starts counting at count = 5 and finishes at count = 67. The count is incremented at positive edge of *clock*. The *clock* has a time period of 10. The *counter* counts through the loop only once and then is disabled. (Hint: Use the **disable** statement).

Tasks and Functions 8

A designer is frequently required to implement the same functionality at many places in a behavioral design. This means that the commonly used parts should be abstracted into routines and the routines must be invoked instead of repeating the code. Most programming languages provide procedures or subroutines to accomplish this. Verilog provides *tasks* and *functions* to break up large behavioral designs into smaller pieces. Tasks and functions allow the designer to abstract Verilog code that is used at many places in the design.

Tasks have **input**, **output**, and **inout** arguments; functions have **input** arguments. Thus, values can be passed into and out from tasks and functions. Considering the analogy of FORTRAN, tasks are similar to *SUBROUTINE* and functions are similar to *FUNCTION*.

Tasks and functions are included in the design hierarchy. Like named blocks, tasks or functions can be addressed by means of hierarchical names.

Learning Objectives

- Describe the differences between tasks and functions.
- Identify the conditions required for tasks to be defined. Understand task declaration and invocation.
- Explain the conditions necessary for functions to be defined. Understand function declaration and invocation.

8.1 Differences Between Tasks and Functions

Tasks and functions serve different purposes in Verilog. We discuss tasks and functions in greater detail in the following sections. However, first it is important to understand differences between tasks and functions, as outlined in Table 8-1.

Table 8-1 Tasks and Functions

Functions	Tasks
A function can enable another function but not another task.	A task can enable other tasks and functions.
Functions always execute in 0 simulation time.	Tasks may execute in non-zero simulation time.
Functions must not contain any delay, event, or timing control statements.	Tasks may contain delay, event, or timing control statements.
Functions must have at least one **input** argument. They can have more than one **input**.	Tasks may have zero or more arguments of type **input, output or inout**.
Functions always return a single value. They cannot have **output** or **inout** arguments.	Tasks do not return with a value but can pass multiple values through **output** and **inout** arguments.

Both tasks and functions must be defined in a module and are local to the module. Tasks are used for common Verilog code that contains delays, timing, event constructs, or *multiple* output arguments. Functions are used when common Verilog code is purely combinational, executes in zero simulation time and provides exactly *one* output. Functions are typically used for conversions and commonly used calculations.

Tasks can have **`input`**, **`output`**, and **`inout`** ports; functions can have **`input`** ports. In addition, they can have local variables, registers, time variables, integers, real, or events. Tasks or functions cannot have wires. Tasks and functions contain behavioral statements only. Tasks and functions do not contain **`always`** or **`initial`** statements but are called from **`always`** blocks, **`initial`** blocks, or other tasks and functions.

8.2 Tasks

Tasks are declared with the keywords **`task`** and **`endtask`**. Tasks must be used if any one of the following conditions is true for the procedure.

- There are delay, timing, or event control constructs in the procedure.
- The procedure has zero or more than one output arguments.
- The procedure has no input arguments.

8.2.1 Task Declaration and Invocation

Task declaration and task invocation syntax is as follows.

```
//Task Declaration Syntax
<task>
        ::= task <name_of_task>;
            <tf_declaration>*
            <statement_or_null>
            endtask

<name_of_task>
        ::= <IDENTIFIER>

<tf_declaration>
        ::= <parameter_declaration>
        ||= <input_declaration>
        ||= <output_declaration>
        ||= <inout_declaration>
        ||= <reg_declaration>
        ||= <time_declaration>
        ||= <integer_declaration>
        ||= <real_declaration>
        ||= <event_declaration>


//Task Invocation Syntax
<task_enable>
        ::= <name_of_task>;
        ||= <name_of_task> (<expression><,<expression>>*);
```

I/O declarations use keywords **input**, **output** or **inout**, based on the type of argument declared. *Input* and *inout* arguments are passed into the task. *Input* arguments are processed in the task statements. *Output* and *inout* argument values are passed back to the variables in the task invocation statement when the task is completed. Tasks can invoke other tasks or functions.

Although the keywords **input**, **inout**, and **output** used for I/O arguments in a task are the same as the keywords used to declare ports in modules, there is a difference. Ports are used to connect external signals to the module. I/O arguments in a task are used to pass values to and from the task.

8.2.2 Task Examples

We discuss two examples of tasks. The first example illustrates the use of input and output arguments in tasks. The second example models an asymmetric sequence generator that generates an asymmetric sequence on the clock signal.

Use of input and output arguments

Example 8-1 illustrates the use of **input** and **output** arguments in tasks. Consider a task called *bitwise_oper,* which computes the *bitwise and, bitwise or,* and *bitwise ex-or* of two 16-bit numbers. The two 16-bit numbers *a* and *b* are inputs and the three outputs are 16-bit numbers *ab_and, ab_or, ab_xor.* A parameter *delay* is also used in the task.

Example 8-1 Input and Output Arguments in Tasks

```
//Define a module called operation that contains the task bitwise_oper
module operation;
...
...
parameter delay = 10;
reg [15:0] A, B;
reg [15:0] AB_AND, AB_OR, AB_XOR;

always @(A or B) //whenever A or B changes in value
begin
       //invoke the task bitwise_oper. provide 2 input arguments A, B
        //Expect 3 output arguments AB_AND, AB_OR, AB_XOR
       //The arguments must be specified in the same order as they
       //appear in the task declaration.
        bitwise_oper(AB_AND, AB_OR, AB_XOR, A, B);
end
...
...
//define task bitwise_oper
task bitwise_oper;
output [15:0] ab_and, ab_or, ab_xor; //outputs from the task
input [15:0] a, b; //inputs to the task
begin
        #delay ab_and = a & b;
        ab_or = a | b;
        ab_xor = a ^ b;
```

Example 8-1 *Input and Output Arguments in Tasks (Continued)*

```
end
endtask
...
endmodule
```

In the above task, the input values passed to the task are *A* and *B*. Hence, when the task is entered, *a* = *A* and *b* = *B*. The three output values are computed after a delay. This delay is specified by the parameter *delay*, which is 10 units for this example. When the task is completed, the output values are passed back to the calling output arguments. Therefore, *AB_AND* = *ab_and*, *AB_OR* = *ab_or*, and *AB_XOR* = *ab_xor* when the task is completed.

Asymmetric Sequence Generator

Tasks can directly operate on **reg** variables defined in the module. Example 8-2 directly operates on the **reg** variable variable clock to continuously produce an asymmetric sequence. The *clock* is initialized with an initialization sequence.

Example 8-2 *Direct Operation on reg Variables*

```
//Define a module that contains the task asymmetric_sequence
module sequence;
...
reg clock;
...
initial
        init_sequence; //Invoke the task init_sequence
...
always
begin
        asymmetric_sequence; //Invoke the task asymmetric_sequence
end
...
...
//Initialization sequence
task init_sequence;
begin
        clock = 1'b0;
end
endtask
```

Example 8-2 Direct Operation on reg Variables (Continued)

```
//define task to generate asymmetric sequence
//operate directly on the clock defined in the module.
task asymmetric_sequence;
begin
        #12 clock = 1'b0;
        #5 clock = 1'b1;
        #3 clock = 1'b0;
        #10 clock = 1'b1;
end
endtask
...
...
endmodule
```

8.3 Functions

Functions are declared with the keywords **function** and **endfunction**. Functions are used if all of the following conditions are true for the procedure.

- There are no delay, timing, or event control constructs in the procedure.
- The procedure returns a single value.
- There is at least one input argument.

8.3.1 Function Declaration and Invocation

The syntax of functions is follows:

```
//Function Declaration Syntax
<function>
        ::= function <range_or_type>? <name_of_function>;
            <tf_declaration>+
            <statement>
            endfunction

<range_or_type>
        ::= <range>
        |= <INTEGER>
        |= <REAL>
```

```
<name_of_function>
        ::= <IDENTIFIER>

<tf_declaration>
        ::= <parameter_declaration>
        ||= <input_declaration>
        ||= <reg_declaration>
        ||= <time_declaration>
        ||= <integer_declaration>
        ||= <real_declaration>


//Function Invocation Syntax
<function_call>
        ::= <name_of_function> (<expression><,<expression>>*)
```

There are some peculiarities of functions. When a function is declared, a register with name *<name_of_function>* is declared implicitly inside Verilog. The output of a function is passed back by setting the value of the register *<name_of_function>* appropriately. The function is invoked by specifying function name and input arguments. At the end of function execution, the return value is placed where the function was invoked. The optional *<range_or_type>* specifies the width of the internal register. If no range or type is specified, the default bit width is 1. Functions are very similar to *FUNCTION* in FORTRAN.

Notice that at least one input argument must be defined for a function. There are no output arguments for functions because the implicit register *<name_of_function>* contains the output value. Also, functions *cannot* invoke other tasks. They can only invoke other functions.

8.3.2 Function Examples

We discuss two examples. The first example models a parity calculator that returns a 1-bit value. The second example models a 32-bit left/right shift register that returns a 32-bit shifted value.

Parity calculation

Let us discuss a function that calculates the parity of a 32-bit *address* and returns the value. We assume even parity. Example 8-3 shows the definition and invocation of the function *calc_parity.*

Example 8-3 Parity Calculation

```
//Define a module that contains the function calc_parity
module parity;
...
reg [31:0] addr;
reg parity;

//Compute new parity whenever address value changes
always @(addr)
begin
       parity = calc_parity(addr); //First invocation of calc_parity
        $display("Parity calculated = %b", calc_parity(addr) );
                                    //Second invocation of calc_parity
end
...
...
//define the parity calculation function
function calc_parity;
input [31:0] address;
begin
        //set the output value appropriately. Use the implicit
        //internal register calc_parity.
       calc_parity = ^address; //Return the xor of all address bits.
end
endfunction
...
...
endmodule
```

Note that in the first invocation of *calc_parity,* the returned value was used to set the **reg** *parity.* In the second invocation, the value returned was directly used inside the **$display** task. Thus, the returned value is placed wherever the function was invoked.

Left/right shifter

To illustrate how a range for the output value of a function can be specified, let us consider a function that shifts a 32-bit value to the left or right by one bit, based on a *control* signal. Example 8-4 shows the implementation of the left/right shifter.

Example 8-4 Left/Right Shifter

```
//Define a module that contains the function shift
module shifter;
…
//Left/right shifter
`define LEFT_SHIFT      1'b0
`define RIGHT_SHIFT     1'b1
reg [31:0] addr, left_addr, right_addr;
reg control;

//Compute the right- and left-shifted values whenever
//a new address value appears
always @(addr)
begin
       //call the function defined below to do left and right shift.
        left_addr = shift(addr, `LEFT_SHIFT);
        right_addr = shift(addr, `RIGHT_SHIFT);
end
…
…
//define shift function. The output is a 32-bit value.
function [31:0] shift;
input [31:0] address;
input control;
begin
       //set the output value appropriately based on a control signal.
        shift = (control == `LEFT_SHIFT) ?(address << 1) : (address
>> 1);

end
endfunction
…
…
endmodule
```

8.4 Summary

In this chapter we discussed tasks and functions used in behavior Verilog modeling.

- *Tasks* and *functions* are used to define common Verilog functionality that is used at many places in the design. Tasks and functions help to make a module definition more readable by breaking it up into manageable subunits. Tasks and functions serve the same purpose in Verilog as subroutines do in C.
- Tasks can take any number of **`input`**, **`inout`** or **`output`** arguments. Delay, event, or timing control constructs are permitted in tasks. Tasks can enable other tasks or functions.
- Functions are used when exactly one return value is required and at least one input argument is specified. Delay, event, or timing control constructs are not permitted in functions. Functions can invoke other functions but cannot invoke other tasks.
- A register with name as the function name is declared implicitly when a function is declared. The return value of the function is passed back in this register.
- Tasks and functions are included in a design hierarchy and can be addressed by hierarchical name referencing.

8.5 Exercises

1. Define a **`function`** to calculate the *factorial* of a 4-bit number. The output is a 32-bit value. Invoke the function by using stimulus and check results.
2. Define a **`function`** to multiply two 4-bit numbers *a* and *b*. The output is an 8-bit value. Invoke the function by using stimulus and check results.
3. Define a **`function`** to design an 8-function ALU that takes two 4-bit numbers *a* and *b* and computes a 5-bit result *out* based on a 3-bit *select* signal. Ignore overflow or underflow bits.

Select Signal	Function Output
3'b000	a
3'b001	a + b
3'b010	a - b

Select Signal	Function Output
3'b011	a / b
3'b100	a % 1 (remainder)
3'b101	a << 1
3'b110	a >> 1
3'b111	(a > b) (magnitude compare)

4. Define a **task** to compute the factorial of a 4-bit number. The output is a 32-bit value. The result is assigned to the output after a delay of 10 time units.
5. Define a **task** to compute even parity of a 16-bit number. The result is a 1-bit value that is assigned to the output after three positive edges of clock. (Hint: Use a **repeat** loop in the task).
6. Using named events, tasks, and functions, design the traffic signal controller in *Traffic Signal Controller* on page 147.

Useful Modeling Techniques 9

We learned the basic features of Verilog in the preceding chapters. In this chapter we discuss additional features that enhance the Verilog language and make it powerful and flexible for modeling and analyzing a design.

Learning Objectives

- Describe procedural continuous assignment statements **assign**, **deassign**, **force**, and **release**. Explain their significance in modeling and debugging.
- Understand how to override parameters by using the **defparam** statement at the time of module instantiation.
- Explain conditional compilation and execution of parts of the Verilog description.
- Identify system tasks for *file output, displaying hierarchy, strobing, random number generation, memory initialization,* and *value change dump.*

9.1 Procedural Continuous Assignments

We studied procedural assignments in Section 7.2, *Procedural Assignments.* Procedural assignments assign a value to a register. The value stays in the register until another procedural assignment puts another value in that register. *Procedural continuous assignments* behave differently. They are procedural statements which allow values of expressions to be driven continuously onto registers or nets for limited periods of time. Procedural continuous assignments override existing assignments to a register or net. They provide an useful extension to the regular procedural assignment statement.

9.1.1 assign and deassign

The keywords **assign** and **deassign** are used to express the first type of procedural continuous assignment. The left-hand side of procedural continuous assignments can only be a *register* or a *concatenation of registers.* It cannot be a *part*

or bit select of a net or *an array* of *registers*. Procedural continuous assignments override the effect of regular procedural assignments. Procedural continuous assignments are normally used for controlled periods of time.

A simple example is the negative edge-triggered D-flipflop with asynchronous reset that we modeled in Example 6-8. In Example 9-1 we now model the same *D_FF*, using **assign** and **deassign** statements.

Example 9-1 *D-flipflop with Procedural Continuous Assignments*

```
// Negative edge-triggered D-flipflop with asynchronous reset
module edge_dff(q, qbar, d, clk, reset);

// Inputs and outputs
output q,qbar;
input d, clk, reset;
reg q, qbar; //declare q and qbar are registers

always @(negedge clk) //assign value of q & qbar at active edge of clock.
begin
        q = d;
        qbar = ~d;
end

always @(reset) //Override the regular assignments to q and qbar
              //whenever reset goes high. Use of procedural continuous
                 //assignments.
        if(reset)
      begin  //if reset is high,override regular assignments to q with
            //the new values, using procedural continuous assignment.
               assign q = 1'b0;
               assign qbar = 1'b1;
        end
        else
        begin   //If reset goes low, remove the overriding values by
                //deassigning the registers. After this the regular
             //assignments q = d and qbar = ~d will be able to change
            //the registers on the next negative edge of clock.
               deassign q;
               deassign qbar;
        end

endmodule
```

In Example 9-1, we overrode the assignment on *q* and *qbar* and assigned new values to them when *reset* signal went high. The register variables retain the continuously assigned value after the **deassign** until they are changed by a future procedural assignment.

9.1.2 force and release

Keywords **force** and **release** are used to express the second form of the procedural continuous assignments. They can be used to override assignments on both registers and nets. **force** and **release** statements are typically used in the interactive debugging process, where certain registers or nets are forced to a value and the effect on other registers and nets is noted. It is recommended that **force** and **release** statements not be used inside design blocks. They should appear only in stimulus or as debug statements.

force and release on registers

A **force** on a register overrides any procedural assignments or procedural continuous assignments on the register until the register is released. The register variables will continue to store the forced value after being released but can then be changed by a future procedural assignment. To override the values of *q* and *qbar* in Example 9-1 for a limited period of time, we could do the following.

```
module stimulus;
...
...
//instantiate the d-flipflop
edge_dff dff(Q, Qbar, D, CLK, RESET);
...
...
initial
begin
    //these statements force value of 1 on dff.q between time 50 and
     //100, regardless of the actual output of the edge_dff.
    #50 force dff.q = 1'b1; //force value of q to 1 at time 50.
    #50 release dff.q;  //release the value of q at time 100.
end
...
...
endmodule
```

force and release on nets

force on nets overrides any continuous assignments until the net is released. The net will immediately return to its normal driven value when it is released. A net can be forced to an expression or a value.

```
module top;
...
...
assign out = a & b & c; //continuous assignment on net out
...
initial
    #50 force out = a | b & c;
    #50 release out;
end
...
...
endmodule
```

In the example above, a new expression is forced on the net from time 50 to time 100. From time 50 to time 100, when the **force** statement is active, the expression *a | b & c* will be reevaluated and assigned to *out* whenever values of signals *a* or *b* or *c* change. Thus, the **force** statement behaves like a continuous assignment except that it is active only for a limited period of time.

9.2 Overriding Parameters

Parameters can be defined in a module definition, as was discussed earlier in Section 3.2.8, *Parameters*. However, during compilation of Verilog modules, parameter values can be altered separately for each module instance. This allows us to pass a distinct set of parameter values to each module during compilation regardless of predefined parameter values.

There are two ways to override parameter values: through the *defparam statement* or through *module instance parameter value assignment*.

9.2.1 defparam Statement

Parameter values can be changed in any module instance in the design with the keyword **defparam**. The hierarchical name of the module instance can be used to override parameter values. Consider Example 9-2, which uses **defparam** to override the parameter values in module instances.

Example 9-2 Defparam Statement

```
//Define a module hello_world
module hello_world;
parameter id_num = 0; //define a module identification number = 0

initial //display the module identification number
        $display("Displaying hello_world id number = %d", id_num);
endmodule


//define top-level module
module top;
//change parameter values in the instantiated modules
//Use defparam statement
defparam w1.id_num = 1, w2.id_num = 2;

//instantiate two hello_world modules
hello_world w1();
hello_world w2();

endmodule
```

In Example 9-2,the module *hello_world* was defined with a default *id_num* = 0. However, when the module instances *w1* and *w2* of the type *hello_world* are created, their *id_num* values are modified with the **defparam** statement. If we simulate the above design, we would get the following output:

```
Displaying hello_world id number = 1
Displaying hello_world id number = 2
```

Multiple **defparam** statements can appear in a module. Any parameter can be overridden with the **defparam** statement.

9.2.2 Module_Instance Parameter Values

Parameter values can be overridden when a module is instantiated. To illustrate this, we will use Example 9-2 and modify it a bit. The new parameter values are passed during module instantiation. The top-level module can pass parameters to

the instances *w1* and *w2* as shown below. Notice that **defparam** is not needed. The simulation output will be identical to the output obtained with the **defparam** statement.

```
//define top-level module
module top;

//instantiate two hello_world modules; pass new parameter values
hello_world #(1) w1; //pass value 1 to module w1
hello_world #(2) w2; //pass value 2 to module w2

endmodule
```

If multiple parameters are defined in the module, during module instantiation they can be overridden by specifying the new values in the same order as the parameter declarations in the module. If an overriding value is not specified, the default parameter declaration values are taken. Consider Example 9-3.

Example 9-3 Module Instance Parameter Values

```
//define module with delays
module bus_master;
parameter delay1 = 2;
parameter delay2 = 3;
parameter delay3 = 7;
...
<module internals>
...
endmodule

//top-level module; instantiates two bus_master modules
module top;

//Instantiate the modules with new delay values
bus_master #(4, 5, 6) b1(); //b1: delay1 = 4, delay2 = 5, delay3 = 6
bus_master #(9,4) b2(); //b2: delay1 = 9, delay2 = 4, delay3 = 7(default)

endmodule
```

Module-instance, parameter value assignment is a very useful method used to override parameter values and to customize module instances.

9.3 Conditional Compilation and Execution

A portion of Verilog might be suitable for one environment and not for the other. The designer does not wish to create two versions of Verilog design for the two environments. Instead, the designer can specify that the particular portion of the code be compiled only if a certain flag is set. This is called *conditional compilation.*

A designer might also want to execute certain parts of the Verilog design only when a flag is set at run time. This is called *conditional execution.*

9.3.1 Conditional Compilation

Conditional compilation can be accomplished by using compiler directives **`ifdef**, **`else**, and **`endif**. Example 9-4 contains Verilog source code to be compiled conditionally.

Example 9-4 Conditional Compilation

```
//Conditional Compilation

//Example 1
`ifdef TEST //compile module test only if text macro TEST is defined
module test;
...
...
endmodule
`else //compile the module stimulus as default
module stimulus;
...
...
endmodule
`endif //completion of `ifdef statement

//Example 2
module top;

bus_master b1(); //instantiate module unconditionally
`ifdef ADD_B2
   bus_master b2(); //b2 is instantiated conditionally if text macro
                    //ADD_B2 is defined
`endif

endmodule
```

The `ifdef statement can appear anywhere in the design. A designer can conditionally compile statements, modules, blocks, declarations, and other compiler directives. The `else statement is optional. A maximum of one `else statement can accompany the `ifdef. An `ifdef is always closed by a corresponding `endif.

The conditional compile flag can be set by using the `define statement inside the Verilog file. In the example above, we could define the flags by defining text macros *TEST* and *ADD_B2* at compile time by using the `define statement. The Verilog compiler simply skips the portion if the conditional compile flag is not set. A boolean expression, such as *TEST && ADD_B2*, is not allowed with the `ifdef statement.

9.3.2 Conditional Execution

Conditional execution flags allow the designer to control statement execution flow at run time. All statements are compiled but executed conditionally. Conditional execution flags can be used only for behavioral statements. The system task keyword **$test$plusargs** is used for conditional execution. This option is not provided as a part of the *IEEE Language Reference Manual*. This facility is available in Verilog-XL but may not be supported in other simulators because it is not a standard.

Consider Example 9-5, which illustrates conditional execution.

Example 9-5 Conditional Execution

```
//Conditional execution
module test;
...
...
initial
begin
   if($test$plusargs("DISPLAY_VAR"))
      $display("Display = %b ", {a,b,c} ); //display only if flag is set
   else
      $display("No Display"); //otherwise no display
end
endmodule
```

The variables are displayed only if the flag *DISPLAY_VAR* is set at run time. Flags can be set at run time by specifying the option *+DISPLAY_VAR* at run time.

9.4 Time Scales

Often, in a single simulation, delay values in one module need to be defined by using certain time unit, e.g., 1 µs, and delay values in another module need to be defined by using a different time unit, e.g., 100 ns. Verilog HDL allows the reference time unit for modules to be specified with the **`timescale** compiler directive.

Usage: **`timescale** *<reference_time_unit> / <time_precision>*

The *<reference_time_unit>* specifies the unit of measurement for times and delays. The *<time_precision>* specifies the precision to which the delays are rounded off during simulation. Only 1, 10, and 100 are valid integers for specifying time unit and time precision. Consider the two modules, *dummy1* and *dummy2*, in Example 9-6.

Example 9-6 Time Scales

```
//Define a time scale for the module dummy1
//Reference time unit is 100 nanoseconds and precision is 1 ns
`timescale 100 ns / 1 ns

module dummy1;

reg toggle;

//initialize toggle
initial
  toggle = 1'b0;

//Flip the toggle register every 5 time units
//In this module 5 time units = 500 ns = .5 µs
always #5
    begin
        toggle = ~toggle;
        $display("%d , In %m toggle = %b ", $time, toggle);
    end

endmodule

//Define a time scale for the module dummy2
//Reference time unit is 1 microsecond and precision is 10 ns
`timescale 1 us / 10 ns

module dummy2;
```

Example 9-6 Time Scales (Continued)

```
reg toggle;

//initialize toggle
initial
  toggle = 1'b0;

//Flip the toggle register every 5 time units
//In this module 5 time units = 5 µs  = 5000 ns
always #5
    begin
        toggle = ~toggle;
        $display("%d , In %m toggle = %b ", $time, toggle);
    end

endmodule
```

The two modules *dummy1* and *dummy2* are identical in all respects, except that the time unit for *dummy1* is 100 ns and time unit for *dummy2* is 1 µs. Thus the **$display** statement in *dummy1* will be executed 10 times for each **$display** executed in *dummy2*. The **$time** task reports the simulation time in terms of the reference time unit for the module in which it is invoked. The first few **$display** statements are shown in the simulation output below to illustrate the effect of the **`timescale** directive.

```
                5 , In dummy1 toggle = 1
               10 , In dummy1 toggle = 0
               15 , In dummy1 toggle = 1
               20 , In dummy1 toggle = 0
               25 , In dummy1 toggle = 1
               30 , In dummy1 toggle = 0
               35 , In dummy1 toggle = 1
               40 , In dummy1 toggle = 0
               45 , In dummy1 toggle = 1
    -->         5 , In dummy2 toggle = 1
               50 , In dummy1 toggle = 0
               55 , In dummy1 toggle = 1
```

Notice that the **$display** statement in *dummy2* executes once for every ten **$display** statements in *dummy1*.

9.5 Useful System Tasks

In this section we discuss the system tasks that are useful for a variety of purposes in Verilog. We discuss system tasks for *file output, displaying hierarchy, strobing, random number generation, memory initialization,* and *value change dump.*

9.5.1 File Output

Output from Verilog normally goes to the standard output and the file *verilog.log*. It is possible to redirect the output of Verilog to a chosen file.

Opening a file

A file can be opened with the system task **$fopen**.

Usage: **$fopen**(*"<name_of_file>"*);

Usage: *<file_handle>* = **$fopen**(*"<name_of_file>"*);

The task **$fopen** returns a 32-bit value called a *multichannel descriptor.* Only one bit is set in a multichannel descriptor. The standard output has a multichannel descriptor with the least significant bit (bit 0) set. Standard output is also called channel 0. The standard output is always open. Each successive call to **$fopen** opens a new channel and returns a 32-bit descriptor with bit 1 set, bit 2 set, and so on, up to bit 31 set. The channel number corresponds to the individual bit set in the multichannel descriptor. Example 9-7 illustrates the use of file descriptors.

Example 9-7 File Descriptors

```
//Multichannel descriptor
integer handle1, handle2, handle3; //integers are 32-bit values

//standard output is open; descriptor = 32'h0000_0001 (bit 0 set)
initial
begin
   handle1 = $fopen("file1.out"); //handle1 = 32'h0000_0002 (bit 1 set)
   handle2 = $fopen("file2.out"); //handle2 = 32'h0000_0004 (bit 2 set)
   handle3 = $fopen("file3.out"); //handle3 = 32'h0000_0008 (bit 3 set)
end
```

The advantage of multichannel descriptors is that it is possible to selectively write to multiple files at the same time. This is explained below in greater detail.

Writing to files

The system tasks **$fdisplay**, **$fmonitor**, **$fwrite**, and **$fstrobe** are used to write to files. Note that these tasks are the similar in syntax to regular system tasks **$display**, **$monitor**, etc., but they provide the additional capability of writing to files.

We will consider only **$fdisplay** and **$fmonitor** tasks.

Usage: **$fdisplay**(*<file_descriptor>, p1, p2 ..., pn);*

$fmonitor(*<file_descriptor>, p1, p2,..., pn);*

p1, p2, ..., pn can be variables, signal names, or quoted strings.

A *file_descriptor* is a multichannel descriptor that can be a file handle or a bitwise combination of file handles. Verilog will write the output to all files that have a 1 associated with them in the file descriptor. We will use the file descriptors defined in Example 9-7 and illustrate the use of the **$fdisplay** and **$fmonitor** tasks.

```
//All handles defined in Example 9-7
//Writing to files
integer desc1, desc2, desc3; //three file descriptors
initial
begin
    desc1 = handle1 | 1; //bitwise or; desc1 = 32'h0000_0003
    $fdisplay(desc1, "Display 1");//write to files file1.out & stdout

    desc2 = handle2 | handle1; //desc2 = 32'h0000_0006
    $fdisplay(desc2, "Display 2");//write to files file1.out & file2.out

    desc3 = handle3 ; //desc3 = 32'h0000_0008
    $fdisplay(desc3, "Display 3");//write to file file3.out only
end
```

Closing files

Files can be closed with the system task **$fclose**.

Usage: **$fclose**(*<file_handle>);*

```
//Closing Files
$fclose(handle1);
```

A file cannot be written to once it is closed. The corresponding bit in the multichannel descriptor is set to 0. The next **$fopen** call can reuse the bit.

9.5.2 Displaying Hierarchy

Hierarchy at any level can be displayed by means of the *%m* option in any of the display tasks, **$display**, **$write** task, **$monitor**, or **$strobe** task as discussed briefly in Section 4.3, *Hierarchical Names*. This is a very useful option. For example, when multiple instances of a module execute the same Verilog code, the *%m* option will distinguish which model instance the output is coming from. No argument is needed for the *%m* option in the display tasks.

Example 9-8 *Displaying Hierarchy*

```
//Displaying hierarchy information
module M;
...
initial
   $display("Displaying in %m");
endmodule

//instantiate module M
module top;
...
M  m1();
M  m2();
M  m3();
endmodule
```

The output from the simulation will look like the following:

```
Displaying in top.m1
Displaying in top.m2
Displaying in top.m3
```

This feature can display full hierarchical names, including module instances, tasks, functions, and named blocks.

9.5.3 Strobing

Strobing is done with the system task keyword **$strobe**. This task is very similar to the **$display** task except for a slight difference. If many other statements are executed in the same time unit as the **$display** task, the order in which the statements and the **$display** task are executed is nondeterministic. If **$strobe** is used, it is always executed after all other assignment statements in the same time unit have executed. Thus, **$strobe** provides a synchronization mechanism to ensure that data is displayed only after all other assignment statements, which change the data in that time step, have executed.

Example 9-9 Strobing

```
//Strobing
always @(posedge clock)
begin
   a = b;
   c = d;
end

always @(posedge clock)
   $strobe("Displaying a=%b, c=%b", a, c); //display values at posedge
```

In Example 9-9, the values at positive edge of clock will be displayed only after statements $a = b$ and $c = d$ execute. If **$display** was used, **$display** might execute before statements $a = b$ and $c = d$, thus displaying different values.

9.5.4 Random Number Generation

Random number generation capabilities are required for generating a random set of test vectors. Random testing is important because it often catches hidden bugs in the design. Random vector generation is also used in performance analysis of chip architectures. The system task **$random** is used for generating a random number.

Usage: **$random**;

$random(<*seed*>);

The value of <*seed*> is optional and is used to ensure the same random number sequence each time the test is run. The task **$random** returns a 32-bit random number. All bits, bit-selects, or part-selects of the 32-bit random number can be used (see Example 9-10).

Example 9-10 Random Number Generation

```
//Generate random numbers and apply them to a simple ROM
module test;
integer r_seed;
reg [31:0] addr;//input to ROM
wire [31:0] data;//output from ROM
...
...
ROM rom1(data, addr);

initial
   r_seed = 2; //arbitrarily define the seed as 2.

always @(posedge clock)
   addr = $random(r_seed); //generates random numbers
...
<check output of ROM against expected results>
...
...
endmodule
```

Note that the algorithm used by **$random** is not standardized and may vary among simulators.

9.5.5 Initializing Memory from File

We discussed how to declare memories in Section 3.2.7, *Memories*. Verilog provides a very useful system task to initialize memories from a data file. Two tasks are provided to read numbers in binary or hexadecimal format. Keywords **$readmemb** and **$readmemh** are used to initialize memories.

Usage: **$readmemb**(*"<file_name>", <memory_name>);*

$readmemb(*"<file_name>", <memory_name>, <start_addr>);*

$readmemb(*"<file_name>", <memory_name>, <start_addr>, <finish_addr>);*

Identical syntax for **$readmemh**.

The *<file_name>* and *<memory_name>* are mandatory; *<start_addr>* and *<finish_addr>* are optional. Defaults are start index of memory array for *<start_addr>* and end of the data file or memory for *<finish_addr>*. Example 9-11 illustrates how memory is initialized.

Example 9-11 Initializing Memory

```
module test;

reg [7:0] memory[0:7]; //declare an 8-byte memory
integer i;

initial
begin
  //read memory file init.dat. address locations given in memory
  $readmemb("init.dat", memory);
  //display contents of initialized memory
  for(i=0; i < 8; i = i + 1)
    $display("Memory [%0d] = %b", i, memory[i]);
end

endmodule
```

The file *init.dat* contains the initialization data. Addresses are specified in the data file with @<*address*>. Addresses are specified as hexadecimal numbers. Data is separated by whitespaces. Data can contain **x** or **z**. Uninitialized locations default to **x**. A sample file, *init.dat*, is shown below.

```
@002
11111111 01010101
00000000 10101010

@006
1111zzzz 00001111
```

When the test module is simulated, we will get the following output.

```
Memory [0] = xxxxxxxx
Memory [1] = xxxxxxxx
Memory [2] = 11111111
Memory [3] = 01010101
Memory [4] = 00000000
Memory [5] = 10101010
Memory [6] = 1111zzzz
Memory [7] = 00001111
```

9.5.6 Value Change Dump File

A *value change dump* (*VCD*) is an ASCII file that contains information about simulation time, scope and signal definitions, and signal value changes in the simulation run. All signals or a selected set of signals in a design can be written to a VCD file during simulation. *Postprocessing tools* can take the VCD file as input and visually display hierarchical information, signal values, and signal waveforms. Many postprocessing tools, such as *Magellan*®, *Signalscan*™, and *VirSim*™ are now commercially available. For simulation of large designs, designers dump selected signals to a VCD file and use a postprocessing tool to debug, analyze, and verify the simulation output. The use of VCD file in the debug process is shown in Figure 9-1.



Figure 9-1 Debugging and Analysis of Simulation with VCD file

System tasks are provided for selecting module instances or module instance signals to dump (**$dumpvars**), name of VCD file (**$dumpfile**), starting and stopping the dump process (**$dumpon**, **$dumpoff**), and generating checkpoints (**$dumpall**). Uses of each task are shown in Example 9-12.

Example 9-12 VCD File System Tasks

```
//specify name of VCD file. Otherwise,default name is
//assigned by the simulator.
initial
      $dumpfile("myfile.dmp"); //Simulation info dumped to myfile.dmp

//Dump signals in a module
initial
   $dumpvars; //no arguments, dump all signals in the design
initial
   $dumpvars(1, top); //dump variables in module instance top.
```

Example 9-12 VCD File System Tasks (Continued)

```
                //Number 1 indicates levels of hierarchy. Dump one
               //hierarchy level below top, i.e., dump variables in top,
              //but not signals in modules instantiated by top.
initial
   $dumpvars(2, top.m1);//dump up to 2 levels of hierarchy below top.m1
initial
   $dumpvars(0, top.m1);//Number 0 means dump the entire hierarchy
                        // below top.m1

//Start and stop dump process
initial
begin
   $dumpon;             //start the dump process.
   #100000 $dumpoff;  //stop the dump process after 100,000 time units
end

//Create a checkpoint. Dump current value of all VCD variables
initial
   $dumpall;
```

The **$dumpfile** and **$dumpvars** tasks are normally specified at the beginning of the simulation. The **$dumpon**, **$dumpoff**, and **$dumpall** control the dump process during the simulation.

Postprocessing tools with graphical displays have emerged as an important part of the simulation and debug process. For large simulation runs, it is very difficult for the designer to analyze the output from **$display** or **$monitor** statements. It is more intuitive to analyze results from graphical waveforms. Formats other than VCD have also emerged, but VCD still remains the popular dump format for Verilog simulators.

However, it is important to note that VCD files can become very large (hundreds of megabytes for large designs). It is important to selectively dump only those signals that need to be examined.

9.6 Summary

In this chapter we discussed the following aspects of Verilog.

- *Procedural continuous assignments* can be used to override the assignments on registers and nets. **assign** and **deassign** can override assignments on registers. **force** and **release** can override assignments on registers and nets. **assign** and **deassign** are used in the actual design. **force** and **release** are used for debugging.
- Parameters defined in a module can be overridden with the **defparam** statement or by passing a new value during *module instantiation.*
- Compilation of parts of the design can be made conditional by using the **`ifdef** statement. Compilation flags are defined at compile time by using the **`define** statement.
- Execution is made conditional in Verilog-XL by means of the **$test$plusargs** system task. The execution flags are defined at run time by +<*flag_name*>. The **$test$plusargs** task is not a part of the Verilog HDL standard defined in the *IEEE Language Reference Manual.*
- Up to 32 files can be opened for writing in Verilog. Each file is assigned a bit in the *multichannel descriptor.* The multichannel descriptor concept can be used to write to multiple files.
- Hierarchy can be displayed with the *%m* option in any display statement.
- *Strobing* is a way to display values at a certain time or event after all other statements in that time unit have executed.
- Random numbers can be generated with the system task **$random**. They are used for random test vector generation.
- Memory can be initialized from a data file. The data file contains addresses and data. Addresses can also be specified in memory initialization tasks.
- *Value Change Dump* is a popular format used by many designers for debugging with postprocessing tools. Verilog allows all or selected module variables to be dumped to the *VCD* file. Various system tasks are available for this purpose.

9.7 Exercises

1. Using **assign** and **deassign** statements, design a positive edge-triggered D-flipflop with asynchronous *clear* (*q=0*) and *preset* (*q=1*).
2. Using primitive gates, design a 1-bit full adder *FA*. Instantiate the full adder inside a stimulus module. Force the *sum* output to *a & b & c_in* for the time between 15 and 35 units.
3. A 1-bit full adder *FA* is defined with gates and with delay parameters as shown below.

```
// Define a 1-bit full adder
module fulladd(sum, c_out, a, b, c_in);
parameter d_sum = 0, d_cout = 0;

// I/O port declarations
output sum, c_out;
input a, b, c_in;

// Internal nets
wire s1, c1, c2;

// Instantiate logic gate primitives
xor (s1, a, b);
and (c1, a, b);

xor #(d_sum) (sum, s1, c_in); //delay on output sum is d_sum
and (c2, s1, c_in);

or  #(d_cout) (c_out, c2, c1); //delay on output c_out is d_cout

endmodule
```

Define a 4-bit full adder *fulladd4* as shown in Example 5-7 on page 74, but pass the following parameter values to the instances using the two methods discussed in the book.

Instance	Delay Values
fa0	d_sum=1, d_cout=1
fa1	d_sum=2, d_cout=2
fa2	d_sum=3, d_cout=3
fa3	d_sum=4, d_cout=4

a. Build the *fulladd4* module with **defparam** statements to change instance parameter values. Simulate the 4-bit full adder using the stimulus shown in Example 5-8 on page 74. Explain the effect of the full adder delays on the times when outputs of the adder appear. (Use delays of 20 instead of 5 used in this stimulus).

b. Build the *fulladd4* with delay values passed to instances *fa0, fa1, fa2,* and *fa3* during instantiation. Resimulate the 4-bit adder, using the stimulus above. Check if the results are identical.

4. Create a design that uses the full adder example above. Use a conditional compilation (**`ifdef**). Compile the *fulladd4* with defparam statements if the text macro *DPARAM* is defined by the **`define** statement; otherwise, compile the *fulladd4* with module instance parameter values.

5. Identify the files to which the following display statements will write.

```
//File output with multi-channel descriptor

module test;

integer handle1,handle2,handle3; //file handles

//open files
initial
begin
  handle1 = $fopen("f1.out");
  handle2 = $fopen("f2.out");
  handle3 = $fopen("f3.out");
end

//Display statements to files
initial
begin
  #5;
  $fdisplay(4, "Display Statement # 1");
  $fdisplay(15, "Display Statement # 2");
  $fdisplay(6, "Display Statement # 3");
  $fdisplay(10, "Display Statement # 4");
  $fdisplay(0, "Display Statement # 5");
end

endmodule
```

6. What will be the output of the **$display** statement shown below?

```
module top;
A a1();
endmodule

module A;
B b1();
endmodule

module B;
initial
   $display("I am inside instance %m");
endmodule
```

7. Consider the 4-bit full adder in Example 6-4 on page 104. Write a stimulus file to do random testing of the full adder. Use a random number generator to generate a 32-bit random number. Pick bits *3:0* and apply them to input *a;* pick bits *7:4* and apply them to input *b*. Use bit *8* and apply it to *c_in*. Apply 20 random test vectors and observe the output.
8. Use the 8-byte memory initialization example in Example 9-11 on page 184. Modify the file to read data in *hexadecimal*. Write a new data file with the following addresses and data values. Unspecified locations are not initialized.

Location Address	Data
1	33
2	66
4	z0
5	0z
6	01

9. Write an **initial** block that controls the VCD file. The **initial** block must do the following:
   - Set *myfile.dmp* as the output VCD file.
   - Dump all variables two levels deep in module instance *top.a1.b1.c1*.
   - Stop dumping to VCD at time 200.
   - Start dumping to VCD at time 400.
   - Stop dumping to VCD at time 500.
   - Create a checkpoint. Dump current value of all VCD variables to the dumpfile.